

# Implementing Functional Languages: a tutorial

Simon L Peyton Jones and David R Lester

平成 35 年 8 月 27 日

## 目次

<b>1</b>	<b>コア言語</b>	<b>13</b>
1.1	コア言語の概要	13
1.1.1	ローカル定義	14
1.1.2	ラムダ抽象化	15
1.1.3	構造化データ型	16
1.1.4	コンストラクタの表現	17
1.1.5	case 式	18
1.2	コア言語の構文	20
1.3	コア言語のデータ型	22
1.4	A small standard prelude	23
1.5	コア言語用のプリティプリンタ	24
1.5.1	文字列を使ったプリティプリンティング	24
1.5.2	プリティプリンティングのための抽象データ型	26
1.5.3	iseq の実装	28
1.5.4	レイアウトとインデント	30
1.5.5	演算子の優先順位	31
1.5.6	iseq のその他の便利な機能	32
1.5.7	概要	32
1.6	コア言語のパース	34
1.6.1	字句解析	35
1.6.2	構文解析のための基本的なツール	37
1.6.3	ツールを鋭利にする	38
1.6.4	コア言語の構文解析	41
1.6.5	左再帰	43
1.6.6	中置演算子の追加	43
1.6.7	まとめ	45
<b>2</b>	<b>テンプレートのインスタンス化</b>	<b>47</b>
2.1	テンプレートインスタンス化のプレビュー	47
2.1.1	事例	48
2.1.2	3つのステップ	49
2.1.3	次の redex を見つけるためのスパインの巻き戻し	50
2.1.4	スーパーコンビネータ redex	52
2.1.5	更新	53
2.1.6	定数適用形式 $\dagger$	54
2.2	状態遷移システム	55
2.3	Mark 1: 最小限のテンプレートインスタンス化グラフ簡約器	58
2.3.1	グラフ簡約の遷移規則	58

2.3.2	実装の構造	60
2.3.3	パーサ	61
2.3.4	コンパイラ	61
2.3.5	評価器	65
2.3.6	結果の書式付け	69
2.4	Mark 2: let(rec) 式	73
2.5	Mark 3: 更新機能の追加	75
2.5.1	間接参照の数を減らす	76
2.6	Mark 4: 算術演算機能の追加	78
2.6.1	算術演算の遷移規則	78
2.6.2	算術演算の実装	78
2.7	Mark 5: 構造化データ型の追加	80
2.7.1	構造化データの構築	80
2.7.2	条件式	81
2.7.3	構造化データの実装	82
2.7.4	ペア	83
2.7.5	リスト	84
2.7.6	リストの表示	85
2.8	代替実装 †	87
2.8.1	プリミティブの代替表現	87
2.8.2	ダンプの代替表現	87
2.8.3	データ値の代替表現	87
2.9	ガベージコレクション	88
2.9.1	マークスキャンコレクション	88
2.9.2	間接参照の排除	90
2.9.3	ポインタ反転	92
2.9.4	2 スペースガベージコレクション	96
<b>3</b>	<b>G マシン</b>	<b>99</b>
3.1	G マシンの紹介	99
3.1.1	例	100
3.1.2	さらなる最適化	102
3.2	テンプレートを構築するためのコード シーケンス	103
3.2.1	算術の後置評価	104
3.2.2	後置コードを使用してグラフを作成する	106
3.2.3	インスタンス化が行われた後はどうなりますか?	106
3.3	Mark 1: 最小限の G マシン	108
3.3.1	全体構造	108
3.3.2	データ型の定義	108
3.3.3	評価器	112

3.3.4	プログラムのコンパイル	116
3.3.5	結果の表示	120
3.3.6	Mark 1 G マシンの改良	122
3.4	Mark 2: 遅延評価させる	123
3.4.1	データ構造	124
3.4.2	評価器	125
3.4.3	コンパイラ	126
3.5	Mark 3: let(rec) 式	127
3.5.1	ローカルにバインドされた変数	129
3.5.2	データ構造	130
3.5.3	評価器	131
3.5.4	コンパイラ	131
3.6	Mark 4: プリミティブの追加	136
3.6.1	データ構造	137
3.6.2	状態の表示	138
3.6.3	新しい命令遷移	140
3.6.4	コンパイラ	144
3.7	Mark 5: 算術演算のより良い処理に向けて	146
3.7.1	問題点	146
3.7.2	解決策	147
3.8	Mark 6: データ構造の追加	151
3.8.1	概要	151
3.8.2	データ構造	152
3.8.3	結果の表示	153
3.8.4	命令セット	154
3.8.5	コンパイラ	157
3.8.6	比較における新しい論理表現の使用	158
3.8.7	使用可能な言語の拡大	159
3.9	Mark 7: さらなる改善	162
3.9.1	V スタックによる階乗関数の実行	162
3.9.2	データ構造	164
3.9.3	命令セット	166
3.9.4	コンパイラ	167
3.10	結論	176
<b>4</b>	<b>TIM: 3 命令マシン</b>	<b>178</b>
4.1	背景: TIM の仕組み	178
4.1.1	平坦化	179
4.1.2	タブル化	179
4.1.3	スパインレスであること	180

4.1.4	例	181
4.1.5	状態遷移規則によるマシンの定義	183
4.1.6	コンパイル	185
4.1.7	更新	186
4.2	Mark 1: 最小限の TIM	187
4.2.1	全体構造	187
4.2.2	データ型の定義	187
4.2.3	プログラムのコンパイル	190
4.2.4	評価器	192
4.2.5	結果の表示	194
4.2.6	ガベージコレクション†	198
4.3	Mark 2: 算術演算の追加	200
4.3.1	概要: 算術演算の仕組み	200
4.3.2	実装への単純な算術演算の追加	202
4.3.3	算術演算のコンパイルスキーム	205
4.4	Mark 3: let(rec) 式	208
4.4.1	let 式	208
4.4.2	letrec 式	211
4.4.3	フレームスロットの再利用†	213
4.4.4	ガベージ コレクション†	213
4.5	Mark 4: 更新	214
4.5.1	基礎技術	214
4.5.2	PushMarker 命令のコンパイル	216
4.5.3	更新メカニズムの実装	216
4.5.4	間接更新の問題	216
4.5.5	let(rec) にバインドされた共有変数の更新	216
4.5.6	間接チェーンの排除	216
4.5.7	部分的な関数適用の更新	216
4.6	Mark 5: 構造化データ	217
4.6.1	一般的なアプローチ	217
4.6.2	データ構造の遷移規則とコンパイル方式	217
4.6.3	試してみる	217
4.6.4	リストの表示	217
4.6.5	データ構造を直接使用する†	217
4.7	Mark 6: CAF とコードストア†	218
4.7.1	CAF の実装	218
4.7.2	コードストアをより忠実にモデリングする	218
4.8	まとめ	219

<b>5</b>	<b>並列 G マシン</b>	<b>221</b>
5.1	Introduction . . . . .	221
5.1.1	Parallel functional programming . . . . .	221
5.1.2	Parallel graph reduction . . . . .	221
5.2	Mark 1: A minimal parallel G-machine . . . . .	222
5.2.1	Data type definitions . . . . .	222
5.2.2	The evaluator . . . . .	222
5.2.3	Compiling a program . . . . .	222
5.2.4	Printing the results . . . . .	222
5.3	Mark 2: The evaluate-and-die model . . . . .	223
5.3.1	The node data structure . . . . .	223
5.3.2	The instruction set . . . . .	223
5.4	Mark 3: A realistic parallel G-machine . . . . .	224
5.4.1	Scheduling policy . . . . .	224
5.4.2	Conservative and speculative parallelism . . . . .	224
5.5	Mark 4: A better way to handle blocking . . . . .	225
5.5.1	Data structures . . . . .	225
5.5.2	Locking and unlocking . . . . .	225
5.6	Conclusions . . . . .	226
<b>6</b>	<b>ラムダリフティング</b>	<b>227</b>
6.1	Introduction . . . . .	227
6.2	Improving the <code>expr</code> data type . . . . .	228
6.3	Mark 1: A simple lambda lifter . . . . .	229
6.3.1	Implementing a simple lambda lifter . . . . .	229
6.3.2	Free variables . . . . .	229
6.3.3	Generating supercombinators . . . . .	229
6.3.4	Making all the variables unique . . . . .	229
6.3.5	Collecting supercombinators . . . . .	229
6.4	Mark 2: Improving the simple lambda lifter . . . . .	230
6.4.1	Simple extensions . . . . .	230
6.4.2	Eliminating redundant supercombinators . . . . .	230
6.4.3	Eliminating redundant local definitions . . . . .	230
6.5	Mark 3: Johnsson-style lambda lifter . . . . .	231
6.5.1	implementation . . . . .	231
6.5.2	Abstracting free variables in functions . . . . .	231
6.5.3	A tricky point $\dagger$ . . . . .	231
6.6	Mark 4: A separate full laziness pass . . . . .	232
6.6.1	A review of full laziness . . . . .	232
6.6.2	Full-lazy lambda lifting in the presence of <code>let(rec)s</code> . . . . .	232

6.6.3	Full laziness without lambda lifting . . . . .	232
6.6.4	A full lazy lambda lifter . . . . .	232
6.6.5	Separating the lambdas . . . . .	232
6.6.6	Adding level numbers . . . . .	232
6.6.7	Identifying MFEs . . . . .	232
6.6.8	Renaming variables . . . . .	232
6.6.9	Floating <code>let(rec)</code> expressions . . . . .	232
6.7	Mark 5: Improvements to full laziness . . . . .	233
6.7.1	Adding <code>case</code> expressions . . . . .	233
6.7.2	Eliminating redundant supercombinators . . . . .	233
6.7.3	Avoiding redundant full laziness . . . . .	233
6.8	Mark 6: Dependency analysis <sup>†</sup> . . . . .	234
6.8.1	Strongly connected components . . . . .	234
6.8.2	Implementing a strongly connected component algo- rithm . . . . .	234
6.8.3	A dependency analysis . . . . .	234
6.9	Conclusions . . . . .	235
<b>7</b>	<b>A ユーティリティモジュール</b>	<b>236</b>
7.1	The heap type . . . . .	236
7.1.1	Specification . . . . .	236
7.1.2	Representation . . . . .	236
7.2	The association list type . . . . .	237
7.3	Generating unique names . . . . .	237
7.3.1	Representation . . . . .	237
7.4	Sets . . . . .	238
7.4.1	Representation . . . . .	238
7.5	Other useful function definitions . . . . .	239
<b>8</b>	<b>B コア言語プログラムの例</b>	<b>240</b>
8.1	基本プログラム . . . . .	240
8.1.1	超基本テスト . . . . .	240
8.1.2	更新のテスト . . . . .	240
8.1.3	もっと興味深い例 . . . . .	240
8.2	<code>let</code> と <code>letrec</code> . . . . .	242
8.3	算術演算 . . . . .	243
8.3.1	条件分岐なし . . . . .	243
8.3.2	条件分岐あり . . . . .	243
8.4	データ構造 . . . . .	244

## 概要



## 序文

この本は、遅延グラフ簡約を使用して非正格な関数型言語の実装を理解するための実用的なアプローチを提供します。この本は、読者がいくつかの重要なコンパイラを開発、変更、および実験するのを支援することにより、関数型言語の実装を「生きた」ものにするのに役立つ実用的なラボワーク資料のソースとなることを目的としています。

この本の珍しい側面は、それが読むだけでなく実行されることを意図しているということです。各実装手法の抽象的な説明を提示するだけでなく、主要な各メソッドの完全に機能するプロトタイプのコードを提示し、一連の改善を行います。

## 本書の概要

この本の主な内容は、コア言語と呼ばれる小さな関数型言語の一連の実装です。コア言語は可能な限り小さく設計されているため、実装は簡単ですが、効率を損なうことなく最新の非厳密な関数型言語をコア言語に変換できるほど rich です。これについては、第 1 章で詳しく説明します。この章では、コア言語用のパーサとプリティプリンタも開発しています。

Appendix B には、本書全体でテストプログラムとして使用するためのコア言語プログラムのセクションが含まれています。

本書の本文は、コア言語の 4 つの異なる実装で構成されています。

- 第 2 章では、テンプレートのインスタンス化に基づいた最も直接的な実装について説明します。
- 第 3 章では、G マシンを紹介し、プログラムを一連の命令 (G コード) にコンパイルして、さらにマシンコードに変換する方法を示します。
- 第 4 章では、評価モデルが G マシンの評価モデルとは大きく異なる、異なる抽象マシンである Three Instruction Machine (TIM) についても同じ演習を繰り返します。TIM は G マシンよりも最近開発されたため、他の文献はほとんどありません。したがって、第 4 章には、G マシンに与えられたものよりもかなり詳細な TIM の評価モデルの開発が含まれています。
- 後に、第 5 章では、並列 G マシン用の関数型プログラムをコンパイルする方法を示すことにより、新しい次元を追加します。

これらの実装のそれぞれについて、コンパイラとマシンインタプリタの2つの主要部分について説明します。コンパイラはコア言語プログラムを受け取り、それをマシンインタプリタによる実行に適した形式に変換します。

マシンインタプリタは、コンパイルされたプログラムの実行をシミュレートします。いずれの場合も、インタプリタは状態遷移システムとしてモデル化されているため、マシンインタプリタと「実際の」実装の間には非常に明確な関係があります。図1は、実装の構造をまとめたものです。

コア言語が制限されている重要な方法の1つは、ローカル関数定義がないことです。ラムダリフティングと呼ばれるよく知られた変換があります。これは、ローカル関数定義をグローバル関数定義に変換し、ローカル関数定義を自由に記述して後で変換できるようにします。第6章では、適切なラムダリフターを開発します。この章は、標準的な資料の単なる再提示ではありません。完全な laziness は、以前はラムダリフティングと切り離せないと見なされていた関数型プログラムの特性です。第6章では、それらが実際にはまったく異なることを示し、ラムダリフティングとは別のパスで完全な laziness を実装する方法を示します。

この本全体を通して、Appendix A で定義されている多くのユーティリティ関数とデータ型を使用しています。

一部のセクションと演習はもう少し進んでおり、大きな損失なしに省略されています。それらは短剣で識別されます、したがって：†

## プロトタイピング言語

実装を作成するためにどの言語を使用するかという問題が発生します。既存の関数型言語である Miranda<sup>1</sup> を使用することを選択しました。関数型言語の主な用途の1つは、ラピッドプロトタイピングです。これにより、管理の詳細にとらわれることなく、プロトタイプ的基本的な側面を表現できるようになります。この本がこの主張を立証するための大きな例として役立つことを願っています。さらに、この本を通して作業することは、実質的な関数型プログラムを書くことの有用な経験を提供するはずです。

この本は関数型プログラミングの入門書ではありません。すでに関数型言語でプログラミングを行っていることを前提としています。(関数型プログラミングの適切な紹介には、[Bird and Wadler1988] および [Holyer1991] が含

---

<sup>1</sup>Miranda は、Research SoftwareLtd の商標です。

まれます。) それにもかかわらず、この本で開発されたプログラムは非常に充実しており、明確でモジュール式の関数型プログラムを作成する能力を伸ばし、開発するためのロールモデルとして役立つことを願っています。

Miranda コードは、「逆コメント規則」を使用してタイプライターファウン  
トで記述されています。たとえば、リストの長さを取る関数の定義は次の  
とおりです。

```
> length [] = 0
> length (x : xs) = 1 + length xs
```

左マージンの

>

マークは、これが実行可能な Miranda コードの行であることを示します。

## 本書がカバーしていないこと

この本では、関数型コンパイラの「バックエンド」にのみ焦点を当ててい  
ます。Miranda などの本格的な関数型言語で書かれたプログラムをコア言語  
に翻訳する方法や、そのようなプログラムの型チェック方法については議論  
しません。

全体の開発は非形式です。コアプログラムの意味とその実装が同等である  
ことを正式に証明するのは良いことですが、これは非常に難しい作業です。  
確かに、私たちが知っているそのような同等性の唯一の完全な証明は [Lester  
1988] です。

## 「The implementation of functional programming lan- guages」との関係

私たちの 1 人による以前の本 [Peyton Jones 1987] は、これと同様の資料  
をカバーしていますが、あまり実用的ではありません。私たちの意図は、学  
生が他の本を参照せずに、現在の本だけを使用して関数型言語の実装に関す  
るコースをたどることができるようにすることです。

この本の範囲はやや控えめで、[Peyton Jones 1987] のパート 2 とパート 3  
に対応しています。後者のパート 1 では、高レベルの関数型言語をコア言語  
に変換する方法について説明していますが、ここではまったく取り上げてい  
ません。

ソースコードの取得  
間違いを見つけたら

## 1 コア言語

すべての実装は、単純なコア言語で記述されたプログラムを使用して実行します。コア言語は非常に貧弱であり、大きなプログラムを書きたくないでしょう。それでも、表現力や効率を損なうことなく、豊富な関数型言語 (Miranda など) のプログラムをコア言語に翻訳できるように慎重に選択されています。したがって、コア言語は、高水準言語構造に関するコンパイラの「フロントエンド」と、さまざまな異なる方法でコア言語を実装することに関する「バックエンド」との間のクリーンなインターフェースとして機能します。

まず、コア言語の非公式な紹介から始めます (セクション 1.1)。これに続いて、次のようにしてコア言語をより正式に定義します。

- その構文 (セクション 1.2)。
- Miranda のデータ型 `coreProgram` と `coreExpr` は、それぞれコア言語のプログラムと式です (セクション 1.3)。これらは、ビルドするコンパイラの入力データ型として機能します。
- コア言語関数の小さな標準プレリユードの定義。これは、すべてのコアプログラムで利用できるようになります (セクション 1.4)。
- コア言語プログラムを文字列に変換するプリティプリンタ。印刷すると、プログラムのフォーマットされたバージョンになります (セクション 1.5)。
- 文字列を解析してコア言語プログラムを生成するパーサ (セクション 1.6)。

この章には 2 番めの目的があります。関数型言語の実装に関与する前に、本全体で使用する Miranda の機能の多くを紹介して使用します。

### 1.1 コア言語の概要

これは、42 と評価されるコアプログラム<sup>2</sup>の例です。

```
main = double 21
double x = x + x
```

コアプログラムは、主要なものを含む一連のスーパーコンビネータ定義で構成されています。プログラムを実行するために、`main` を評価します。スー

---

<sup>2</sup>コアプログラムにはタイプライターフォントを使用しますが、実行可能な Miranda コードを区別する最初の `>` 記号はありません。

パーコンビネータは、double の定義などの機能を定義できます。double は、1 つの引数  $x$  の関数であり、引数の 2 倍を返します。プログラムは、関数の引数にパターンマッチングが許可されていないことを除いて、Miranda スクリプトのトップレベルと非常によく似ています。パターンマッチングは、別のコア言語構造である case 式によって実行されます。これについては、以下で説明します。各スーパーコンビネータは、引数がすべて単純な変数である単一の等式によって定義されます。

すべてのスーパーコンビネータに引数があるわけではないことに注意してください。main など一部のスーパーコンビネータは引数を取りません。引数のないスーパーコンビネータは、定数適用形式または CAF と呼ばれ、後で説明するように、実装では特別な処理が必要になることがよくあります。

#### 1.1.1 ローカル定義

スーパーコンビネータは、コア言語の let 構造を使用して、ローカル定義を持つことができます。

```
main = quadruple 20 ;
quadruple x = let twice_x = x+x
              in twice_x + twice_x
```

ここで、twice\_x は quadruple の本体内でローカルに  $x+x$  と定義され、quadruple は  $twice\_x + twice\_x$  を返します。Miranda の where 句と同様に、ローカル定義は、中間値に名前を付けることと、同じ値を 2 回再計算することを節約することの両方に役立ちます。プログラマーは、quadruple で 2 つの加算だけが実行されることを合理的に期待できます。

let 式は非再帰的です。再帰的定義の場合、コア言語は letrec 構造を使用します。これは、定義が再帰的である可能性があることを除いて、let とまったく同じです。例えば:

```
infinite n = letrec ns = cons n ns
            in ns
```

(letrec だけを提供するのではなく) コア言語で let と letrec を区別する理由は、let は letrec よりも実装が少し簡単であり、コードが少し良くなる可能性があるためです。

let と letrec は Miranda の where 句に似ていますが、いくつかの重要な違いがあります。

- where 句は、常に再帰スコープを定義します。非再帰的な形式はありません。
- where 句を使用して、ローカル関数を定義し、パターンマッチングを実行できます。

```
... where f x = x+y
      (p,q) = zip xs ys
```

これらの機能はどちらも、コア言語の let および letrec 式では提供されません。

関数はスーパーコンビネータとしてトップレベルでのみ定義でき、パターンマッチングは case 式によってのみ実行されます。

つまり、let または letrec バインディングの左側は単純な変数でなければなりません。

- let/letrec 構造は式です。したがって、次のように書くことは非常に合法です (例えば):

```
quad_plus_one x = 1 + (let tx = x+x in tx+tx)
```

対称的に、Miranda の where 句は、定義にのみ付加できます。(この理由の 1 つは、Miranda の where 句の定義がいくつかのガードされた右辺に及ぶことを可能にすることです。)

### 1.1.2 ラムダ抽象化

関数は通常、トップレベルのスーパーコンビネータ定義を使用してコア言語で表現されます。ほとんどの本では、これが関数を表す表す表す唯一の方法です。ただし、明示的なラムダ抽象化を使用して関数を指定できると便利な場合があり、コア言語はそのための構造を提供します。たとえば、次のプログラムで

```
double_list xs = map (\ x. 2*x) xs
```

ラムダ抽象化 (`\ x. 2*x`) は、引数を 2 倍にする関数を示します。

明示的なラムダ抽象化を含むプログラムを、トップレベルのスーパーコンビネータ定義のみを使用する同等のプログラムに変換することができます。このプロセスはラムダリフティングと呼ばれ、第 6 章で詳しく説明します。

他の章では、このラムダリフティングプロセスが実行されたと想定しているため、明示的なラムダ抽象化を使用していません。

コア言語の最後の主要な構成要素は、パターンマッチングを表す `case` 式です。パターンマッチングを処理する方法はいくつかあるため、構造化データ型の確認から始めます。

### 1.1.3 構造化データ型

最新の関数型プログラミング言語すべての普遍的な機能は、代数的データ型と呼ばれることが多い構造化型の提供です。たとえば、Miranda で書かれたいくつかの代数的型の定義は次のとおりです。

```
colour ::= Red | Green | Blue

complex ::= Rect num num | Polar num num

numPair ::= MkNumPair num num

tree * ::= Leaf * | Branch (tree *) (tree *)
```

各定義は、1 つ以上のコンストラクタ (`Red`, `Green` など) とともに、新しい型 (`colour` など) を導入します。それらは次のように読むことができます: 「`colour` 型の値は `Red` または `Green` または `Blue` のいずれかです」、および 「`complex` は 2 つの `num` を含む `Rect`、または 2 つの `num` を含む `Polar` のいずれかです」。

型 `tree` は、パラメータ化された代数的データ型の例です。型 `tree` は、型変数 `*` に関してパラメータ化されます。次のように読む必要があります。「`*の tree`」`s` は、`*` を含む `Leaf`、または `*` の 2 つの `tree` 「`s`」を含む `Branch` のいずれかです。特定の `tree` には、同じ型の `Leaf` が必要です。たとえば、型 `tree num` は、`Leaf` に `num` がある `tree` であり、型 `tree colour` は、`Leaf` に `colour` がある `tree` です。

構造化された値は、これらのコンストラクタを使用して構築されます。たとえば、次の式は構造化された値を示します。

```
Green
Rect 3 4
Branch (Leaf num) (Leaf num)
```

構造化された値は、パターンマッチングを使用して分解されます。例えば:



```

isRed Red = True
isRed Green = False
isRed Blue = False

first (MkNumPair n1 n2) = n1

depth (Leaf n) = 0
depth (Branch t1 t2) = 1 + max (depth t1) (depth t2)

```

通常「組み込み」とみなされるいくつかのデータ型は、構造化型の特殊なケースにすぎません。例えば、ブール値は構造化型です。代数的データ型宣言によって定義できます。

```
bool ::= False | True
```

特別な構文とは別に、Miranda が提供するリストとタプルは構造化型のさらなる例です。:と [] の特別な構文ではなく、コンストラクタとして Cons と Nil を使用する場合、次のようなリストを定義できます。

```
list * ::= Nil | Cons * (list *)
```

[Peyton Jones 1987] の第 4 章では、構造化型について詳しく説明しています。従って、疑問が生じます。小さなコア言語で構造化型をどのように表現および操作するのでしょうか。特に、私たちの目標は、コア言語でのデータ型宣言を完全に回避することです。私たちが採用するアプローチは、2 つの部分に別れています。

- コンストラクタには、単純で統一された表現を使用します。
- パターンマッチングを単純な case 式に変換します。

#### 1.1.4 コンストラクタの表現

コア言語で Red や Branch などのユーザ定義のコンストラクタを許可する代わりに、コンストラクタの単一ファミリ

$$\text{Pack}\{tag, arity\}$$

を提供します。

ここで、tag はコンストラクタを一意に識別する整数であり、arity はそれを取る引数の数を示します。たとえば、colour、complex、tree、numPair のコンストラクタを次のように表すことができます。

```

Red      = Pack{1, 0}
Green    = Pack{2, 0}
Blue     = Pack{3, 0}

Rect     = Pack{4, 2}
Polar    = Pack{5, 2}

Leaf     = Pack{6, 1}
Branch   = Pack{7, 2}

MkNumPair = Pack{8, 2}

```

したがって、コア言語では、

```
Branch (Leaf 3) (Leaf 4)
```

の代わりに

```
Pack{7,2} (Pack{6,1} 3) (Pack{6,1} 4)
```

と記述します。この tag は、異なるコンストラクタで作成されたオブジェクトを互いに区別できるようにするためです。適切に型指定されたプログラムでは、実行時に異なる型のオブジェクトを区別する必要はないため、tag はデータ型内で一意である必要があります。したがって、新しいデータ型ごとに 1 から tag を新たに開始して、次の表現を与えることができます。

```

Red      = Pack{1, 0}
Green    = Pack{2, 0}
Blue     = Pack{3, 0}

Rect     = Pack{1, 2}
Polar    = Pack{2, 2}

Leaf     = Pack{1, 1}
Branch   = Pack{2, 2}

MkNumPair = Pack{1, 2}

```

#### 1.1.5 case 式

一般に、最新の関数型プログラミング言語で許可されているパターンマッチングは、複数のネストされたパターン、重複するパターン、ガードなど、かなり複雑になる可能性があります。コア言語の場合、パターンマッチング

のすべての複雑な形式を非合法化することにより、これらの複雑さを排除します。これを行うには、コア言語では `case` 式のみを提供します。それらの正式な構文はセクション 1.2 に記載されていますが、いくつかの例を示します。

```
isRed c = case c of
    <1> -> True ;
    <2> -> False ;
    <3> -> False

depth t = case t of
    <1> n -> 0 ;
    <2> t1 t2 -> 1 + max (depth t1) (depth t2)
```

`case` 式で重要なことは、各選択肢がタグとそれに続くいくつかの変数 (コンストラクタのアリティと同じである必要があります) のみで構成されていることです。ネストされたパターンは許可されていません。

`case` 式は、多方向ジャンプのように、非常に単純な操作上の解釈を持っています。分析する式を評価し、それが構築されているコンストラクタのタグを取得して、適切な選択肢を評価します。

## 1.2 コア言語の構文

図 1.1 に、コア言語の構文を示します。文法では、中置二項演算子を使用できますが、(簡潔にするために)それらの優先順位については明示されていません。代わりに、次の優先順位の表を示します。優先順位が高いほど、バインディングが厳しくなります。

優先度	結合の向き	演算子
6	左結合	関数適用
5	右結合	*
	なし	/
4	右結合	+
	なし	-
3	なし	==   =   >   >=   <   <=
2	右結合	&
1	右結合	

演算子の結合法則は、演算子の繰り返しの前後で括弧を省略できる場合を決定します。たとえば、+は右結合であるため、 $x+y+z$  は  $x+(y+z)$  と同じ意味です。一方、/は結合多元環ではないため、式  $x/y/z$  は不正です。

単項否定のための特別な演算子記号はありません。代わりに、通常の間数と構文的に同じように動作する `negate` 関数が提供されます。例えば：

`f x = x + (negate x)`

ブール否定演算子 `not` は、同じ方法で処理されます。

プログラム	$program \rightarrow sc_1; \dots; sc_n$	$n \geq 1$
スーパーコンビネータ	$sc \rightarrow var\ var_1 \dots var_n = expr$	$n \geq 0$
式	$expr \rightarrow expr\ expr$ $\quad   \quad expr_1\ binop\ expr_2$ $\quad   \quad let\ defns\ in\ expr$ $\quad   \quad letrec\ defns\ in\ expr$ $\quad   \quad case\ expr\ of\ alts$ $\quad   \quad \backslash\ var_1 \dots var_n .\ expr$ $\quad   \quad aexpr$	関数適用 中置 2 項演算 局所定義 局所再帰定義 case 式 ラムダ抽象 ( $n \geq 1$ ) 原子式
	$aexpr \rightarrow var$ $\quad   \quad num$ $\quad   \quad Pack\{num, num\}$ $\quad   \quad ( expr )$	変数 数値 コンストラクタ 括弧で囲まれた式
定義	$defns \rightarrow defn_1; \dots; defn_n$ $defn \rightarrow var = expr$	$n \geq 1$
選択肢	$alts \rightarrow alt_1; \dots; alt_n$ $alt \rightarrow \langle num \rangle\ var_1 \dots var_n \rightarrow expr$	$n \geq 1$ $n \geq 0$
二項演算子	$binop \rightarrow arithop \quad   \quad relop \quad   \quad boolop$ $arithop \rightarrow + \quad   \quad - \quad   \quad * \quad   \quad /$ $relop \rightarrow < \quad   \quad <= \quad   \quad == \quad   \quad \sim = \quad   \quad >= \quad   \quad >$ $boolop \rightarrow \& \quad   \quad  $	算術演算子 比較演算子 論理演算子
変数	$var \rightarrow alpha\ varch_1 \dots varch_n$ $alpha \rightarrow alphabeticcharacter$ $varch \rightarrow alpha \quad   \quad digit \quad   \quad -$	$n \geq 0$
数値	$num \rightarrow digit_1 \dots digit_n$	$n \geq 1$

### 1.3 コア言語のデータ型

この本で説明されている実装ごとに、コンパイラとマシンインタプリタを構築します。コンパイラはコアプログラムを受け取り、それをマシンインタプリタによる実行に適した形式に変換します。これを行うには、コアプログラムを表す Miranda データ型が必要です。これを、このセクションで定義します。実際、コアプログラム用の型、コア式用の型、およびその他のいくつかの補助的な型を定義します。

コア言語式 `expr` のデータ型は、次のように定義されます。

```
module Language where
import Utils

data Expr a
  = EVar Name                -- Variables
  | ENum Int                 -- Numbers
  | EConstr Int Int          -- Constructor tag arity
  | EAp (Expr a) (Expr a)    -- Applications
  | ELet                     -- Let(rec) expressions
      IsRec                 --   boolean with True = recursive,
      [(a, Expr a)]         --   Definitions
      (Expr a)              --   Body of let(rec)
  | ECase                    -- Case expression
      (Expr a)              --   Expression to scrutinise
      [Alter a]             --   Alternatives
  | ELam [a] (Expr a)        -- Lambda abstraction
  deriving (Text)
```

バインダに関して `expr` のデータ型をパラメータ化することを選択します。バインダは、変数のバインディングオカレンスで使用される名前です。つまり、`let(rec)` 定義の左側、またはラムダ抽象化です。コア言語式 `expr` の型は、宣言は、「\*の `expr` は、`name` 含む `EVar`、または `...`、またはタイプ\*の値のリストと\*の `expr` を含む `ELam` のいずれかです」と読み取ることができます。

本書のほとんどの部分では、これらのバインディング位置で常に `name` を使用しているため、型シノニムを使用して、通常使用するタイプである `coreExpr` のタイプを定義します。

```
type CoreExpr = Expr Name
```

## 1.4 A small standard prelude

## 1.5 コア言語用のプリティプリンタ

`CoreProgram` 型の値を取得したら、それを表示できると便利なのがよくあります。Miranda の組み込み機能は、ここではあまり役に立ちません。たとえば、Miranda プロンプトに回答して `preludeDefs` と入力すると、生成される出力を理解するのはかなり困難です。(それを試してみてください。)

必要なのは、次の型が付いた「プリティプリント」関数 `pprint` です。

```
pprint :: CoreProgram -> String
```

次に、`pprint preludeDefs` と入力して、印刷すると `preludeDefs` の適切にフォーマットされたバージョンのように見える文字のリストを取得することを期待できます。このセクションの目標は、そのような関数を作成することです。

プログラムの結果がリストの場合、Miranda は通常、リスト項目をコンマで区切り、角かっこで囲んで印刷します。ただし、プログラムの結果が `[char]` 型の特殊なケースでは、Miranda は角かっことコンマを付けずに「すべて押しつぶされた」リストを表示します。たとえば、値 `'Hi \nthere'` は次のように表示されます。

```
Hi
there
```

次のようには表示されません。

```
['H', 'i', '\n', 't', 'h', 'e', 'r', 'e']
```

このようにして、`pprint` は出力フォーマットを完全に制御できます。

### 1.5.1 文字列を使ったプリティプリンティング

まず、コア言語の式に集中しましょう。プリティプリンティング関数 `pprExpr` が必要のように見えます。次のように定義されています。

```
pprExpr :: CoreExpr -> String
pprExpr (ENum n) = show n
pprExpr (EVar v) = v
pprExpr (EAp e1 e2) = pprExpr e1 ++ " " ++ pprAExpr e2
```

(現時点では、`pprExpr` のケースの多くを意図的に除外しています。) `pprAExpr` の型は `pprExpr` と同じですが、変数または数値でない限り、式を括弧で囲む点が異なります。



```
pprAExpr :: CoreExpr -> String
pprAExpr e = isAtomicExpr e | pprExpr e
pprAExpr e = otherwise | "(" ++ pprExpr e ++ ")"
```

この方法で進めることはできますが、そうすることには深刻な問題があります。プリティプリンタは、リスト連結関数`++`を大量に使用します。次の例に示すように、これは非常に厄介なパフォーマンスをもたらす可能性があります。次の式を考えてみましょう。

$$(x1 \mathrel{++} x2) \mathrel{++} x3$$

内側の`++`は $\# \text{xs1}^3$ に比例する時間がかかりますが、外側の`++`は $\text{xs1} \mathrel{++} \text{xs2}$ の長さに比例する時間がかかるため、合計時間は $(2 * \# \text{xs1}) + \# \text{xs2}$ になります。一般に、このネストされた `append` にさらにリストを追加した場合、コストは結果の長さの2次式になる可能性があります！もちろん、式を逆に括弧でくくった場合、コストは結果の長さに比例しますが、残念ながら、きれいなプリンターでこれを保証することはできません。

この効果を実証するために、最初に関数 `mkMultiAp` を記述します。これにより、特定のサイズのサンプル式を簡単に作成できます。呼び出し (`mkMultiAp n e1 e2`) は、式を表す `coreExpr` を生成します

$$e_1 \underbrace{e_2 e_2 \dots e_2}_n$$

```
mkMultiAp :: Int -> CoreExpr -> CoreExpr -> CoreExpr
mkMultiAp n e1 e2 = foldl1 EAp e1 (take n e2s)
  where
    e2s = e2 : e2s
```

この定義では、`take` は、リストの最初の  $n$  個の要素を取り、リストの残りを破棄する Miranda の標準関数です。関数 `foldl1` は標準関数であり、Appendix A<sup>4</sup> で定義されています。二項関数  $\otimes$ 、値  $acc$  およびリスト  $xs = [x_1, \dots, x_n]$  が与えられると、`foldl1  $\otimes$  acc xs` は  $acc'$  を計算します。

$$acc' = (\dots((acc \otimes x_1) \otimes x_2) \otimes \dots x_n)$$

これを図 1.2 に示します。 `mkMultiAp` では、`foldl1` を使用して `EAp` ノードの左分岐チェーンを構築します。初期のアキュムレータ  $acc$  は  $e_1$  であり、結合関数は `EAp` コンストラクターです。最後に、`e2s` は無限のリスト  $[e_2, e_2, \dots]$  です。最初の  $n$  個の要素のみが `take` によって使用されます。

<sup>3</sup> # 関数は、リストの長さを取得するための標準の Miranda 関数です。

<sup>4</sup> Miranda のバージョンが異なれば `foldl` の定義も異なるため、Miranda の標準関数 `foldl` ではなく `foldl1` を使用します。



Figure 1.2: An illustration of  $\text{foldl1 } \otimes \text{ acc } [x_1, \dots, x_n]$

演習 1.1  $n$  のさまざまな値に対して、計算に必要なミランダのステップ数を測定します。

```
# (pprExpr (mkMultiAp n (EVar "f") (EVar "x")))
```

(Miranda ディレクティブ / `count` を使用して、Miranda に実行統計を出力するように指示できます。画面が巨大なプリントアウトでいっぱいにならないように、結果の長さをとります。) 実行コストが  $n$  でどのように上昇するかを示すグラフをスケッチし、 $n$  に関してほぼ 2 次であることを確認します。

### 1.5.2 プリティプリンティングのための抽象データ型

コストが印刷されるプログラムのサイズの 2 倍になるプリティ プリンタは、明らかに容認できないので、それを回避する方法を見つけたほうがよいでしょう。

この問題を 2 つの部分に分けることができます: 「実行したい操作は?」と「それらを実行する効率的な方法は?」です。他の言語と同様に、Miranda は抽象データ型を導入することでこの区別を明確にする方法を提供します。

```
iNil      :: Iseq                -- The empty iseq
iStr      :: String -> Iseq     -- Turn a string into an iseq
iAppend   :: Iseq -> Iseq -> Iseq -- Append two iseqs
iNewline  :: Iseq                -- New line with indentation
iIndent   :: Iseq -> Iseq       -- Indent an iseq
iDisplay  :: Iseq -> String     -- Turn an iseq into a string
```

abstype キーワードは、抽象データ型 `iseq` を導入します。その後に、データ型のインターフェイスが続きます。つまり、データ型 `iseq` で実行できる操作と、各操作の型です。

このようなデータ型が与えられた場合、`pprExpr` を書き直して、文字のリストではなく `iseq` を返すようにします。

```
pprExpr :: CoreExpr -> Iseq
pprExpr (EVar v)      = iStr v
pprExpr (EAp e1 e2) = (pprExpr e1) 'iAppend' (iStr " ") 'iAppend' (pprAExpr e2)
```

`++` を `iAppend`<sup>5</sup> に単純に置き換え、リテラル文字列の周りに `iStr` を追加しました。

`iseq` と文字のリストの違いは何ですか？ まず、リスト追加の予想外の二次動作を持たない `iAppend` の実装を作成することを目指しています。次に `iseq` は、インデントの制御に役立つ新しい操作 `iIndent` および `iNewline` を提供します。アイデアは、`iIndent` がその引数をインデントして現在の列に揃えるというものです。引数が何行にもわたっており、それ自体に `iIndent` の呼び出しが含まれている場合でも機能するはずです。`iNewline` は、改行の後に、現在のインデント レベルによって決定されるいくつかのスペースが続くことを表します。

`iIndent` と `iNewline` の使用方法の例として、`pprExpr` を拡張して `let` と `letrec` 式を処理してみましょう。

```
pprExpr (ELet isrec defns expr)
  = iConcat [ iStr keyword, iNewline,
              iStr " ", iIndent (pprDefns defns), iNewline,
              iStr "in ", pprExpr expr ]
  where
    keyword | not isrec = "let"
            | isrec     = "letrec"

pprDefns :: [(Name, CoreExpr)] -> Iseq
pprDefns defns = iInterleave sep (map pprDefn defns)
  where
    sep = iConcat [ iStr ";", iNewline ]
```

---

<sup>5</sup>Miranda では、識別子の前にドル記号を書くと、それが中置演算子に変わり、引数の前ではなく引数の間に `iAppend` を書くことができます。このような中置演算子は右結合です。

```
pprDefn :: (Name, CoreExpr) -> Iseq
pprDefn (name, expr) = iConcat [ iStr name, iStr " = ", iIndent (pprExpr expr) ]
```

定義をより読みやすくするために、`iConcat` と `iInterleave` の 2 つの新しい関数を使用しました。

```
iConcat :: [Iseq] -> Iseq
iInterleave :: Iseq -> [Iseq] -> Iseq
```

`iConcat` は `iseq` のリストを受け取り、`iAppend` を使用してそれらを単一の `iseq` に連結します。`iInterleave` は、隣接する各ペア間に指定された `iseq` をインターリーブする点を除いて、`iConcat` に似ています。

演習 1.2 `iAppend` と `iNil` に関して `iConcat` と `iInterleave` を定義します。

一般に、すべてのプリティ プリンティング関数は `iseq` を返し、表示したいものの全体を表す `iseq` に `iDisplay` をトップ レベルで 1 回だけ適用します。

演習 1.3 `pprExpr` に式を追加して `case` 式とラムダ式を処理し、`pprAExpr` と `pprProgram` の定義を同じスタイルで記述します。

### 1.5.3 `iseq` の実装

`iseq` 型の実装に移ります。すべてのインデントを無視する実装を作成することから始めます。抽象データ型を実装するには、型が何であるかを言う必要があります `iseq` を表すために使用されます。

```
data Iseq = INil
          | IStr String
          | IAppend Iseq Iseq
```

最初の宣言は `iseqRep` が `iseq` を表すために使用されることを示し、2 番目の宣言は `iseqRep` が 3 つのコンストラクター `INil`、`IStr`、および `IAppend` を持つ代数データ型であることを宣言します。

この特定の表現の一般的な考え方は、`iDisplay` の最終的なすべてまでのすべての作業を延期することです。操作 `iNil`、`iStr`、`iAppend` はすべて、関連するコンストラクターを使用するだけです。

```
iNil = INil
iStr str = IStr str
iAppend seq1 seq2 = IAppend seq1 seq2
```

インデントを無視しているので、`iIndent` と `iNewline` は自明に定義されています。次のセクションでそれらを改善します。

```
iNewline = IStr "\n"
iIndent seq = seq
```

`iseq` を文字のリストに変換する操作 `iDisplay` にすべての関心があります。目標は、`iseq` のサイズに比例した時間だけかかるようにすることです。より一般的な機能である `flatten` に関して `iDisplay` を定義すると便利であることが判明しました。

```
flatten :: [Iseq] -> String

iDisplay seq = flatten [seq]
```

関数 `flatten` は `iseqReps` のリストを受け取り、リスト内の各 `iseqReps` を連結した結果を返します。このリストを持つ理由は、すぐにわかるように、保留中の作業のリストを蓄積できるようにするためです。`flatten` は、抽象型 `iseq` ではなく、表現型 `iseqRep` を操作することに注意してください。

ワークリストと呼ばれる引数のケース分析によって `flatten` を定義します。ワークリストが空の場合、完了です。

```
flatten [] = ""
```

それ以外の場合は、ワークリストの最初の要素でケース分析を行うことで作業します。`INil` の場合は、ワークリストからアイテムをポップするだけです。

```
flatten (INil : seqs) = flatten seqs
```

`IStr` の場合は、指定された文字列を追加して、残りのワークリストをフラット化することで機能します。

```
flatten (IStr s : seqs) = s ++ (flatten seqs)
```

これまでのところ、`flatten` がリストを取得するという事実は、私たちにあまり役に立ちませんでした。`IAppend` を扱うと、リスト引数の正当性がより明確にわかります。実行する必要があるのは、もう1つのアイテムをワークリストの先頭にプッシュすることだけです。

```
flatten (IAppend seq1 seq2 : seqs) = flatten (seq1 : seq2 : seqs)
```

演習 1.4 `iseq` のサイズに関して平坦化のコストはいくらですか？ 上記のように `iseq` を使用するように `pprExpr` を変更し、前の演習と同じ実験を使用して新しい実装の効果を測定します。 `pprExpr` の結果に `iDisplay` を適用することを忘れないでください。

演習 1.5 抽象データ型を使用する主な利点は、インターフェイスに影響を与えずに ADT の実装を変更できることです。この例として、引数のいずれかが `INil` の場合に単純化された結果を返すように `iAppend` を再定義します。

#### 1.5.4 レイアウトとインデント

これまでのところ、`iIndent` 操作に対してかなり些細な解釈しか与えていません。次に、それを改善します。以前と同じ精神で、最初に `iseqRep` 型を追加の 2 つのコンストラクター `IIndent` と `INewline` で拡張し、これらのコンストラクターを使用するように操作を再定義します。

```
data Iseq = INil
  | IStr String
  | IAppend Iseq Iseq
  | IIndent Iseq
  | INewline
```

```
iIndent seq = IIndent seq
iNewline    = INewline
```

次に、`flatten` をより強力にする必要があります。まず、現在のカラムを追跡する必要があります。次に、そのワークリストは `(iseq, num)` ペアで構成されている必要があります。ここで、数値は対応する `iseq` に必要なインデントを示します。

```
flatten :: Int          -- Current column; 0 for first column
        -> [(Iseq, Int)] -- Work list
        -> String       -- Result
```

`flatten` を適切に初期化するには、`iDisplay` を変更する必要があります。

```
iDisplay seq = flatten 0 [(seq, 0)]
```

`flatten` の興味深いケースは、`INewline` を扱う場合です。これは、インデントを実行する必要があるためです。

```
flatten col ((INewline, indent) : seqs)
  = '\n' : (space indent) ++ (flatten indent seqs)
```

新しい行に移動して indent スペースを追加したので、flatten の再帰呼び出しには現在の列引数 indent があることに注意してください。これは、新しい行に移動してインデント スペースを追加したためです。

IIndent ケースは、現在の列から現在のインデントを設定するだけです。

```
flatten col ((IIndent seq, indent) : seqs) = flatten col ((seq, col) : seqs)
```

演習 1.6 IAppend、IStr、および INil の flatten の定義を追加します。

Elet を含む式で pprExpr を試し、レイアウトが適切に機能することを確認します。

演習 1.7 IStr に与えられた文字列に改行文字 '\n' が埋め込まれていると、プリティプリンタは正しく動作しません。これをチェックするように iStr を変更し、改行文字を INewline を使用して置き換えます。

#### 1.5.5 演算子の優先順位

セクション 1.3 で説明したように、coreExpr 型には中置演算子適用のための構造がありません。代わりに、そのような関数適用は、他の関数適用と同様に、前置形式で表現されます。私たちの プリティプリンタ がそのような関数適用を認識し、それらを中置形式で印刷できれば素晴らしいと思います。これは、次の形式の pprExpr に式を追加することで簡単に実行できます。

```
pprExpr (EAp (EAp (EVar "+") e1) e2)
  = iConcat [ pprAExpr e1, iStr " + ", pprAExpr e2 ]
```

挿入する括弧が多すぎるため、これでもあまりうまくいきません。次の式

```
x + y > p * length xs
```

を見たいですか？または完全に括弧で囲まれたバージョン

```
(x + y) > (p * (length xs))
```

の方がいいですか？

これを実現する最も簡単な方法は、pprExpr にそのコンテキストの優先レベルを示す追加の引数を与え、これを使用して、生成される式の周りに括弧を追加するかどうかを決定することです。(関数 pprAExpr は冗長になります。)

演習 1.8 これらの変更を pprExpr に加えてテストします。

### 1.5.6 iseq のその他の便利な機能

後で iseq で動作する関数がいくつかあると便利です。これらはすべて iseq インターフェイス関数の観点から定義されているため、これらの定義を変更せずに実装を変更できます。

iNum は数値を iseq にマップし、iFNum は同じことを行いますが、結果は指定された幅までスペースで左がパディングされます。

```
iNum :: Int -> Iseq
iNum n = iStr (show n)

iFNum :: Int -> Int -> Iseq
iFNum width n
  = iStr (space (width - length digits) ++ digits)
  where
    digits = show n
```

(数値が必要な幅より広い場合、負の数値がスペースに渡され、空のリストが返されます。したがって、最終的な効果は、含めるのに十分な幅のフィールドを返すことです数字。) iLayn は、標準関数 layn と同じように、項目に番号を付け、それぞれの後に改行文字を付けてリストをレイアウトします。

```
iLayn :: [Iseq] -> Iseq
iLayn seqs
  = iConcat (map lay_item (zip [1..] seqs))
  where
    lay_item (n, seq)
      = iConcat [ iFNum 4 n,
                  iStr ")",
                  iIndent seq,
                  iNewline
                ]
```

### 1.5.7 概要

私たちのプリティプリンタにはまだ欠点があります。特に、優れたプリティプリンタは、収まる場合は 1 行にレイアウトし、収まらない場合は複数の行にレイアウトします。次のように iseq データ型を精巧にすることは十分に可能です。これを行いますが、ここでは行いません。iseq 型は、プログラム以外のデータをきれいに印刷するのに役立ちます。この本では、さまざまな



目的で使います。このセクションから導き出したい 2 つの一般的なポイントがあります。

- 抽象データ型のインターフェースをその実装から分離することは、非常にしばしば役に立ちます。Miranda は、抽象型に対する関数が表現を検査しないようにすることで、この抽象化を直接サポートします。
- `flatten` に関する `iDisplay` の定義は、汎化と呼ばれる非常に一般的な手法の例です。より一般的な関数への単純な呼び出しという観点から、本当に必要な関数を定義することがよくあります。これは通常、より一般的な関数が、簿記をまっすぐに保つために必要ないくつかの追加の引数を持ち運ぶためです。

一般化が適切な手法である場合について、一般的な意見を述べるのは困難です。実際、適切な一般化を行うことは、多くの場合、プログラムを作成する際の主要な創造的ステップです。ただし、この本には一般化の例がたくさんあり、アイデアを伝えるのに役立つことを願っています。

## 1.6 コア言語のパーサ

それぞれの実装をさまざまなコアプログラムで実行する必要があります。これは、具体的な構文で Core プログラムを含むファイルを取得し、それを CoreProgram 型の値に解析する方法が必要であることを意味します。

パーサを作成するのは一般的にかなり面倒なので、文法を受け入れてパーサを作成するツールの作成に多大な労力が費やされてきました。Unix Yacc ユーティリティは、そのようなパーサジェネレータの例です。ただし、関数型言語では、単純なパーサを作成するのは非常に簡単です。このセクションでは、コア言語について説明します。タスクを 3 つの段階に分割します。

- まず、名前付きファイルの内容を文字のリストとして取得します。これは、組み込みの Miranda 関数 read によって実行されます。
- 次に、字句解析関数 lex は、入力を識別子、数字、記号などの小さなチャンクのシーケンスに分割します。これらの小さなチャンクはトークンと呼ばれます。

```
Clex :: String -> [Token]
```

- 最後に、構文解析関数の構文は、この一連のトークンを使用して CoreProgram を生成します。

```
syntax :: [Token] -> CoreProgram
```

完全なパーサは、これら 3 つの関数の合成にすぎません。

```
parse :: String -> CoreProgram
parse = syntax . clex
-- In Gofer I propose to compose this with some function
-- CoreProgram -> String, which will illustrate some sort of
-- execution machine, and then give this composition to catWith
-- from my utils
```

記号 '.' は、Miranda の中置関数合成演算子であり、次のように定義できます。

$$(f \cdot g) x = f (g x)$$

同様に、次のように、合成を使用せずに解析関数を定義することもできます。

```
parse filename = syntax (lex (read filename))
```

ただし、構文解析を3つの関数のパイプラインとして定義していることが特に簡単にわかるため、関数合成を使用する方が適切なスタイルです。

### 1.6.1 字句解析

字句解析器から始めます。トークンの種類はまだ定義されていません。最初に最も簡単なことは、トークンをまったく処理せず、(空でない) 文字列として残すことです。

```
type Token = String -- A token is never empty
```

今、字句解析自体。空白(空白、タブ、改行)を破棄する必要があります：

```
clex (c : cs) | isWhiteSpace c = clex cs
```

数字を単一のトークンとして認識する必要があります。

```
clex (c : cs) | isDigit c = num_token : clex rest_cs
  where
    num_token = c : takeWhile isDigit cs
    rest_cs   = dropWhile isDigit cs
```

標準関数 `isDigit` は文字を受け取り、その文字が10進数の場合にのみ、`True` を返します。`takeWhile` と `dropWhile` も標準機能です。`takeWhile` は、述語が満たされている間、リストの先頭から要素を取得し、`dropWhile` は、述語が満たされている間、リストの先頭から要素を削除します。例えば、

```
takeWhile isDigit "123abc456"
```

の結果はリスト `"123"` です。

字句解析プログラムは、アルファベット文字で始まり、文字、数字、アンダースコアのシーケンスで続く変数も認識する必要があります。

```
clex (c : cs) | isAlpha c = var_tok : clex rest_cs
  where
    var_tok = c : takeWhile isIdChar cs
    rest_cs = dropWhile isIdChar cs
```

ここで、`letter` はアルファベット文字で `True` を返す `digit` のような標準関数であり、`isIdChar` は以下で定義されています。

上記の式のいずれにも当てはまらない場合、字句解析プログラムは 1 文字を含むトークンを返します。

```
cllex (c : cs) = [c] : cllex cs
```

最後に、入力文字列が空の場合、`lex` は空のトークンリストを返します。

```
cllex [] = []
```

上記で使用した補助関数の定義で締めくくります。(演算子 `'` は Miranda のブール `'or'` 演算です。)

```
isIdChar, isWhiteSpace :: Char -> Bool
isIdChar c = isAlpha c || isDigit c || (c == '_' )
isWhiteSpace c = c `elem` " \t\n"
```

**演習 1.9** コメントと空白を無視するように字句アナライザを変更します。コメントが二重の縦棒 `||` によって導入されるのと同じ規則を使用し、行の終わりまで延長します。

**演習 1.10** 字句解析プログラムは現在、`<=` や `==` などの 2 文字の演算子を単一のトークンとして認識しません。それらのリストを提供することにより、そのような演算子を定義します。

```
twoCharOps :: [String]
twoCharOps = ["==", "~=", ">=", "<=", "->"]
```

`twoCharOps` の `member` をトークンとして認識するように `lex` を変更します。(標準関数メンバーが役立つ場合があります。)

**演習 1.11** 字句解析では空白が破棄されるため、パーサは構文エラーの行番号を報告できません。この問題を解決する 1 つの方法は、各トークンに行番号を付けることです。つまり、タイプ `token` は次のようになります。

```
token == (num, [char])
```

これを行うように字句アナライザを変更します。これを行うには、現在の行番号であるパラメータを `lex` に追加する必要があります。

### 1.6.2 構文解析のための基本的なツール

コア言語のパーサを作成する準備として、パーサを作成するときに使用するいくつかの汎用関数を開発します。以下で説明する手法はよく知られていますが [Fairbairn1986、Wadler 1985]、関数型プログラミングで何ができるかをかなりうまく示しています。実行例として、次の小さな文法を使用します。

```
greeting  →  hg person !
hg         →  hello
           |  goodbye
```

ここで、*person* は文字で始まるトークンです。

関数型プログラミングで非常に一般的な私たちの一般的なアプローチは、小さなパーサを接着して大きなパーサを構築しようとすることです。重要な質問は、パーサのタイプはどうあるべきかということです。これは、トークンのリストを引数として受け取る関数であり、最初は、解析された値を返すだけでよいように見えます。しかし、これは2つの理由から、十分に一般的ではありません。

1. まず、トークンの残りのリストも返す必要があります。たとえば、入力から2つの項目を次々に解析する場合は、最初のパーサを入力に適用できますが、最初のパーサから返された残りの入力に2番目のパーサを適用する必要があります。
2. 次に、文法があいまいな場合があるため、入力を解析する方法は複数あります。または、入力が文法に準拠していない可能性があります。その場合、入力を正常に解析する方法はありません。これらの可能性に対応するためのエレガントな方法は、可能な解析のリストを返すことです。このリストは、入力を解析する方法がない場合は空であり、入力を解析する独自の方法がある場合は1つの要素を含みます。

次のように、型シノニムを使用してパーサの型を定義することにより、結論を要約できます。

```
type Parser a = [Token] -> [(a, [Token])]
```

つまり、型\*の値のパーサは、トークンのリストを受け取り、解析のリストを返します。各解析は、型\*の値と残りのトークンのリストを組み合わせたものです。

### 1.6.3 ツールを鋭利にする

これで、パーサを開発するための基本的なツールが完成しました。このセクションでは、さまざまな方法でそれらを開発します。

文法では人の名前の後に感嘆符が必要なため、上記の `pGreeting` の定義は完全には正しくありません。次のように問題を解決できます。

```
pGreeting = pThen keep_first
              (pThen mk_pair pHelloOrGoodbye pVar)
              (pLit "!")
    where
      keep_first hg_name exclamation = hg_name
      mk_pair hg name = (hg, name)
```

最終的な感嘆符は常に存在するため、解析された値の一部として返さないことを選択しました。keep\_first によって破棄されます。ただし、この定義はかなり不器用です。新しい関数 `pThen3` を定義すると、次のように記述できるので便利です。

```
pGreeting = pThen3 mk_greeting
              pHelloOrGoodbye
              pVar
              (pLit "!")
    where
      mk_greeting hg name exclamation = (hg, name)
```

演習 1.12 `pThen3` の型を指定し、その定義を書き留めて、新しいバージョンの `pGreeting` をテストします。同様に、後で必要になる `pThen4` を記述します。

文法のもう 1 つの非常に一般的な機能は、記号の 0 回以上の繰り返しを要求することです。これを反映するために、パーサ `p` を受け取り、`p` が認識する 0 回以上の出現を認識する新しいパーサを返す関数 `pZeroOrMore` が必要です。成功した解析によって返される値は、`p` の継続的な使用によって返される値のリストである可能性があります。従って、`pZeroOrMore` の型は次のようになります。

```
pZeroOrMore :: Parser a -> Parser [a]
```

例えば、0 個以上の挨拶を認識するパーサは次のようになります。

```
pGreetings :: Parser [(String, String)]
pGreetings = pZeroOrMore pGreeting
```

1 つ以上の出現、または出現なしのいずれかを確認する必要があることを確認することで、`pZeroOrMore` になることができます。

```
pZeroOrMore p = (pOneOrMore p) 'pAlt' (pEmpty [])
```

ここで、`pEmpty` は常に成功するパーサであり、入力から何も削除せず、最初の引数として指定された値を返します。

```
pEmpty :: a -> Parser a
```

関数 `pOneOrMore` は、`pZeroOrMore` と同じ型です。

```
pOneOrMore :: Parser a -> Parser [a]
```

演習 1.13 `pOneOrMore` と `pEmpty` の定義を記述します。(ヒント:`pOneOrMore` から `pZeroOrMore` を呼び出すと便利です。) 1 つまたは複数の挨拶を認識するパーサを定義するためにそれらを使用して、定義をテストします。

成功した解析によって返された値を処理すると便利ながよくあります。たとえば、`pGreetings` が内容ではなく、挨拶の数を返すようにしたいと思います。これを行うには、長さ関数 `#` を `pZeroOrMore` によって返される値に適用します。

```
pGreetingsN :: Parser Int
pGreetingsN = (pZeroOrMore pGreeting) 'pApply' length
```

ここで、`pApply` は新しいパーサ操作関数であり、パーサと関数を受け取り、パーサによって返される値に関数を適用します。

```
pApply :: Parser a -> (a -> b) -> Parser b
```

演習 1.14 `pApply` の定義を記述し、それをテストします。(ヒント: リスト内包表記を使用してください。)

文法のもう 1 つの非常に一般的なパターンは、他の記号で区切られた 1 つ以上の記号の出現を探すことです。たとえば、図 1.1 のプログラムは、セミコロンで区切られた 1 つ以上のスーパーコンビネータ定義のシーケンスです。さらに別のパーサ構築関数 `pOneOrMoreWithSep` が必要です。この関数のタイプは、次の通りです。

```
pOneOrMoreWithSep :: Parser a -> Parser b -> Parser [a]
```

2 番目の引数は、結果の一部として返されないセパレータを認識するパーサです。そのため、型に\*\*が 1 回だけ出現します。

演習 1.15 `pOneOrMoreWithSep` を定義してテストします。*program* の次の文法を考えると役立つ場合があります。

$$\begin{array}{ll} \text{program} & \rightarrow \text{sc programRest} \\ \text{programRest} & \rightarrow ; \text{program} \\ & | \epsilon \end{array}$$

ここで、 $\epsilon$  は空の文字列 (pEmpty パーサに対応) です。

パーサ `pLit` と `pVar` は互いに非常に似ています。どちらも最初のトークンのプロパティをテストし、失敗するか (プロパティがない場合)、成功してトークン内の文字列を返します (トークンがある場合)。このアイデアを一般化するには、パーサ `pSat` ('sat' は 'satisfies' の略) を次のように記述します。

```
pSat :: (String -> Bool) -> Parser String
```

`pSat` は、トークン内の文字列に目的のプロパティがあるかどうかを通知する関数を受け取り、そのプロパティを持つトークンを認識するパーサを返します。これで、`pSat`<sup>6</sup>の観点から `pLit` を記述できます。

```
pLit s = pSat (== s)
```

演習 1.16 `pSat` を定義し、テストします。`pLit` と同様の方法で、`pSat` の観点から `pVar` を記述します。

`pSat` は、有用なレベルのモジュール性を追加します。たとえば、`pVar` は現在、すべてのアルファベットトークンを変数として認識しますが、最終的には、言語キーワード (`let` や `case` など) を変数として認識しないようにする必要があります。

演習 1.17 上記の `pVar` の定義で `pSat` に渡される関数を変更して、リストキーワードの文字列を変数として扱わないようにします。

```
keywords :: [String]
keywords = ["let", "letrec", "case", "in", "of", "Pack"]
```

演習 1.18 別の例として、`pSat` を使用して、以下に示す型の数値のパーサを定義します。

---

<sup>6</sup>式 (`= s`) はセクションと呼ばれます。これは、等式演算子`=`を 1 つの引数 `s` に部分適用し、その引数が `s` に等しいかどうかをテストする関数を生成します。



```
pNum :: ParserInt
```

pNum は pSat を使用して数値トークンを識別し、次に pApply を使用して文字列を数値に変換する必要があります。(Miranda は、ハードワークを実行するために使用できる [Char] -> num 型の標準関数 numval を提供します。)

pOneOrMore とそれに関連する機能に関連する興味深いパフォーマンスの問題があります。次のコアプログラムについて考えてみます。

```
f x = let x1 = x; x2 = x; ...; xn = x
      of x1
```

アイデアは、 $x_1, x_2, \dots, x_n$  の定義を持つ大きな let 式があるということです。(定義はかなり些細なものですが、目的を果たします。) このプログラムには構文エラーがあります。let 式の後に「in」ではなく「of」を記述しました。

#### 1.6.4 コア言語の構文解析

コア言語のパーサを定義する準備が整いました。まず、「ラッパー」関数、構文を扱います。トークンのリストを取得し、coreProgram 型の結果を提供することを思い出してください。これを行うには、非終端プログラムを解析するパーサ pProgram(図 1.1) を呼び出し、それが返す最初の完全な解析を選択します。完全な解析が返されない場合、つまり、解析後に残っているトークンのシーケンスが空である場合、構文は(ひどく情報が少ない)エラーメッセージを生成します。

```
syntax = take_first_parse . pProgram
      where
        take_first_parse ((prog, []) : others) = prog
        take_first_parse (parse      : others) = take_first_parse others
        take_first_parse other              = error "Syntax error"
```

私たちの構文解析ツールの利点は、文法を Miranda に直訳するだけでパーサを記述できることです。たとえば、図 1.1 の *program* と *sc* のプロダクションについて考えてみます。

$$\begin{array}{ll} \text{program} & \rightarrow \text{sc}_1; \dots; \text{sc}_n & (n \geq 1) \\ \text{sc} & \rightarrow \text{var } \text{var}_1 \dots \text{var}_n = \text{expr} & (n \geq 0) \end{array}$$

これらを直接 Miranda に直訳することができます。

```
pProgram :: Parser CoreProgram
pProgram = pOneOrMoreWithSep pSc (pLit ";")
```

```
pSc :: Parser CoreScDefn
pSc = pThen4 mk_sc pVar (pZeroOrMore pVar) (pLit "=") pExpr
```

演習 1.20 関数 `mk_sc` を記述します。これは、`pSc` で使用される 4 つのパースによって返される 4 つの引数を取り、次のタイプの値を作成します。

```
(name, [name], coreExpr)
```

関数適用と中置演算子の生成を除いて、残りの文法の定義を完了するのは簡単なことです。

演習 1.21 これらの 2 つのプロダクションを除外して、パーサを完成させます。型が `Parser CoreExpr` であるパーサ `pAexpr` には少し注意が必要です。`pApply` 関数は、`pVar` によって返される値の周りに `EVar` コンストラクタをラップし、`pNum` によって返される値の周りに `ENum` コンストラクタをラップするために必要です。次のプログラムでパーサをテストします。

```
f = 3 ;
g x y = let z = x in z ;
h x = case (let y = x in y) of
    <1> -> 2 ;
    <2> -> 5
```

より大きなプログラムでパーサを実行すると、出力が判読でなくなることがわかります。これを解決するには、プリティプリンティング関数 `pprint` を使用して、パーサの出力をフォーマットします。

演習 1.22 次のプログラムを検討します。

```
f x y = case x of
    <1> -> case y of
        <1> -> 1;
        <2> -> 2
```

<2>で始まる選択枝は、内側の `case` または外側の `case` のどちらに取り付けられますか？答えを見つけて、パーサが期待どおりに動作するかどうかを確認します。これは「ぶら下がり `else`」の問題として知られています。

ここで、上記の 2 つの問題に目を向けます。

### 1.6.5 左再帰

関数適用の問題は比較的簡単に解決できます。関数適用のためのプロダクションは次のようになります。

$$expr \rightarrow expr \ aexpr$$

これを単に直訳すると

```
pExpr = pThen EAp pExpr pAexpr
```

残念ながら、pExpr はそれ自体を無期限に呼び出し続けるため、終了することはありません。問題は、*expr* が *expr* の生成の最初のシンボルとして表示されることです。これは左再帰と呼ばれます。私たちの構文解析ツールは、単に左再帰文法に対処できません。幸いなことに、通常、文法を変換して左再帰にならないようにすることができますが、結果の文法は、構築しようとしている結果の構造を反映していません。この場合、たとえば、繰り返しを使用して、問題のあるプロダクションを次のように変換できます。

$$expr \rightarrow aexpr_1 \dots aexpr_n \ (n \geq 1)$$

これで、パーサ (pOneOrMore pAexpr) を使用できるようになりました。問題は、これが EAp コンストラクターで作成された単一の式ではなく、式のリストを返すことです。pApply を使用してこれを解決し、パーサを提供します

```
(pOneOrMore pAexpr) $ pApply mk_ap_chain
```

演習 1.23 型が [CoreExpr] -> CoreExpr の適切な関数 mk\_ap\_chain を定義します。関数適用のプロダクションをパーサに追加してテストします。

### 1.6.6 中置演算子の追加

中置演算子の最初の問題は、それらの優先順位が図 1.1 の文法に暗黙的に含まれていることです。これを明示的にする標準的な方法は、図 1.3 に示すように、いくつかの種類の式を使用することです。

$$\begin{aligned}
expr &\rightarrow \text{let } defns \text{ in } expr \\
&| \text{letrec } defns \text{ in } expr \\
&| \text{case } expr \text{ of } alts \\
&| \backslash var_1 \dots var_n . expr \\
&| expr1 \\
expr1 &\rightarrow expr2 \mid expr1 \\
&| expr2 \\
expr2 &\rightarrow expr3 \& expr2 \\
&| expr3 \\
expr3 &\rightarrow expr4 \text{ relop } expr4 \\
&| expr4 \\
expr4 &\rightarrow expr5 + expr4 \\
&| expr5 - expr5 \\
&| expr5 \\
expr5 &\rightarrow expr6 * expr5 \\
&| expr6 / expr6 \\
&| expr6 \\
expr6 &\rightarrow aexpr_1 \dots aexpr_n \quad (n \geq 1)
\end{aligned}$$

この文法が次の事実を表現する方法に注意してください。|および&は右結合ですが、関係演算子は非結合です。非常に多くのルールを書き出さなければならぬのはかなり面倒ですが、私たちはずっと意味していることを明確にしているだけです。しかし、ここで2番目の問題が発生します。これらのルールから直接実装されたパーサはひどく非効率的です。*expr1*のプロダクションを考えてみましょう。ナイーブなパーサは*expr2*を認識しようとし、次に垂直バー|を探します。検出されなかった場合(よくあることですが)、元の入力を手間をかけて再解析して、*expr2*を再度検索します。さらに悪いことに、*expr2*を解析する各試行には、*expr3*を解析する2回の試行が含まれる可能性があり、したがって、*expr4*を解析する4回の試行が含まれる可能性があります。

*expr2*の解析を2つのプロダクション間で共有したいのですが、*expr1*プロダクションを2つに分割することで、これを行うのは難しくありません。

$$\begin{aligned}
expr1 &\rightarrow expr2 \text{ } expr1c \\
expr1c &\rightarrow \mid expr1 \\
&| \epsilon
\end{aligned}$$

ここで、 $\epsilon$ は空の文字列を表します。*expr1*の生成では、*expr1*は垂直バー|の後に*expr1*が続くか、空であると示されています。私たちは、ほぼ、そこにいる!最後の質問は、*expr1*のパーサの型は何ですか。|というフレーズがあ

るため、`parser coreExpr` 型にすることはできません。*expr1* は式の一部にすぎず、空の文字列  $\epsilon$  も式ではありません。いつものように、文法を変換すると構造が破壊されます。

解決策はかなり簡単です。このような新しいデータ型 `partialExpr` を定義します

```
data PartialExpr = NoOp
                  | FoundOp Name CoreExpr
```

これで、次のように *expr1c* のパーサを定義できます。

```
pExpr1c :: Parser PartialExpr
pExpr1c = (pThen FoundOp (pLit "|") pExpr1) 'pAlt' (pEmpty NoOp)
```

*expr1* のパーサは、`pExpr1` によって返される中間結果を分解します：

```
pExpr1 :: Parser CoreExpr
pExpr1 = pThen assembleOp pExpr2 pExpr1c
```

```
assembleOp :: CoreExpr -> PartialExpr -> CoreExpr
assembleOp e1 NoOp = e1
assembleOp e1 (FoundOp op e2) = EAp (EAp (EVar op) e1) e2
```

演習 1.24 提案された行に沿って文法を変換し、変更を Miranda コードに直訳して、結果のパーサをテストします。

### 1.6.7 まとめ

パーサ構築関数のライブラリによって効率的に処理できる文法は、LL(1) 文法と呼ばれます。これは、従来の再帰降下パーサによって処理できるものとまったく同じクラスです [Aho et al. 1986]。

ライブラリを使用すると、非常に簡潔なパーサを簡単に作成できます。ほとんどすべてのプログラムには何らかの入力言語があり、プログラムによって解析される必要があるため、これは重要で便利なプロパティです。

気をつけなければならないことはたくさんありますが (左再帰、演算子の優先順位、共有)、実装されている言語に関係なく、再帰降下パーサではまったく同じ問題が発生します。

```
module Template where
import Language
import Utils
```

## 2 テンプレートのインスタンス化

この章では、関数型言語の可能な限りの単純な実装、つまりテンプレートのインスタンス化に基づくグラフ簡約器を紹介します。

初期バージョン (マーク 1) の完全なソースコードが示され、その後に基本設計の一連の改善とバリエーションが続きます。まず、グラフの簡約とテンプレートのインスタンス化のレビューから始めます。

### 2.1 テンプレートインスタンス化のレビュー

テンプレートのインスタンス化かの概要から始めます。この資料は、[Peyton Jones 1987] の第 11 章と第 12 章で詳しく説明されています。

次の重要な事実を思い出します。

- 関数型プログラムは、式を評価することによって「実行」されます。
- 式はグラフで表されます。
- 評価は、一連の簡約を実行することによって行われます。
- 簡約は、グラフ内の簡約可能な式をその簡約形式に置き換えます (または更新します)。「簡約可能な式」という用語は、しばしば「redex」と省略されます。
- redex がなくなると、評価は完了です。式は正規形であると言います。
- 評価される式にはいつでも複数の redex が含まれる可能性があるため、次にどれを簡約するかを選択できます。幸い、どの簡約手順を選択しても、常に同じ答え (つまり、正規形) が得られます。注意点が 1 つあります。一部の簡約手順は終了しない場合があります。
- ただし、どの redex を選択しても評価が終了する場合、常に最も外側の redex を選択するというポリシーも終了します。この簡約順序の選択は、最外最左簡約 (normal order reduction) と呼ばれ、常に使用されるものです。

従って、評価のプロセスは次のように説明できます。

```
until there are more redexes
  select the outermost redex
  reduce it
  update the (root of the) redex with the result
end
```

### 2.1.1 実例

例として、次のコア言語プログラムについて考えてみます。

```
square x = x * x ;  
main = square (square 3)
```

このプログラムは、スーパーコンビネータと呼ばれる一連の定義で構成されています。square と main はどちらもスーパーコンビネータです。慣例により、評価される四季はスーパーコンビネータ main です。従って、最初に評価される式は、次のかなり些細なツリーで表されます (ツリーは単なる特別な種類のグラフであることを忘れないでください)。

main

ここで、main には引数がないため、main 自体が redex となり、main をその本体で置き換えます。

```
main          reduces to      @  
                               / \  
                             square  @  
                               / \  
                             square  3
```

これらの図と後に続くすべての図では、関数適用は@記号で表されています。

ここで、最も外側の redex は外側の関数適用 (square の適用) です。関数適用を減らすために、redex を関数本体のインスタンスに置き換え、仮引数が出現するたびに引数へのポインタを置き換えます。

```
      @!          reduces to      @!  
      / \  
    square  @  
      / \  
    square  3  
  
      @!          reduces to      @!  
      / \  
    @  \  
    / \_ _ _ @  
    *   / \  
    square  3
```


結果で上書きされる redex のルートは、!でマークされます。内側の square 3 の redex が共有され、ツリーがグラフになっていることに注意してください。

square の定義では、式  $x*x$  (\*は中置演算子として書かれています) は  $((x) x)$  の略で、2つの引数に\*を適用します。カーリー化を使用して、1引数の関数適用の形で複数の引数を持つ関数を記述します。\*は、引数  $p$  に適用され




たとき、関数を返す関数となります。 $(* p)$  が返す関数は、別の引数  $q$  に適用されると  $p$  と  $q$  の積を返すような関数です。

現在、唯一の redex は内側の関数適用 (square の 3 への適用) です。 $*$  の引数を評価する必要があるため、 $*$  の適用は簡約できません。内側の関数適用は次のように簡約されます。



まだ 1 つの redex、内側の乗算があります。redex を乗算結果の 9 に置き換えます。



redex のルートの結果で物理的に更新することにより、外側の乗算の両方の引数が内側の乗算の結果を「参照」することに注意してください。最終的な簡約は簡単です。



### 2.1.2 3 つのステップ

前に見たように、グラフ簡約は、正規形に達するまで次の 3 つのステップを繰り返すことで構成されます。

1. 次の redex を見つける。
2. それを簡約する。
3. 簡約結果で redex(のルート) を更新する。

前のセクションの例からわかるように、2 種類の redex があり、さまざまな方法で簡約されます。

**スーパーコンビネータ** 最も外側の関数適用がスーパーコンビネータの適用である場合、それは確かに redex でもあり、以下に説明するように簡約することができます (セクション 2.1.4)。

**組み込みのプリミティブ** 最も外側の関数適用が組み込みプリミティブの適用である場合、引数が評価されるかどうかに応じて、適用は redex である場合とそうでない場合があります。redex でない場合は、引数を評価する必要があります。これは、まったく同じプロセスを使用して行われます。引数の最も外側の redex を繰り返し見つけて、それを簡約します。これが完了すると、外側の組み込みプリミティブ適用の簡約に戻ることができます。

### 2.1.3 次の redex を見つけるためのスパインの巻き戻し

簡約サイクルの最初のステップは、実行する次の簡約の部位を見つけることです。つまり、最も外側の簡約可能な関数適用です。次のように、最も外側の関数適用を見つけるのは簡単です (ただし、簡約できない場合があります)。

1. スーパーコンビネータまたは組み込みプリミティブに到達するまで、ルートから開始して、関数適用ノードの左側の枝をたどります。この関数適用ノードの左分岐チェーンはエクスプレッションのスパインと呼ばれ、このプロセスはスパインの巻き戻しと呼ばれます。通常、スタックは、途中で遭遇したノードのアドレスを記憶するために使用されます。
2. ここで、スーパーコンビネータまたはプリミティブが取る引数の数を確認し、その数の関数適用ノードに戻ります。これで、最も外側の関数適用のルートが見つかりました。

たとえば、式  $(f\ E1\ E2\ E3)$  では、 $f$  は 2 つの引数を取ります。たとえば、最も外側の関数適用は  $(f\ E1\ E2)$  です。式とスタックは次のようになります。

Stack  
-----

```

      | ---|-----> @
      -----
      / \
      | ---|-----> @! E3
      -----
      / \
      | ---|----> @ E2
      -----
      / \
      | ---|-> f E1
      -----

```

最も外側の関数適用 (のルート) は!でマークされています。

f が 2 つではなく 4 つの引数をとった場合のように、評価の結果が部分適用である可能性がある場合は、上記の手順 2 の前に、スパインに十分な関数適用ノードがあることを確認する必要があります。そうでない場合、式は Weak Head Normal Form (WHNF) に達しています。サブ式 E1、E2、および E3 にはまだ redex が含まれている可能性があります。ほとんどの評価器は、サブ式も簡約しようとするのではなく、WHNF に到達すると停止します。プログラムが型チェックされており、結果が数値、たとえば、リストであることが保証されている場合、このアンダーフローチェックは省略できます。

最も外側の関数適用のルートのみが見つかったことに注意してください。それは同様に redex であるかもしれないしそうでないかもしれません。関数がスーパーコンビネータである場合、それは確かに redex になりますが、+などのプリミティブである場合、引数が評価されるかどうかによって異なります。もしスーパーコンビネータなら、私たちは最も外側の redex を見つけました。スーパーコンビネータでない場合は、まだやるべきことがあります。

プリミティブが現在評価されていない引数の値を必要とする場合、プリミティブの簡約を進める前に引数を評価する必要があります。これを行うには、以前と同じように、現在のスタックを片側に配置し、引数を減らすために新しいスタックから開始する必要があります。これは、前のセクションの例でステージに到達したときの状況でした。

```

      @
      / \
      @ \
      / \___@!
      *   / \
square 3

```

新しいスタックで引数 (square 3) を評価する必要があります。この評価

中に、未評価の引数を持つプリミティブに再び遭遇する可能性があるため、新しい評価を再度開始する必要があります。正しい順序でスタックに戻ることができるように、すべての「古い」スタックを追跡する必要があります。これは、ダンプと呼ばれるスタックのスタックを保持することによって便利に行われます。引数を評価する必要がある場合は、現在のスタックをダンプにプッシュします。評価が終了したら、古いスタックをダンプから取り出します。

もちろん、実際の実装では、スタック全体をコピーすることはありません。「新しい」スタックは「古い」スタックが再び必要になる前に終了するため、「新しい」スタックは「古い」スタックの上に物理的に構築できます。ダンプスタックは、「新しい」と「古い」の境界がどこにあったかを追跡するだけです。ただし、概念的には、ダンプはスタックのスタックであり、この方法でモデル化します。

#### 2.1.4 スーパーコンビネータ redex

スーパーコンビネータ redex は、引数を本体に置き換えることで簡約されます。より正確には：

スーパーコンビネータの簡約。スーパーコンビネータ redex は、redex をスーパーコンビネータ本体のインスタンスに置き換え、実引数へのポインタで対応する仮引数の発生箇所を置き換えることによって簡約されます。引数はコピーされないことに注意してください。むしろ、それらへのポインタを使用するデバイスによって、それらは共有されます。

スーパーコンビネータ本体には、let 式と letrec 式を含めることができます。例えば：

```
f x = let y = x*x
      in y+y
```

let および letrec 式は、グラフのテキスト記述として扱われます。ここでは、たとえば、f の定義の可能な使用方法があります。

$\begin{array}{c} @ \\ / \quad \backslash \\ f \quad 3 \end{array}$	reduces to	$\begin{array}{c} @ \\ / \quad \backslash \\ @ \quad \backslash \\ / \quad \backslash \quad \backslash \\ + \quad \backslash \_\_\_ @y \\ \quad \quad / \quad \backslash \end{array}$
---	------------	---

```

      @   \
     / \   \
    *   \___3

```

let 式は、y という名前の部分式 x\*x を定義します。let 式の本体 y+y は、y の代わりに部分式へのポインタを使用します。したがって、通常の表現は木を表します。式に循環グラフを記述させます。および letrec 式は循環グラフを記述します。

### 2.1.5 更新

簡約を実行した後、結果で redex のルートを更新する必要があります。これにより、redex が共有されている場合 (例 (sqrt (sqrt 3))) のように、簡約は 1 回だけ実行されます。この更新は、遅延評価の本質です。redex はまったく評価されない場合がありますが、評価された場合、更新により、そのコストが最大で 1 回発生することが保証されます。

更新を省略してもエラーは発生しません。これは、一部の式が複数回評価される可能性があることを意味するだけであり、非効率的です。

更新を実行するときに少し注意が必要なケースが 1 つあります。以下のプログラムを検討します。

```

id x = x
f p = (id p) * p
main = f (sqrt 4)

```

f の簡約が行われた後、グラフは次のようになります。

```


      @
     / \
    @   \
   / \   \
  +   \___@y
        / \
        @   \
       / \   \
      *   \___3

```

sqrt は、平方根を取るための組み込みプリミティブであると想定しています。ここで、次に選択された redex が\*の最初の引数、つまり id の関数適用であると仮定します。(どちらの引数も通常の形式ではないため、\*の 2 番目

の引数でも同様に適切である可能性があります、最初の引数であると想定します。) id 簡約を実行した後、redex のルートを何で上書きする必要がありますか? (sqrt 4) が 2 回評価されるため、(sqrt 4) 関数適用ノードのコピーで上書きしないでください。

このジレンマから抜け出す最も簡単な方法は、新しい種類のグラフノードである間接ノードを追加することです。これは#記号で示されます。間接ノードを使用して、簡約の結果を指すように redex のルートを更新できます。



[Peyton Jones 1987] のセクション 12.4 には、更新に関連する問題の詳細な説明が含まれています。

## 2.1.6 定数適用形式 †

一部のスーパーコンビネータには引数がありません。それらは、定数適用形式または CAF と呼ばれます。たとえば、fac20 は CAF です。

```
fac20 = factorial 20
```

CAF の興味深い点は、スーパーコンビネータ自体が redex であるということです。fac20 が呼び出されるたびに、factorial20 の新しいコピーをインスタンス化する必要はありません。これは、factorial20 の計算を繰り返すことを意味するためです。むしろ、スーパーコンビネータ fac20 は fac20 簡約のルートであり、その本体をインスタンス化した結果で上書きする必要があります。

実際の結果は、スーパーコンビネータを通常の方法で更新できるようにするために、グラフノードで表す必要があるということです。これは、各実装で実際に発生することがわかります。

これでグラフ簡約のレビューは終わりです。

## 2.2 状態遷移システム

ここで、グラフ簡約の実装に注意を向けます。状態遷移システムを使用して、各実装について説明します。このセクションでは、状態遷移システムを紹介します。

状態遷移システムは、シーケンシャルマシンの動作を説明するための表記法です。いつでも、マシンは指定された初期状態から始まるある状態にあります。マシンの状態が状態遷移ルールの1つと一致する場合、ルールはマシンの新しい状態を指定します。一致する状態遷移ルールがない場合、実行は停止します。複数のルールが一致する場合は、1つが任意に選択されて発火します。その場合、マシンは非決定論的です。すべてのマシンは決定論的です。

これは、(かなり非効率的な)乗算器を指定するために使用される状態遷移システムの簡単な例です。状態は4つ組  $(n, m, d, t)$  です。掛ける数は  $n$  と  $m$ 、現在の合計は  $t$  で、マシンは状態  $(n, m, 0, 0)$  に初期化されます。

マシンの動作は、2つの遷移ルールによって指定されます。最初のルールで指定されているように、 $d$  は0に向かって繰り返しデクリメントされると同時に、 $t$  をインクリメントします。

$$\begin{array}{cccc} n & m & d & t \\ \Rightarrow & n & m & d - 1 & t + 1 \\ & \text{where } d > 0 \end{array}$$

古い状態の同じ値の真下に新しい状態の各値を使用して遷移ルールを常に作成するため、どの値が変更されたかを簡単に確認できます。

$d$  が0に達すると、再び  $n$  に初期化され、 $m$  が0に達するまで  $m$  がデクリメントされます。これは、2番目のルールで指定されています。

$$\begin{array}{cccc} n & m & 0 & t \\ \Rightarrow & n & m - 1 & n & t + 1 \\ & \text{where } m > 0 \end{array}$$

ルールが適用されない場合、マシンは終了します。この時点で、状態  $(n, 0, 0, t)$  になります。ここで、 $t$  は初期状態からの  $n$  と  $m$  の積です。

**演習 2.1** 初期状態  $(2; 3; 0; 0)$  から開始して、各ステップでどのルールを実行するかを指定して、マシンを手動で実行しなさい。最終状態が  $(2; 0; 0; 6)$  であることを確認しなさい。

演習 2.2 一連の状態の不変量は、すべての状態に当てはまる述語です。n と m の初期値 (N と M と呼びます) と m、d、t の現在の値との関係を表す不変量を見つけなさい。次に、マシンが乗算を実行するという推測を証明しなさい。証明を行うには、以下を示す必要があります。

1. 不変量は初期状態では真になる。
2. 不変量が特定の状態で真になれば、次の状態でも真になる。
3. 不変量と終了条件 ( $m = d = 0$ ) が与えられると、 $t = N * M$  になる。
4. マシンは終了する。

状態遷移システムは、次の理由で私たちの目的に便利です。

- それらは十分に抽象的であるため、非常に低レベルの詳細に巻き込まれることがない。
- それらは十分に具体的であるため、ルールに多くの複雑さを隠すことで「不正行為」をしていないことを確認できる。
- 状態遷移システムを直接 Miranda に書き直して、システムの実行可能な実装を提供できる。

最後のポイントを説明するために、乗算器を Miranda に書き直します。このマシンの状態の型を定義するために、型シノニムを与えることから始めます。

```
type MultState = (Int, Int, Int, Int) -- (n, m, d, t)
```

次に、関数 evalMult は状態を取得し、その状態とそれに続くすべての状態で構成されるリストを返します。

```
evalMult :: MultState -> [MultState]
evalMult state = if multFinal state
                  then [state]
                  else state : evalMult (stepMult state)
```

関数 stepMult は非最終状態を取り、次の状態を返します。遷移ルールごとに stepMult の式が 1 つあります。

```
stepMult (n, m, d, t) | d > 0 = (n, m, d - 1, t + 1)
                      | d == 0 = (n, m - 1, n, t)
```

関数 multFinal は状態を取得し、その状態が最終状態であるかどうかをテストします。



```
multFinal :: MultState -> Bool
```

演習 2.3 関数 `multFinal` を定義し、結果のマシンを初期状態  $(2, 3, 0, 0)$  で実行し、結果リストの最後の状態が  $(2, 0, 0, 6)$  であることを確認します。標準関数 `layn` は、結果をより読みやすくレイアウトするのに役立つ場合があります。

```
multFinal state = m == 0 && d == 0
                  where (_, m, d, _) = state
```

```
*Main> evalMult (2, 3, 0, 0)
[(2,3,0,0),(2,2,2,0),(2,2,1,1),(2,2,0,2),
 (2,1,2,2),(2,1,1,3),(2,1,0,4),
 (2,0,2,4),(2,0,1,5),(2,0,0,6)]
```

## 2.3 Mark 1: 最小限のテンプレートインスタンス化グラフ簡約器

これで、かなり単純なグラフ簡約器の定義を開始する準備が整いました。シンプルですが、より洗練されたグラフ簡約器が持つ多くの部分が含まれているため、説明するのに数ページかかります。

### 2.3.1 グラフ簡約の遷移規則

テンプレートインスタンス化グラフ簡約マシンの状態は、4 つ組

(スタック、ダンプ、ヒープ、グローバル)

または略して (s, d, h, f) です。

- スタックはアドレスのスタックであり、各アドレスはヒープ内のノードを識別します。これらのノードは、評価される式のスパインを形成します。表記  $a_1 : s$  は、最上位の要素が  $a_1$  で、残りが  $s$  であるスタックを示します。
- ダンプは、厳密なプリミティブの引数を評価する前に、スパインスタックの状態を記録します。ダンプは Mark1 マシンではまったく使用されませんが、後続のバージョンでは役立ちます。
- ヒープは、タグ付けされたノードのコレクションです。表記  $h[a : node]$  は、ヒープ  $h$  内のアドレス  $a$  がノード  $node$  を指すことを意味します。
- 各スーパーコンビネータ (および後で各プリミティブ) について、グローバルはスーパーコンビネータ (またはプリミティブ) を表すヒープノードのアドレスを提供します。

ヒープノードは、次の 3 つの形式のいずれかを取ることができます (最も原始的なマシンの場合)。

- $\text{NApp } a_1 \ a_2$  は、アドレスが  $a_1$  であるノードからアドレスが  $a_2$  であるノードへの関数適用を表します。
- $\text{NSupercomb } args \ body$  は、引数  $args$  と本体  $body$  を持つスーパーコンビネータを表します。
- $\text{NNum } n$  は、数  $n$  を表します。

このプリミティブテンプレートインスタンス化マシンには、2つの状態遷移ルールしかありません。1つ目は、単一のアプリケーションノードをスパインスタックに巻き戻す方法を説明しています。

(2.1)	$\frac{a : s \quad d \quad h[a : \text{NAp } a_1 \ a_2] \quad f}{\Rightarrow \quad a_1 : a : s \quad d \quad h \quad f}$
-------	--

(このルールの2行目のヒープコンポーネントには、アドレス  $a$  から  $\text{NAp } a_1 \ a_2$  へのマッピングが含まれていますが、混乱を避けるために、再度書き出すことはありません。) このルールを繰り返し適用すると、スタックの最上位のノードが  $\text{NAp}$  ノードでなくなるまで、式のスパイン全体がスタックに巻き戻されます。

2番目のルールは、スーパーコンビネータの簡約を実行する方法を説明しています。

(2.2)	$\frac{a_0 : a_1 : \dots : a_n : s \quad d \quad h[a_0 : \text{NSupercomb } [x_1, \dots, x_n] \text{ body}] \quad f}{\Rightarrow \quad a_r : s \quad d \quad h' \quad f}$ <p style="margin: 0; padding: 0;">where <math>(h', a_r) = \text{instantiate body } h \ f[x_1 \mapsto a_1, \dots, x_n \mapsto a_n]</math></p>
-------	--

このルールへの関心のほとんどは、関数のインスタンス化の中に隠されています。その引数は次のとおりです。

- インスタンス化する式
- ヒープ
- スタックから取得したアドレスへの引数名のマッピングによって拡張された、ヒープアドレスへの名前のグローバルマッピング  $f$

新しいヒープと、新しく構築されたインスタンス(のルート)のアドレスを返します。このような強力な操作は、各ステップが単純なアトミックアクションであることが意図されている状態遷移システムの精神とは実際には異なりますが、それがテンプレートインスタンス化マシンの性質です。後の章の実装はすべて、真にアトミックなアクションを持ちます！

`redex` のルート自体はこのルールの影響を受けないことに注意してください。スタック上で結果のルートに置き換えられるだけです。言い換えると、これらのルールは、グラフ簡約マシンではなく、`redex` のルートを更新しないツリー簡約マシンを記述しています。これについては、セクション 2.5 で後ほど改善します。

### 2.3.2 実装の構造

マシンの仕様がわかったので、その実装に着手する準備が整いました。

関数型言語で実装を記述しているので、たとえば、ジョブを実行するために関数 `run` を記述しなければなりません。そのタイプはどうあるべきですか？それは、ファイル名を取り、その中でプログラムを実行し、結果を出力する必要があります。これは、最終結果またはある種の実行トレースのいずれかである可能性があります。したがって、`run` のタイプは、次の型シグネチャによって指定されます。

```
runProg :: [Char] -> [Char] -- name changed to not conflict
```

これで、`run` がどのように構築されるかを考えることができます。プログラムの実行は、次の 4 つの段階で構成されます。

1. 指定されたファイルにある式からプログラムを解析します。 `parse` 関数はファイル名を受け取り、解析されたプログラムを返します。

```
parse :: [Char] -> CoreProgram
```

2. プログラムを実行に適した形式に変換します。このタスクを実行する `compile` 関数は、プログラムを受け取り、テンプレートインスタンス化マシンの初期状態を生成します。

```
compile :: CoreProgram -> TiState
```

`tiState` は、テンプレートインスタンス化マシンの状態の型です。(接頭辞「`ti`」はテンプレートのインスタンス化の略です。)

3. 最終状態に達するまで状態遷移を繰り返し実行して、プログラムを実行します。結果は、通過したすべての状態のリストです。これから、その後、最終状態を抽出するか、すべての状態のトレースを取得できます。今のところ、結果として数値を返すプログラムに限定するので、この実行関数を `eval` と呼びます。

```
eval :: TiState -> [TiState]
```

4. 印刷用に結果をフォーマットします。これは、印刷する情報を選択し、それを文字のリストにフォーマットする関数 `showResults` によって実行されます。

```
showResults :: [TiState] -> [Char]
```

関数 `run` は、次の 4 つの関数を組み合わせたものです。

```
runProg = showResults . eval . compile . parse -- ‘run’: name conflict
```

これらの各フェーズにサブセクションを割り当てます。

### 2.3.3 パーサ

パーサ関数を含むソース言語は、第 1 章で定義されている別のモジュール `language` で定義されています。モジュールをインポートするために `%include` ディレクティブを使用して利用できるようにします。

```
-- import Language
```

### 2.3.4 コンパイラ

このセクションでは、`compile` 関数を定義します。`utils` モジュールで定義されたデータ型と関数が必要になるため、`%include` を使用して使用できるようにします。

```
-- import Utils
```

次に、コンパイラが操作するデータ型の表現を検討する必要があります。

#### データ型

コンパイラは、型 `tiState` を持つマシンの初期状態を生成するため、次に行うことは、型シノニムを使用して、マシンの状態がどのように表されるかを定義することです。

```
type TiState = (TiStack, TiDump, TiHeap, TiGlobals, TiState)
```

マシンの状態は、最初の 4 つのコンポーネントがセクション 2.3.1 で指定されたコンポーネントに正確に対応し、5 番目のコンポーネントが統計の累積に使用される 5 つです。

次に、これらの各コンポーネントの表現を検討する必要があります。

- スパインスタックは、ヒープアドレスの単なるスタックです。

```
type TiStack = [Addr]
```

スタックをリストとして表すことを選択します。スタックの要素は、`utils` モジュール (付録 A.1) で定義された抽象データ型 `addr` のメンバーです。これらはヒープアドレスを表し、抽象化することで、`utils` モジュールによって提供される操作のみを使用できるようにします。したがって、たとえば誤ってアドレスに 1 つ追加することは不可能です。

- ダンプはセクション 2.6 まで必要ありませんが、後で追加すると状態遷移ルールに多くの面倒な変更が必要になるため、すでに状態の一部にしています。今のところ、引数のない単一のコンストラクタで構成される簡単な型の定義を与えます。

```
data TiDump = DummyTiDump
initialTiDump = DummyTiDump
```

- ヒープは、`utils` モジュールで定義された `heap` 抽象データ型によって表されます。ヒープに含まれるもの、つまり `node` 型のオブジェクト (まだ定義されていません) を指定する必要があります。

```
type TiHeap = Heap Node
```

ヒープ `node` は、次の代数的データ型宣言で表されます。これは、セクション 2.3.1 に記載されている可能性のリストに対応しています。

```
data Node = NApp Addr Addr -- Application
          | NSupercomp Name [Name] CoreExpr -- Supercombinator
          | NNum Int -- A number
```

唯一の違いは、スーパーコンビネータの名前を保持するために使用される `NSupercomb` コンストラクタにタイプ `name` のフィールドを追加したことです。これは、文書化とデバッグの目的でのみ使用されます。

- グローバルコンポーネントは、各スーパーコンビネータ名を、その定義を含むヒープノードのアドレスに関連付けます。

```
type TiGlobals = ASSOC Name Addr
```

`assoc` タイプは、その操作とともに `utils` モジュールで定義されます (付録 A.2)。リストの組み込み構文を使用して関連付けを操作できると非常に便利なため、実際には (抽象データ型ではなく) 型シノニムとして定義されています。ここでは、抽象化とプログラミングの容易さの間に緊張関係があります。

- 状態の `tiStats` コンポーネントは遷移ルールには記載されていませんが、これを使用して、マシンの動作に関する実行時のパフォーマンス統計を収集します。収集する統計を簡単に変更できるように、抽象型にします。まず、実行したステップ数のみを記録します。

```
tiStatInitial  :: TiStats
tiStatIncSteps :: TiStats -> TiStats
tiStatGetSteps :: TiStats -> Int
```

実装はかなり単純です。

```
type TiStats = Int
tiStatInitial  = 0
tiStatIncSteps s = s + 1
tiStatGetSteps s = s
```

便利な関数 `applyToStats` は、特定の関数を状態の統計コンポーネントに適用します。

```
applyToStats :: (TiState -> TiState) -> TiState -> TiState
applyToStats stats_fun (stack, dump, heap, sc_defs, stats)
  = (stack, dump, heap, sc_defs, stats_fun stats)
\end{itemize}
```

これで、関連するデータ型の定義が完了しました。

## コンパイラ自体

コンパイラの仕事は、プログラムを取得し、そこからマシンの初期状態を作成することです。

```
compile program
  = (initial_stack, initialTiDump, initial_heap, globals, tiStatInitial)
  where
    sc_defs = program ++ preludeDefs ++ extraPreludeDefs

    (initial_heap, globals) = buildInitialHeap sc_defs

    initial_stack = [address_of_main]
    address_of_main = aLookup globals "main" (error "main is not defined")
```

where 節の各定義を順番に考えてみましょう。最初の `sc.defs` は、プログラムに含まれるすべてのスーパーコンビネータ定義のリストにすぎません。`preludeDefs` はセクション 1.4 で定義されており、すべてのプログラムに常に含まれている標準のスーパーコンビネータ定義のリストであることを思い出してください。`extraPreludeDefs` は、追加したい標準関数のリストです。今のところそれは空です：

```
extraPreludeDefs = []
```

2 番目の定義は、補助関数 `buildInitialHeap` を使用して、各スーパーコンビネータの名前をそのノードのアドレスにマップする関連付けリスト `globals` とともに、各スーパーコンビネータの `NSupercomb` ノードを含む初期ヒープを構築します。

最後に、`initial_stack` は、`globals` から取得されたスーパーコンビネータ `main` のノードのアドレスである 1 つの項目のみを含むように定義されています。

次に、`buildInitialHeap` の定義を検討する必要があります。これは少し注意が必要です。リスト `sc.defs` の要素ごとに何かを行う必要がありますが、厄介なのは、「何か」にヒープの割り当てが含まれることです。各ヒープ割り当ては新しいヒープを生成するため、`sc.defs` の 1 つの要素から次の要素にヒープを渡す方法を見つける必要があります。このプロセスは、空のヒープ `hInitial`(付録 A.1) から始まります。

このアイデアを高階関数 `mapAccum1` にカプセル化します。これは、この本で非常に多く使用します。`mapAccum1` は 3 つの引数を取ります：「処理関数」 $f$ 、「アキュムレータ」 $acc$ 、およびリスト  $[x_1, \dots, x_n]$  です。入力リストの各要素を受け取り、それと現在のアキュムレータに  $f$  を適用します。 $f$  は、結果のペア、結果リストの要素、およびアキュムレータの新しい値を返します。`mapAccum1` は、 $f$  の 1 つの呼び出しから次の呼び出しにアキュムレータを渡し、最終的に 2 つの結果を返します：「アキュムレータの最終値」 $acc'$  と、「結果リスト」 $[y_1, \dots, y_n]$  です。図 2.1 は、この配管を示しています。`mapAccum1` の定義は、付録 A.5 に記載されています。

この場合、「アキュムレータ」はヒープであり、初期値は `hInitial` です。リスト  $[x_1, \dots, x_n]$  はスーパーコンビネータ定義、`sc.defs` であり、結果リスト  $[y_1, \dots, y_n]$  は、スーパーコンビネータの名前とアドレス、`sc_addrs` の関連付けです。ここに、`buildInitialHeap` の定義があります。

```
buildInitialHeap :: [CoreScDefn] -> (TiHeap, TiGlobals)
```



```
buildInitialHeap sc_defs = mapAccuml allocateSc hInitial sc_defs
```

`allocateSc` と呼ぶ「処理関数」は、単一のスーパーコンビネータを割り当て、新しいヒープと `sc_addrs` アソシエーションリストのメンバーを返します。

```
allocateSc :: TiHeap -> CoreScDefn -> (TiHeap, (Name, Addr))
allocateSc heap (name, args, body)
  = (heap', (name, addr))
  where
    (heap', addr) = hAlloc heap (NSupercomb name args body)
```

これでコンパイラの定義は完了です。次に、評価器に注意を向けます。

### 2.3.5 評価器

評価器の `eval` は、マシンの初期状態を取得し、一度に 1 ステップずつマシンを実行して、通過したすべての状態のリストを返します。

`eval` は、常に現在の状態を結果の最初の要素として返します。現在の状態が最終状態の場合、それ以上の状態は返されません。それ以外の場合、`eval` は次の状態に再帰的に適用されます。後者は、(`step` を使用して) 単一のステップを実行し、次に `doAdmin` を呼び出して、ステップ間に必要な管理作業を実行することによって取得されます。

```
eval state = state : rest_states
  where
    rest_states | tiFinal state = []
                | otherwise     = eval next_state
    next_state  = doAdmin (step state)
```

```
doAdmin :: TiState -> TiState
doAdmin state = applyToStats tiStatIncSteps state
```

#### 最終状態のテスト

関数 `tiFinal` は最終状態を検出します。スタックに単一のオブジェクトが含まれていて、それが数値またはデータオブジェクトのいずれかである場合にのみ、終了します。

```

tiFinal :: TiState -> Bool

tiFinal ([sole_addr], dump, heap, globals, stats)
    = isDataNode (hLookup heap sole_addr)

tiFinal ([], dump, heap, globals, stats) = error "Empty stack!"
tiFinal state = False --Stack contains more than one item

```

スタック要素はアドレスであることに注意してください。これは、数値であるかどうかを確認する前に、ヒープで検索する必要があります。スタックが空である必要がある場合 (これは決して発生しないはずですが) にも、適切なエラーメッセージを生成する必要があります。

最後に、isDataNode を定義できます。

```

isDataNode :: Node -> Bool
isDataNode (NNum n) = True
isDataNode node     = False

```

ステップを進める

関数 step は、ある状態をその後続状態に写像します。

```

step :: TiState -> TiState

```

スパインスタックの最上位のノードでケース分析を行う必要があるため、ヒープからこのノードを抽出し、dispatch を使用して適切な関数を呼び出し、ノードの各形式のハードワークを実行します。

```

step state
    = dispatch (hLookup heap (hd stack))
      where
        (stack, dump, heap, globals, stats) = state
        dispatch (NNum n)                    = numStep state n
        dispatch (NApp a1 a2)                 = apStep state a1 a2 -- 状態遷移ルール 2.1
        dispatch (NSupercomb sc args body)    = scStep state sc args body -- 状態遷移ルール 2.2

```

数値や関数適用のケースもほとんどトラブルなく対応できます。数値を関数として適用してはならないため、スタックの最上位に数値があるのはエラーです。(スタック上の唯一のオブジェクトである場合、実行は tiFinal によって停止されます。)

```

numStep :: TiState -> Int -> TiState
numStep state n
    = error "Number applied as a function!"

```

関数適用ノードの処理は、アンワインドルール (ルール 2.1) で説明されており、Miranda に直接変換できます。

```

apStep :: TiState -> Addr -> Addr -> TiState
apStep (stack, dump, heap, globals, stats) a1 a2
    = (a1 : stack, dump, heap, globals, stats)

```

### スーパーコンビネータの適用

スーパーコンビネータを適用するには、その本体をインスタンス化し、引数名をスタック内の引数アドレスにバインドする必要があります (ルール 2.2)。次に、redex のルートを含む引数をスタックから破棄し、代わりにリダクションの結果 (のルート) をスタックにプッシュします。(この最初のバージョンのマシンでは、更新を実行していないことに注意してください。)

```

scStep :: TiState -> Name -> [Name] -> CoreExpr -> TiState
scStep (stack, dump, heap, globals, stats) sc_name arg_names body
    = (new_stack, dump, new_heap, globals, stats)
    where
        new_stack          = result_addr : (drop (length arg_names + 1) stack)
        (new_heap, result_addr) = instantiate body heap env
        env                 = arg_bindings ++ globals
        arg_bindings       = zip2 arg_names (getargs heap stack)

```

スーパーコンビネータとプリミティブを適用するには、補助関数が必要です。関数 getArgs は、スタック (関数適用ノードのスタックの最上位にあるスーパーコンビネータで構成されている必要があります) を受け取り、スタック上の各関数適用ノードの引数から形成されたリストを返します。

```

-- now getargs since getArgs conflicts with Gofer standard.prelude
getargs :: TiHeap -> TiStack -> [Addr]
getargs heap (sc : stack)
    = map get_arg stack
    where
        get_arg addr = arg
                        where
                            (NAp fun arg) = hLookup heap addr

```

`instantiate` 関数は、式、ヒープ、および名前とアドレスを関連付ける環境を取ります。ヒープ内に式のインスタンスを作成し、インスタンスのルート 新しいヒープとアドレスを返します。この環境は、`instantiate` によって、スーパーコンビネータとローカル変数の代わりに使用されるアドレスを指定するために使用されます。

```
instantiate :: CoreExpr          -- Body of supercombinator
            -> TiHeap            -- Heap before instantiation
            -> ASSOC Name Addr  -- Association of names to addresses
            -> (TiHeap, Addr)   -- Heap after instantiation,
                                -- and address of root of instance
```

数字の場合は、非常に簡単です。

```
instantiate (ENum n) heap env = hAlloc heap (NNum n)
```

関数適用の場合も簡単です。2つのブランチをインスタンス化し、関数適用ノードを構築するだけです。`instantiate` の再帰呼び出しを介して、ヒープを「スレッド化」する方法に注目してください。つまり、`instantiate` の最初の呼び出しにヒープが与えられ、新しいヒープが生成されます。後者は、`instantiate` の2番目の呼び出しに与えられ、さらに別のヒープを生成します。後者は、新しい関数適用ノードが割り当てられるヒープであり、呼び出し元に返される最終ヒープを生成します。

```
instantiate (EAp e1 e2) heap env
  = hAlloc heap2 (NAp a1 a2)
  where
    (heap1, a1) = instantiate e1 heap env
    (heap2, a2) = instantiate e2 heap1 env
```

変数の場合、指定された環境で名前を検索するだけで、バインディングが見つからない場合は適切なエラーメッセージが生成されます。

```
instantiate (EVar v) heap env
  = (heap, aLookup env v (error ("Undefined name " ++ show v)))
```

Appendix A.2 で定義されている `aLookup` は、連想リストで変数を検索しますが、検索が失敗した場合は3番目の引数を返します。

```
instantiate (EConstr tag arity) heap env
  = instantiateConstr tag arity heap env
instantiate (ELet isrec defs body) heap env
  = instantiateLet isrec defs body heap env
instantiate (ECase e alts) heap env
  = error "Can't instantiate case exprs"
```

コンストラクタと `let(rec)` 式をインスタンス化する問題を延期するために、補助関数 `instantiateConstr` と `instantiateLet` を呼び出します。これらはそれぞれ、現在のエラーを生成します。後で、それらを運用定義に置き換えます。最後に、後で説明するように、テンプレートマシンは `case` 式をまったく処理できません。

```
instantiateConstr tag arity heap env
  = error "Can't instantiate constructors yet"
instantiateLet isrec defs body heap env
  = error "Can't instantiate let(rec) yet"
```

### 2.3.6 結果の書式付け

`eval` からの出力は状態のリストであり、全体を表示するとかなり大量になります。さらに、ヒープとスタックは抽象オブジェクトであるため、Miranda はそれらをまったく表示しません。したがって、`showResults` 関数は、セクション 1.5 で紹介した `iseq` データ型を使用して、出力をフォーマットします。

```
showResults :: [TiState] -> [Char]
showResults states
  = iDisplay (iConcat [ iLayn (map showState states),
                        showStats (last states)
                      ])
])
```

スタックの内容を表示するだけで状態を表示します。各ステップの後にヒープ全体を表示するのは面倒なので、スタックから直接参照されるノードの内容を表示することで満足します。状態の他のコンポーネントは変更されないため、それらも表示しません。

```
showState :: TiState -> Iseq
showState (stack, dump, heap, globals, stats)
  = iConcat [ showStack heap stack, iNewline ]
```

スタック上のアドレスと、それが指すノードの内容を表示することにより、スタックを最上位の要素から最初に表示します。これらのノードのほとんどは関数適用ノードであり、これらのそれぞれについて、引数ノードの内容も表示します。

```
showStack :: TiHeap -> TiStack -> Iseq
showStack heap stack
  = iConcat [ iStr "stk [",
             iIndent (iInterleave iNewline (map show_stack_item stack)),
```

```

        iStr " ]"
    ]
where
    show_stack_item addr
        = iConcat [ showFWAddr addr,
                    iStr ":",
                    showStkNode heap (hLookup heap addr)
                  ]

showStkNode :: TiHeap -> Node -> Iseq
showStkNode heap (NAP fun_addr arg_addr)
    = iConcat [ iStr "NAP ",
                showFWAddr fun_addr,
                iStr " ",
                showFWAddr arg_addr,
                iStr " (",
                showNode (hLookup heap arg_addr),
                iStr ")"
              ]

showStkNode heap node = showNode node

    showNode は、ノードの値を表示します。完全な値を出力するのではなく、
    NSupercomb ノード内に格納されている名前のみを出力します。実際、これ
    が名前がこれらのノード内に保存される唯一の理由です。

showNode :: Node -> Iseq
showNode (NAP a1 a2)
    = iConcat [ iStr "NAP ",
                showAddr a1,
                iStr " ",
                showAddr a2
              ]

showNode (NSupercomb name args body)
    = iStr ("NSupercom " ++ name)
showNode (NNum n)
    = (iStr "NNum ") 'iAppend' (iNum n)

showAddr :: Addr -> Iseq
showAddr addr = iStr (show addr)

showFWAddr :: Addr -> Iseq -- Show address in field of width 4

```

```
showFWAddr addr
  = iStr (space (4 - length str) ++ str)
  where
    str = show addr
```

showStats は、累積された統計を出力する責任があります。

```
showStats :: TiState -> Iseq
showStats (stack, dump, heap, globals, stats)
  = iConcat [ iNewline,
              iNewline,
              iStr "Total number of steps = ",
              iNum (tiStatGetSteps stats)
            ]
```

演習 2.4 これまでに与えられた実装をテストします。適切なテストプログラムは次のとおりです。

```
main = S K K 3
```

結果は 3 になります。さらにいくつかのテストプログラムを考案し、それらが機能することを確認します。算術演算はまだ定義されていないことを忘れないでください。

演習 2.5 ヒープの内容全体を出力するように showState を変更します。(ヒント: ヒープ内のすべてのノードのアドレスを検出するには、hAddresses を使用します。) このようにして、ヒープが 1 つのステップから次のステップにどのように進化するかを確認できます。

演習 2.6 スーパーコンビネータまたはプリミティブが適用される引数が少なすぎると、scStep は失敗します。このケースを検出するために、適切なチェックとエラーメッセージを scStep に追加します。

演習 2.7 より多くの実行統計を収集するようにインタープリタを変更します。たとえば、次のような統計情報を累積できます。

- 簡約の数。おそらくスーパーコンビネータの簡約とプリミティブの簡約に分けられます。
- 各種のヒープ操作の数、特に割り当て。これを行う最も便利な方法は、heap 抽象データ型を変更してこの情報自体を累積することですが、これは結果の一部として新しいヒープを返すヒープ操作でのみ機能します。

- 最大スタック深度。

演習 2.8 `scStep` の定義では、`instantiate` に渡される環境 `env` は次のように定義されます。

```
env = arg\_bindings ++ globals
```

`++` の引数を逆にすると、どのような違いがありますか？

演習 2.9 (少し注意が必要です。) 次の `eval` の定義は、与えられたものよりも明白だと思うかもしれません。

```
eval state = [state],          tiFinal state  
            = state : eval next_state, otherwise
```

(`next_state` は以前と同じように定義されています)。なぜこれが劣った定義なのですか？(ヒント：すべての状態が `showResults` によってフォーマットされていて、存在しないヒープノードにアクセスしようとするなど、`tiFinal state` を評価するときにエラーが発生した場合にどうなるかを考えてください。エラーの原因となった状態が出力されますか？そうでない場合は、なぜですか？)



## 2.4 Mark 2: let(rec) 式

最初の機能強化は、マシンが `let` および `letrec` 式を処理できるようにすることです。2.1.4 で説明したように、スーパーコンビネータの本体には、グラフのテキスト記述とみなされる `let(rec)` 式が含まれる場合があります。

したがって、実装に加える必要がある唯一の変更は、インスタンス化を拡張することであり、その結果、`Elet` コンストラクの等式が得られます。

演習 2.10 非再帰的な `let` 式用の `instantiate` 関数定義を追加します。インスタンス化するために必要なこと (`Elet nonRecursive defs body`) は次のとおりです。

1. `defs` の各定義の右辺をインスタンス化します。
2. 環境を拡張して、`defs` の名前を新しく構築されたインスタンスのアドレスにバインドします。
3. 拡張した環境と式本体を渡す `instantiate` を呼び出します。

これはまだ `let` 式のみを処理します。`letrec` 式をインスタンス化した結果は閉路グラフですが、`let` 式は閉路グラフを生成するもとなります。

演習 2.11 非再帰的 `Elet` 用 `instantiate` 関数の定義をコピーし、再帰的な場合に機能するように変更します (または、両方を処理するように定義を変更します)。(ヒント: ステップ 1 で、既存の環境の代わりに拡張環境 (ステップ 2 で構築) を渡してインスタンス化することを除いて、`let` の場合とまったく同じようにすべてを実行します。)

この演習のヒントは、ステップ 2 で作成された名前とアドレスのバインディングをステップ 1 への入力として使用する必要があるため、不思議に思われます。Miranda でこれを試してみると、すべてが完全に機能します。これは、他の非厳密な関数型言語と同様に、関数を呼び出す前に関数への入力を評価する必要がないためです。実際の実装では、`letrec` 内の各 (ルート) ノードが割り当てられるアドレスを計算し、この情報を反映するように環境を拡張してから、定義式の右辺をインスタンス化することにより、このトリックを「手動で」実行する必要があります。

実装が機能するかどうかを確認するためのテストプログラムを次に示します。

```
pair x y f = f x y ;
fst p = p K ;
snd p = p K1 ;
```

```

f x y = letrec
    a = pair x b ;
    b = pair y a
  in
    fst (snd (snd (snd a))) ;
main = f 3 4

```

結果は 4 になります。このプログラムがどのように機能するか理解できますか？(すべてはセクション 2.8.3 で明らかにされます。)

演習 2.12 次のプログラムを検討します。

```

main = letrec f = f x in f

```

このプログラムを実行するとどうなりますか？この問題は、Miranda のような強く型付けされた言語で発生する可能性がありますか？

## 2.5 Mark 3: 更新機能の追加

これまでのところ簡約器は更新を実行しないため、共有部分は何度も評価される可能性があります。

セクション 2.1.5 で説明した様に、問題を解決する最も簡単な方法は、結果を指す間接ノードで `redex` のルートを更新することです。

(2.3)	$a_0 : a_1 : \dots : a_n : s \quad d \quad h[a_0 : \text{NSupercomb } [x_1, \dots, x_n] \text{ body}] \quad f$ $\Rightarrow \quad a_r : s \quad d \quad h'[a_n : \text{NInd } a_r] \quad f$ <p>where <math>(h', a_r) = \text{instantiate body } h \ f[x_1 \mapsto a_1, \dots, x_n \mapsto a_n]</math></p>
-------	---

違いは、インスタンス化関数によって返されるヒープ  $h'$  は、ノード  $a_n$  (redex のルート) を  $a_r$  (インスタンス化によって返される結果のルート) への間接参照で上書きすることによってさらに変更されることです。スーパーコンビネータが CAF (セクション 2.1.6 を参照) の場合、 $n = 0$  であり、変更されるノードはスーパーコンビネータノード自体であることに注意してください。

さらに 1 つの変更が必要です。スパインをほどくときに間接に遭遇する可能性があるため、この場合に対処するために新しいルールを追加する必要があります。

(2.4)	$a : s \quad d \quad h[a : \text{NInd } a_1] \quad f$ $\Rightarrow \quad a_1 : s \quad d \quad h \quad f$
-------	---

間接ノードのアドレス  $a$  は、あたかもそこになかったかのように、スタックから削除されます。

これらの新しいルールを実装するには、いくつかのことは行う必要があります。

- 新しいノードコンストラクター `NInd` を `node` データ型に追加します。これにより、次の改訂された定義が得られます。

```
data Node = NAp Addr Addr -- Application
          | NSupercomb Name [Name] CoreExpr -- Supercombinator
          | NNum Int -- Number
          | NInd Addr -- Indirection
```

この余分なコンストラクターを考慮に入れるために、`showNode` に新しい定義式を追加する必要があります。

- 結果への間接参照で `redex` のルートを更新するために `hUpdate` を使用するように `scStep` を変更します (ルール 2.3)。
- 間接化に対処するために、`dispatch` の定義に新たな定義式を追加します (ルール 2.4)。

演習 2.13 間接ノードで更新を実行するように変更を加えます。リダクションマシンの Mark1 バージョンと Mark3 バージョンの両方で次のプログラムを実行して、変更の効果を試してください。

```
id x = x ;
main = twice twice id 3
```

(`twice` は `preludeDefs`(セクション 1.4) で定義されていることを思い出してください。) 最初に手でそれを簡約することによって何が起るかを理解するようにしてください。 `main` を `twice twice twice id3` と定義するとどうなりますか？

### 2.5.1 間接参照の数を減らす

多くの場合、`instantiate` によって (または、後で説明するように、プリミティブによって) 新しく作成されたノードへの間接参照を使用して、`redex` のルートを更新します。このような状況では、間接参照を使用するのではなく、結果のルートノードを `redex` のルートの上に直接構築する方が安全です。結果のルートが新しく作成されるため、これを行うことで共有が失われることはなく、余分な間接ノードを構築 (およびその後トラバース) する手間が省けます。

これを行うには、新しいインスタンス化関数 `instantiateAndUpdate` を定義します。これは、追加の引数、結果で更新されるノードのアドレスを取り、結果のグラフのアドレスを返さないことを除いて、`instantiate` と同じです。

```
instantiateAndUpdate
  :: CoreExpr          -- Body of supercombinator
  -> Addr              -- Address of node to update
  -> TiHeap             -- Heap before instantiation
  -> ASSOC Name Addr    -- Associate parameters to addresses
  -> TiHeap             -- Heap after instantiation
```

たとえば、式が関数適用の場合のインスタンス化と更新の定義は次のとおりです。

```

instantiateAndUpdate (EAp e1 e2) upd_addr heap env
= hUpdate heap2 upd_addr (NAP a1 a2)
  where
    (heap1, a1) = instantiate e1 heap env
    (heap2, a2) = instantiate e2 heap1 env

```

再帰的なインスタンス化は、古いインスタンス化によって引き続き実行されることに注意してください。ルートノードのみを更新する必要があります。

演習 2.14 `instanceiateAndUpdate` の定義を完了します。次の点には少し注意が必要です。

- インスタンス化される式が単純な変数である場合でも、間接参照を使用する必要があります。なんで？
- `let(rec)` 式の方程式の再帰的なインスタンス化について慎重に検討してください。

`scStep` を変更して、インスタンス化の代わりに `instantiateAndUpdate` を呼び出し、更新するノードのアドレスとして `redex` のルートを渡します。 `scStep` 自体から更新コードを削除します。削減の数と割り当てられたヒープノードの数に対するこの変更の影響を測定します。

## 2.6 Mark 4: 算術演算機能の追加

このセクションでは、算術プリミティブを追加します。これには、初めてダンプを使用することが含まれます。

### 2.6.1 算術演算の遷移規則

### 2.6.2 算術演算の実装

算術演算を実装するには、いくつかの変更を加える必要があります。まず、型 `tiDump` を、初期値が空のスタックのスタックに再定義する必要があります。

```
type TiDump = [TiStack]
initialTiDump = []
```

次に、新しい種類のヒープノードを追加する必要があります。 `NPrim n p` は、名前が `n` で、値が `p` であるプリミティブを表します。ここで、`p` はプリミティブタイプです。 `NSupercomb` ノードの場合と同様に、名前はデバッグと文書化の理由でのみ `NPrim` ノードに存在します。

```
data Node = NAp Addr Addr           -- Application
          | NSupercomb Name [Name] CoreExpr -- Supercombinator
          | NNum Int                -- Number
          | NInd Addr               -- Indirection
          | NPrim Name Primitive     -- Primitive
```

いつものように、 `NPrim` ノードを表示するには、 `showNode` も拡張する必要があります。前のセクションで示した遷移ルールは、データ型 `primitive` を次のように定義する必要があることを示しています。

```
data Primitive = Neg | Add | Sub | Mul | Div
```

必要なプリミティブごとに1つのコンストラクターを使用します。

ここで、各スーパーコンビネータの初期ヒープに `NSupercomb` ノードを割り当てる必要があるのと同じように、各プリミティブの初期ヒープに `NPrim` ノードを割り当てる必要があります。次に、マシン状態のグローバルコンポーネントに追加のバインディングを追加できます。これにより、スーパーコンビネータの場合と同様に、各プリミティブの名前がそのノードのアドレスにマップされます。次のように `buildInitialHeap` の定義を変更することで、これを簡単に行うことができます。

```
buildInitialHeap :: [CoreScDefn] -> (TiHeap, TiGlobals)
buildInitialHeap sc_defs
```

```

= (heap2, sc_addrs ++ prim_addrs)
  where
    (heap1, sc_addrs) = mapAccum1 allocateSc Initial sc_defs
    (heap2, prim_addrs) = mapAccum1 allocatePrim heap1 primitives

```

変数名からプリミティブへのマッピングを提供する関連付けリストを定義します。したがって、次のようになります。

```

primitives :: ASSOC Name Primitive
primitives = [ ("negate", Neg),
               ("+", Add), ("-", Sub),
               ("*", Mul), ("/", Div)
             ]

```

さらにプリミティブを追加するには、`primitive` 型にコンストラクターを追加し、`primitives` 関連付けリストに要素を追加します。

次に、`allocateSc` を定義したのと同じように、`allocatePrim` を定義できます。

## 2.7 Mark 5: 構造化データ型の追加

このセクションでは、構造化データ型をリダクションマシンに追加します。コア言語の `case` 式の実装を提供するのは良いことですが、テンプレートインスタンス化マシンのフレームワーク内で実装するのはかなり難しいことがわかりました。(後の実装ではこの問題は発生しません。) 代わりに、`if`, `casePair`, `caseList` などの組み込み関数のコレクションを使用して、特定の構造化型を操作できるようにします。テンプレートマシンは、一般的な構造化オブジェクトを処理できません。

演習 2.18 テンプレートインスタンス化マシンに `case` 式を導入するのが難しいのはなぜですか?(ヒント: インスタンス化が `case` 式で何をするかを考えてください。)

### 2.7.1 構造化データの構築

構造化データは、コンストラクターのファミリー `Pack{t, a}` で構築されます。ここで、`t` はコンストラクターのタグを示し、`a` はそのアリティを示します(セクション 1.1.4)。したがって、グラフでこれらのコンストラクター関数を表現する必要があります。これらは実際には新しい形式のプリミティブであるため、プリミティブ型に新しいコンストラクタ `PrimConstr` を追加することでこれを行うことができます。これで、`instantiateConstr` の式で、式 `EConstr t a` をヒープノード `NPrim "Pack"(PrimConstr t a)` にインスタンス化できます。

次に、このプリミティブがどのように実装されているかという疑問が生じます。新しい補助関数 `primConstr` を呼び出す `PrimConstr` コンストラクターと一致するように、`primStep` にケースを追加する必要があります。これにより、十分な引数が指定されていることを確認し、指定されている場合は、ヒープ内に構造化データオブジェクトを構築します。

次に、このプリミティブがどのように実装されているかという疑問が生じます。新しい補助関数 `primConstr` を呼び出す `PrimConstr` コンストラクターと一致するように、`primStep` にケースを追加する必要があります。これにより、十分な引数が指定されていることを確認し、指定されている場合は、ヒープ内に構造化データオブジェクトを構築します。

これを行うには、構造化データオブジェクトを表す新しいコンストラクタ `NData` をノードタイプに追加する必要があります。`NData` コンストラクタには、オブジェクトのタグとそのコンポーネントが含まれています。



```

data Node = NAp Addr Addr          -- Application
          | NSupercomb Name [Name] CoreExpr -- Super combinator
          | NNum Int                -- Number
          | NInd Addr               -- Indirection
          | NPrim Name Primitive    -- Primitive
          | NData Int [Addr]        -- Tag, list of components

```

これで、NPrim (PrimConstr  $t$   $n$ ) のルールを指定できます。

(2.10)	$a_0 : a_1 : \dots : a_n : [] \quad d \quad h$	$\left[ \begin{array}{l} a : \text{NPrim (PrimConstr } t \ n) \\ a_1 : \text{NAp } a \ b_1 \\ \dots \\ a_n : \text{NAp } a_{n-1} \ b_n \end{array} \right]$	$f$
	$\Rightarrow$	$a_n : [] \quad d \quad h[a_n : \text{NData } t \ [b_1, \dots, b_n]]$	$f$

構造化オブジェクトの構築についてはこれだけです。次の質問は、それらをどのように分解するかです。これは、コア言語の case 式で表されます。すでに述べたように、case 式を直接実装するのは難しいため、ブール値から始まるいくつかの特殊なケースに満足しています。

### 2.7.2 条件式

ブール型は、Miranda で次のように宣言される可能性があります。

```
boolean ::= False | True
```

True と False の 2 つのコンストラクタがあります。それぞれアリティは 0 であり、タグ 1 を False に、タグ 2 を True に任意に割り当てます。したがって、次のコア言語定義を与えることができます。

```

False = Pack{1, 0}
True  = Pack{2, 0}

```

一般的な case 式を使用することはできないため、次の簡約ルールを使用して条件付きプリミティブを追加するだけで十分です。

```

if Pack{2, 0} t e = t
if Pack{1, 0} t e = e

```

動作上、データオブジェクトであると想定される最初の引数を評価し、タグを調べ、タグが 2(True) または 1(False) のどちらであるかに応じて、2 番目または 3 番目の引数を選択します。

演習 2.19 条件付きプリミティブの状態遷移ルールを記述します。3つのルールが必要です。ブール条件がすでに評価されている場合に削減を実行するための2つのルール。1つは、条件が評価されていない場合に、古いスタックをダンプにプッシュし、条件のアドレスを新しい空のスタックにプッシュすることによって評価を開始します（ルール 2.9 を参照）。最初の2つのルールの更新では、間接参照を使用する必要があることに注意してください。もう1つのルールがありません。それは何ですか？（ヒント：条件の評価が完了すると、条件はどのように再試行されますか？）

if があれば、それと False および True に関して、他のブール演算子のコア言語定義を与えることができます。例えば：

```
and x y = if x y False
```

演習 2.20 or、xor および not のコア言語定義を指定します。これらのコア言語定義をすべて extraPreludeDefs に追加します。

最後に、数値を比較する何らかの方法が必要です。これには、新しいプリミティブ >、>= などがが必要です。

### 2.7.3 構造化データの実装

構造化データオブジェクト、条件、比較操作を追加するために実装に必要な変更のリストを次に示します。

- NData コンストラクターを node データ型に追加します。showNode を拡張して、NData ノードを表示します。
- PrimConstr、If、Greater、GreaterEq、Less、LessEq、Eq、NotEq を primitive 型に追加します。最初のものを除くすべての場合、primitive の関連付けリストに適切なペアを追加して、これらのプリミティブの名前を instantiateVar によってそれらの値にマップできるようにします。
- instanceiateConstr(および必要に応じて instantiateAndUpdateConstr) の定義を追加します。
- isDataNode 関数は、NNum ノードだけでなく NData ノードも識別する必要があります。
- step の dispatch コードには、新しい補助関数 dataStep を呼び出す、NData ノードの追加のケースが必要です。

- `dataStep` の定義; `numStep` と非常によく似ています。
- `primStep` を拡張して、新しいプリミティブ `PrimConstr`、`If`、`Greater` などに対応します。`PrimConstr` と `If` の場合、新しい補助関数 `primConstr` と `primIf` を呼び出す必要があります。比較プリミティブは、`primArith` をほとんど使用できますが、完全には使用できません。必要なのは、`primArith` のわずかな一般化です。

```
primDyadic :: TiState -> (Node -> Node -> Node) -> TiState
```

これは、数値を組み合わせる関数ではなく、ノードを組み合わせる関数を取ります。`primDyadic` に関して、`primArith` と同様の関数 `primComp` を比較プリミティブとして定義するのは簡単です。そして、`primArith` の定義を一般化することによって `primDyadic` を定義します。

**演習 2.21** これらすべての変更を行います。再帰を終了する条件があるので、ついに、賢明な再帰関数を書くことができます。たとえば、階乗関数を試してください

```
fac n = if (n == 0) 1 (n * fac (n-1)) ;
main = fac 3
```

#### 2.7.4 ペア

ブール値のコンストラクタは両方ともアリティが 0 となっています。次に、ペアのデータ型を追加します。これは、Miranda で次のように宣言される可能性があります。

```
pair * ** ::= MkPair * **
```

`Pack 1, 2` コンストラクタを使用してペアを作成できます。

```
MkPair = Pack{1, 2}
```

`case` 式を使わずに、それらを分解してみませんか？たとえば、`case` 式を含む次のコア言語プログラムについて考えてみます。

```
f p = case p of
      <1> a b -> b*a*a end
```

`case` 式がないため、代わりに次のように翻訳できます。

```
f p = casePair p f'
f' a b = b*a*a
```

ここで、`f'` は補助関数であり、`casePair` は次のように定義された組み込みのプリミティブです。

```
casePair (Pack{1, 2} a b) f = f a b
```

操作上、`casePair` は最初の引数を評価します。これはペアを生成することを期待しています。次に、2 番目の引数をペアの 2 つのコンポーネントに適用します。これを実装するには、プリミティブ型にさらに別のコンストラクタ `PrimCasePair` を追加し、それを処理するコードをさらに記述します。

たとえば、次のコア言語定義を使用して、ペアの 1 番目と 2 番目のコンポーネントを抽出する `fst` と `snd` を定義できます。

```
fst p = casePair p K
snd p = casePair p K1
```

**演習 2.22** `casePair` の状態遷移ルールを記述します。いつものように、2 つのルールが必要になります。1 つは最初の引数が評価された場合に削減を実行し、もう 1 つは評価されなかった場合にその評価を開始します。上記のように、ペアを実装するために必要な変更を加えます。次のプログラム (および独自の他のプログラム) を使用して実装をテストします。

```
main = fst (snd (fst (MkPair (MkPair 1 (MkPair 2 3)) 4)))
```

### 2.7.5 リスト

ペアとブール値を作成したので、リストは簡単にはずです。リストのデータ型は、Miranda で次のように定義されている可能性があります。

```
list * ::= Nil | Cons * (list *)
```

タグ 1 を `Nil` に、タグ 2 を `Cons` に割り当てます。

唯一の問題は、リストを分解する `caseList` プリミティブが正確に何をすべきかということです。`casePair` には、ペアのコンポーネントを引数として取る関数である「継続」が 1 つあることを思い出してください。`if` には 2 つの「継続」があり、最初の引数の値に応じていずれかを選択します。したがって、`caseList` は、これら両方のアイデアを組み合わせたものにすぎません。

```
caseList Pack{1, 0}      cn cc = cn
caseList (Pack{2, 2} x xs) cn cc = cc x xs
```

3つの引数を取り、最初の引数を評価します。それが空のリストである場合 (つまり、タグが1でコンポーネントがない場合)、caseList は単に2番目の引数 cn を選択します。それ以外の場合は、リストセル (つまり、タグが2でコンポーネントが2つ) である必要があり、caseList は3番目の引数をこれらのコンポーネントに適用します。

たとえば、Miranda で記述される length 関数を実装したいとします。

```
length [] = 0
length (x : xs) = 1 + length xs
```

caseList を使用すると、次のように length を記述できます。

```
length xs = caseList xs 0 length'
length' x xs = 1 + length xs
```

**演習 2.23** Cons、Nil、head、tail のコア言語定義を記述します。head と tail を定義するには、新しいプリミティブ abort を導入する必要があります。これは、空リストの head または tail を取得した場合に返されます。abort は、Miranda の error プリミティブを呼び出してプログラムを停止することで便利に実装できます。

**演習 2.24** caseList の状態遷移ルールを記述し、それと abort を実装して、preludeDefs に Cons、Nil、head、tail の定義を追加します。実装をテストするためのプログラムをいくつか作成します。

これで、考えたい構造化データ型に適した case プリミティブを記述できるようになります。

**演習 2.25** 完全な case 式を実装するのではなく、case プリミティブを使用して構造化データ型を分解することの主な欠点は何ですか？

### 2.7.6 リストの表示

これを説明するための状態遷移ルールを書き始めるとすぐに、状態遷移の世界で「数値を出力する」という考えをどのように表現するかを決定する必要があります。最も簡単な解決策は、出力と呼ばれる新しいコンポーネントを状態に追加し、「出力に数値を追加する」ことによって「数値を出力する」モデルを作成することです。また、Print と Stop という2つの新しいプリミティブを追加します。

Stop プリミティブは簡単です。スタックを空にします。(tiFinal は、現在のようにエラーを出すのではなく、空のスタックを検出したときにマシン

を停止するように変更されます。) Stop は、ダンプが空であることを期待しています。

(2.11)	$\begin{array}{ccccccc} o & a : [] & [] & h[a : \text{NPrim Stop}] & f \\ \Rightarrow & o & [] & h & f \end{array}$
--------	---

Print プリミティブは、最初の引数を整数に評価し、その値を出力リストに付加します。次に、結果として 2 番目の引数を返します。また、ダンプが空であることも想定しています。最初の引数がすでに評価されている場合、最初のルールが適用されます。

(2.12)	$\begin{array}{ccccccc} o & a : a_1 : a_2 : [] & [] & h & \left[ \begin{array}{l} a : \text{NPrim Print} \\ a_1 : \text{NAp } a \ b_1 \\ a_2 : \text{NAp } a_1 \ b_2 \\ b_2 : \text{NNum } n \end{array} \right] & f \\ \Rightarrow & o \ ++ \ [n] & b_2 : [] & [] & h & f \end{array}$
--------	---

Print は、出力  $o$  に副作用があるため、かなり奇妙なスーパーコンビネータです。Print は明らかに注意して使用する必要があります！2 番目のルールは、Print の最初の引数が評価されない場合に適用されます。つまり、通常の方法で評価を開始します。

(2.13)	$\begin{array}{ccccccc} o & a : a_1 : a_2 : [] & [] & h & \left[ \begin{array}{l} a : \text{NPrim Print} \\ a_1 : \text{NAp } a \ b_1 \\ a_2 : \text{NAp } a_1 \ b_2 \end{array} \right] & f \\ \Rightarrow & o & b_1 : [] & (a_2 : []) : [] & h & f \end{array}$
--------	---

ここで、extraPreludeDefs で次の追加関数を定義します。

```
printList xs = caseList xs stop printCons
printCons h t = print h (printList t)
```

ここで、print は primitives によって Print プリミティブにバインドされ、stop は Stop にバインドされます。

最後に、スタックに式 (printList main) のアドレスが最初に含まれるように、compile 関数を変更します。これが正しいことをしていることを自分自身に納得させるのにそれほど時間はかからないはずです。

**演習 2.26** これらの変更を実装し、数値のリストを返すプログラムを作成して実装をテストします。

## 2.8 代替実装 †

これらの演習では、私たちが行ったことに対するいくつかの代替実装について説明します。

### 2.8.1 プリミティブの代替表現

プリミティブに対して行うのはそれを実行することだけなので、次のトリックを実行できます。primitive をケース分析を実行する列挙型にする代わりに、tiState を tiState に変換する以下のような関数にすることができます。

```
Type Primitive = TiState -> TiState
```

コンストラクターの Add、Sub などには完全に消滅しました。これで、primStep によって実行される「ケース分析」はかなり簡単になりました。関数を適用するだけです。primStep とプリミティブの改訂された定義は次のとおりです。

```
primStep state prim = prim state
primitives = [ ("negate", primNeg),
               ("+", primArith (+)), ("-", primArith (-)),
               ("*", primArith (*)), ("/", primArith (/))
             ]
```

これは、実際の実装では直接対応します。プリミティブを区別するために NPrim ノードに小さな整数タグを格納する代わりに、コードポインタを格納し、それにジャンプしてプリミティブを実行します。

演習 2.27 この変更を実装してテストします。

### 2.8.2 ダンプの代替表現

現在、ダンプをスタックのスタックとして実装していますが、実際の実装では、古いスタックの上に新しいスタックを直接構築することは間違いありません。ダンプには、スパインスタックのベースからのオフセットが含まれ、1つのサブスタックが終了し、次のサブスタックが開始する場所を示します。

次の型宣言を使用して、これをマシンで直接モデル化できます。

```
type TiDump = Stack Num
```

演習 2.28 この変更を実装してテストします。引数の評価の開始と終了、および tiFinal の定義を処理する関数定義を変更する必要があります。

### 2.8.3 データ値の代替表現

## 2.9 ガベージコレクション

実行が進むにつれて、ヒープに割り当てられるノードが増えるため、ヒープを表す Miranda データ構造はますます大きくなります。最終的に、Miranda はスペースを使い果たします。これは実際の実装に直接対応しているため、当然のことです。ノードが割り当てられると、ヒープはどんどん大きくなり、最終的にはいっぱいになります。スペースを解放するためにガベージコレクションを実行する必要があります。

より具体的には、型も含めて関数 `gc` を定義する必要があります。

```
gc :: TiState -> TiState
```

その結果の状態は、ヒープが(うまくいけば)小さいことを除いて、入力状態とまったく同じように動作します。この小さなヒープには、マシン状態の他のコンポーネントから直接または間接的にアクセスできるすべてのノードが含まれています。`gc` は、不要になったノードのアドレスで `hFree` を呼び出すことにより、ヒープを小さくします(`hFree` の説明については、Appendix A.1 を参照してください)。

`doAdmin` 関数は、各ステップの後に (`hSize` を使用して) ヒープサイズをチェックし、指定されたサイズよりも大きい場合はガベージコレクタを呼び出すことができます。

### 2.9.1 マークスキャンコレクション

まず、マークスキャンコレクタを開発します。これは3つのフェーズで機能します。

1. 最初のフェーズでは、すべてのルートを識別します。つまり、マシンの状態に含まれるすべてのヒープアドレスです。そのようなアドレスはどこに潜んでいる可能性がありますか？`addr` の出現について、マシンの状態に関するタイプを調べることで簡単に見つけることができます。答えは、アドレスがスタック、ダンプ、およびグローバルで発生する可能性があるということです。したがって、次の関数が必要です。

```
findStackRoots  :: TiStack  -> [Addr]
findDumpRoots   :: TiDump    -> [Addr]
findGlobalRoots :: TiGlobals -> [Addr]
```



2. マークフェーズでは、アドレスがマシンのステートにある各ノードがマークされます。ノードがマークされると、そのすべての子孫もマークされ、以下同様に再帰的にマークされます。markFrom 関数は、ヒープとアドレスを受け取り、そのアドレスからアクセス可能なすべてのノードがマークされた新しいヒープを返します。

```
markFrom :: TiHeap -> Addr -> TiHeap
```

3. スキャンフェーズでは、ヒープ内のすべてのノード (マークされているかどうかに関係なく) が検査されます。マークされていないノードは解放され、マークされたノードはマーク解除されます。

```
scanHeap :: TiHeap -> TiHeap
```

演習 2.30 findRoots、markFrom、scanHeap の観点から gc の定義を記述し、doAdmin から適切に呼び出します。

演習 2.31 findRoots の定義を記述します。

markFrom と scanHeap を実装する前に、ノードをマークする方法が必要です。実際の実装では、これはノードのビットを使用して、ノードがマークされているかどうかを示すことによって行われます。次のように、新しいコンストラクタである node 型を追加してこれをモデル化します。

```
data Node = NAp Addr Addr -- Application
          | NSupercomb Name [Name] CoreExpr -- Supercombinator
          | NNum Int -- Number
          | NInd Addr -- Indirection
          | NPrim Name Primitive -- Primitive
          | NData Int [Addr] -- Tag, list of components
          | NMarked Node -- Marked node
```

新しい種類のノードはNMarked ノードであり、マーキングが行われる前に存在していた node がその中に含まれています。NMarked ノード内の node が別の NMarked ノードになることはありません。

これで、markFrom を定義する準備が整いました。アドレス  $a$  とヒープ  $h$  が与えられると、次のようになります。

1.  $h$  で  $a$  を検索し、ノード  $n$  を与えます。すでにマークされている場合、markFrom はすぐに戻ります。これが、ヒープ内で循環構造に遭遇したときにマーキングプロセスが永久に続くのを防ぐものです。

2. `hUpdate` を使用してノードをマークし、`NMarked n` に置き換えます。
3.  $n$  内から任意のアドレスを抽出し (そのようなアドレスは 0 個以上ある場合があります)、それぞれに対して `markFrom` を呼び出します。

残っているのは `scanHeap` だけです。 `hAddresses` を使用して、ヒープで使われているすべてのアドレスのリストを抽出し、それぞれを順番に調べます。参照するノードがマークされていない場合 (つまり、`NMarked` ノードではない場合)、`hFree` を呼び出してノードを解放します。それ以外の場合は、`hUpdate` を使用してノードのマークを解除し、`NMarked` コンストラクタ内にあるノードに置き換えます。

演習 2.32 `markFrom` と `scanHeap` の定義を記述します。

これで、マークスキャンガベージコレクタは完了です。

ガベージコレクションを実行する方法はマークスキャンだけではありません。さらに調査するためのいくつかの方向性を提案します。ガベージコレクション手法の簡単な調査は、[Peyton Jones 1987] の第 17 章にあります。より包括的なレビューは [Cohen 1981] です。

## 2.9.2 間接参照の排除

まず、開発したばかりのコレクタの最適化から始めます。評価中に、間接ノードを導入する場合があります。図 2.2 に示すように、ポインタを再調整することにより、間接ノードを排除すると便利です。




Figure 2.2: Eliminating indirections during garbage collection

これを行うには、`markFrom` の機能を少し変更する必要があります。これで、アドレスとヒープを取得し、そのアドレスからアクセス可能なすべてのノードにマークを付け、古いアドレスの代わりに使用する必要がある新しいアドレスとともに新しいヒープを返す必要があります。

```
markFrom :: TiHeap -> Addr -> (TiHeap, Addr)
```

図では、ノード (a) のアドレスを使用して `markFrom` を呼び出すと、ノード (c) をマークし (ノード (b) はマークしない)、ノード (c) のアドレスを返す必要があります。

`markFrom` から返されたアドレスをどのように利用しますか? `markFrom` が呼び出されたアドレスの代わりに挿入する必要があります。これを行う最も簡単な方法は、最初の 2 つのフェーズをマージすることです。これにより、各ルートがマシン状態で識別されると、`markFrom` が呼び出され、返されたアドレスを使用して、マシン状態の元のルートが置き換えられます。したがって、`findStackRoots` とそのコンパニオンを次のように置き換えます。

```
markFromStack :: TiHeap -> TiStack -> (TiHeap, TiStack)
```

```
markFromDump      :: TiHeap -> TiDump      -> (TiHeap, TiDump)
markFromGlobals   :: TiHeap -> TiGlobals -> (TiHeap, TiGlobals)
```

演習 2.33 markFrom の改訂版を実装し、間接をマークせずに「スキップオーバー」し、再帰的に呼び出すときに各ノード内のアドレスを更新します。次に、markFrom の観点から他のマーキング関数を実装し、それらを新しいバージョンの gc で接着します。この新しいコレクタで取得したヒープサイズを以前に取得したヒープサイズと比較して、改善を測定します。(markFrom から NInd の特別な処理を削除することで、以前の状態に簡単に戻すことができます。)

### 2.9.3 ポインタ反転

ヒープ内のすべての  $N$  ノードがたまたま 1 つの長いリストにリンクされている場合、markFrom はそれ自体を  $N$  回再帰的に呼び出します。実際の実装では、これはヒープが大きいと同じくらい深いスタックを構築します。非常にまれな状況で、ヒープと同じ大きさのスタックをカウントに割り当てる必要があるのは非常に面倒です。

ポインタ反転と呼ばれる巧妙なトリックがあります。これは、マークされているノードそのものをリンクすることでスタックを排除できます [Schorr and Waite1967]。アルゴリズムによって課せられる唯一の追加要件は、マークされたノードが数ビットの追加の状態情報を必要とすることです。これは、NMarked コンストラクターをいくらか拡張することで表現できます。

```
data Node = NAp Addr Addr          -- Application
          | NSupercomb Name [Name] CoreExpr -- Supercombinator
          | NNum Int                -- Number
          | NInd Addr               -- Indirection
          | NPrim Name Primitive    -- Primitive
          | NData Int [Addr]        -- Tag, list of components
          | NMarked MarkState Node  -- Marked node

data markState = Done              -- Marking on this node finished
               | Visits Int -- Node visited n times so far
```

markState のコンストラクターの意味については、簡単に説明します。

別の (まったく別の) 状態遷移システムを使用して、ポインタ反転アルゴリズムを説明できます。マーキングマシンの状態には、フォワードポインタ、

バックワードポインタ、およびヒープの3つのコンポーネント  $(f, b, h)$  があります。markFrom を呼び出すたびに、マシンの新しい実行が開始されます。markFrom がアドレス  $a$  とヒープ  $h_{init}$  で呼び出されると、マシンは次の状態から開始されます。

$$(a, \text{hNull}, h_{init})$$

(hNull は、ヒープ内のオブジェクトをアドレス指定しない型 addr の識別値であり、通常のアドレスと区別できます。) マシンは、つぎの状態になると終了します。

$$(f, \text{hNull}, h[f : \text{NMarked Done } n])$$

つまり、 $f$  がマークされたノードを指し、 $b = \text{hNull}$  の場合です。 $(f$  が指すノードがすでにマークされている場合は、初期状態も最終状態である可能性があります。)

遷移規則は、マシンがマークされていないノードを処理する方法を説明することから始めます。現在のところ、NData ノードは無視されます。まず、関数適用の場合を扱います。マークされていない関数適用に遭遇すると、最初のサブグラフに「降下」し、NAp ノードの最初のフィールドに古いバックポインタを記録します。新しいフォワードポインタは最初のサブグラフをアドレス指定し、新しいバックポインタは関数適用ノード自体をアドレス指定します。状態情報 Visits 1 は、バックポインタが NAp ノードの最初のフィールドに保持されているという事実を記録します。

(x.xx)	$f \quad b \quad h[f : \text{NAp } a_1 \ a_2]$
$\Rightarrow$	$a_1 \quad f \quad h[f : \text{NMarked (Visits 1) (NAp } b \ a_2)]$

それまでのフォワードポインタが指していたアドレスの内容を NMarked (Visits 1) (NAp b a2) b は、それまでのバックワードポインタが指していたアドレスの値。に書き換えて、ここをバックワードポインタが指すようにする。

これを図 2.3(a) および (b) に示します。



Figure 2.3: Marking a graph using pointer reversal

この図では、マークは「Visits 1」の場合は「V1」、「Visits 2」の場合は「V2」、「Done」の場合は「D」と省略されています。逆方向のポインタのチェーンが、アクセスされたアプリケーションノードに構築される方法に注意してください。

次のルールは、NMarked Done コンストラクタを使用して、マークされていないNPrim ノードを完了としてマークする必要があることを示しています (図 2.3(c))。

$$\begin{array}{|c|} \hline (x.xx) \\ \hline \end{array} \quad \begin{array}{c} f \quad b \quad h[f : \text{NPrim } p] \\ \Rightarrow \quad f \quad b \quad h[f : \text{NMarked Done (NPrim } p)] \end{array}$$

NSupercomb ノードと NNum ノードは、それ以上のアドレスを含まないため、同様に扱われます。

マークされていないノードについてはこれだけです。マシンは、 $f$  がマークされたノードを指していることを検出すると、 $b$  が指しているノードを検査します。hNull の場合、マシンは終了します。( フォワードポインタが指しているアドレスの中身が NMarked Done  $n$  でバックワードポインタが hNull の場合ということ?) それ以外の場合は、マークされた NAp ノードである必要があります。まず、状態情報が (Visits 1) の場合、ノードが 1 回訪問されたとしましょう。したがって、NAp ノードの最初のサブグラフのマーク付けが完了し、2 番目のサブグラフをマークする必要があります。これは、 $f$  をポイントにして、 $b$  を変更せずに、ノード ( $b'$ ) に保存されているバックポインタを最初のフィールドを 2 番目のフィールドに変更し、状態情報を変更します (図 2.3(d))。

$$\begin{array}{|c|} \hline (x.xx) \\ \hline \end{array} \quad \begin{array}{c} f \quad b \quad h \left[ \begin{array}{l} f : \text{NMarked Done } n \\ b : \text{NMarked (Visits 1) (NAp } b' a_2) \end{array} \right] \\ \Rightarrow \quad a_2 \quad b \quad h[b : \text{NMarked (Visits 2) (NAp } f b')] \end{array}$$

しばらくして、マシンは 2 番目のサブグラフのマーキングを完了します。この場合、ノードを元の形式に復元し、バックポインタのチェーンを 1 段階バックアップできます (図 2.3(e))。

$$\begin{array}{|c|} \hline (x.xx) \\ \hline \end{array} \quad \begin{array}{c} f \quad b \quad h \left[ \begin{array}{l} f : \text{NMarked Done } n \\ b : \text{NMarked (Visits 2) (NAp } a_1 b') \end{array} \right] \\ \Rightarrow \quad b \quad b' \quad h[b : \text{NMarked Done (NAp } a_1 f)] \end{array}$$

最後に、間接処理を扱います。 $f$  を変更することでスキップされますが、 $b$  は変更されません。ヒープは変更されないままなので、間接参照自体はマークされません。したがって、ガベージコレクションが完了すると、すべてのインダイレクションが再利用されます。ご覧のとおり、このマーキングアルゴリズムを使用すると、ガベージコレクション中に間接参照を「ショートアウト」するのは非常に簡単です。

(x.xx)	$f \quad b \quad h[f : \text{NInd } a]$
	$\Rightarrow \quad a \quad b \quad h$

これで、ポインタ反転アルゴリズムの状態遷移が完了しました。

演習 2.34 NData ノードのルールを追加します。

演習 2.35 アルゴリズムを実装します。必要な主な変更は、`node` 型と `markFrom` 関数です。NMarked コンストラクタの形式が変更されたため、`scan` は簡単な方法で変更する必要があります。

#### 2.9.4 2 スペースガベージコレクション

ガベージコレクションを実行するもう 1 つの非常に一般的な方法は、すべてのライブデータを 1 つのヒープから別のヒープにコピーすることです。これは、[Fenichel and Yochelson 1969] によって発明されたいわゆる 2 スペースコレクタです ([Baker 1978, Cheney 1970] も参照)。コレクタは 2 つの段階で機能します。

1. マシンの状態 (スタック、ダンプなど) が指すすべてのノードは、古いヒープ (*from-space* と呼ばれる) から最初は空の新しいヒープ (*to-space* と呼ばれる) に退避されます。ノードは、そのコピーを *to-space* に割り当て、*from-space* コピーを、新しいノードの *to-space* アドレスを含む *forwarding pointer* で上書きすることによって退避されます。`markFrom` と同様に、退避ルーチンは新しいノードの *to-space* アドレスを返します。これは、マシン状態の古いアドレスを置き換えるために使用されます。
2. 次に、*to-space* 内のすべてのノードが最初から線形にスキャンされ、それぞれがスカベンジされます。ノード  $n$  は、それが指すノードを退避させ、 $n$  内のそれらのアドレスを新しい *to-space* アドレスに置き換えることによってスカベンジされます。スキャンポインタが *allocation pointer* に追いつくと、スキャンが停止します。

これを実装するには、`node` 型のさらに別のバリエーションを追加する必要があります。今回は、単一のアドレス (*to-space* アドレス) を含む `NForward` コンストラクタを使用します。(このコレクターには `NMarked` は必要ありません。) `markFromStack` の代わりに、次のタイプの `evacuateStack` が必要です。

```
evacuateStack :: TiHeap -> TiHeap -> TiStack -> (TiHeap, TiStack)
```



呼び出し (`evacuateStack fromheap toheap stk`) は、*stk* から *toheap* に参照される *fromheap* 内のすべてのノードを退避させ、新しい *toheap* と新しい *stk* を返します。ダンプとグローバルにも同様の関数が必要です。

最後に、関数が必要です。

```
scavengeHeap :: TiHeap-> TiHeap-> TiHeap
```

ここで、呼び出し (`scavengeHeap fromheap toheap`) は、*toheap* 内のノードをスカベンジし、必要に応じて *fromheap* から *toheap* にノードを退避させます。

演習 2.36 このガベージコレクターを実装します。

```
module GM where
import Language
import Utils
```

### 3 G マシン

この章では、最初のコンパイラベースの実装である G マシンを紹介します。これは、スウェーデンのヨーテボリにあるチャルマース工科大学で、オーガストソンとジョンソンによって開発されました。この章の内容は、彼らの一連の論文 [Augustsson 1984、Johnsson 1984] に基づいており、博士論文 [Augustsson 1987、Johnsson1987] で最高潮に達します。

#### 3.1 G マシンの紹介

テンプレートのインスタンス化マシンの基本的な操作は、`instantiate` 関数によって実装されるスーパーコンビネータ本体のインスタンスを構築することでした。インスタンス化が実行されるたびに、インスタンス化はテンプレートを再帰的にトラバースする必要があるため、これはかなり遅い操作です。`instantiate` によって実行される機械語命令について考えると、テンプレートのトラバースに関係するものと、実際にインスタンスを構築することに関係するものの 2 種類があることがわかります。

G マシンおよびその他のコンパイルされた実装の「大きなアイデア」は次のとおりです。

プログラムを実行する前に、各スーパーコンビネータ本体を一連の命令に変換し、実行時にスーパーコンビネータ本体のインスタンスを構築します。

このコードを実行すると、インスタンス化関数を呼び出すよりも高速になります。これは、すべての命令がインスタンスの構築に関係しているためです。テンプレートをトラバースするために必要な指示はありません。これは、すべてが変換プロセス中に行われたためです。したがって、プログラムの実行は 2 つの段階に分けられます。最初の段階では、コンパイラを使用してプログラムの中間形式を生成します。これはコンパイルタイムと呼ばれます。第 2 段階では、中間形式が実行されます。これはランタイムと呼ばれます。

スーパーコンビネータに対して行うことは、それをインスタンス化することだけなので、変換が完了したら元のスーパーコンビネータを破棄して、コンパイルされたコードのみを保持できます。

原則として、G マシンコンパイラを使用して、ソース言語のプログラムを一連の機械語命令に変換します。多くの異なるハードウェア (68000 ベース、または VAX など) に言語を実装したい場合があるため、抽象マシンがあると便利です。優れた抽象マシンには 2 つの特性があります。まず、具体的な機

機械語コード (たとえば 68000 アセンブラ) に簡単に変換できます。2 つ目は、ソースから抽象的な機械語コードを簡単に生成できることです。

ここでトレードオフに直面していることに注意してください。抽象マシンを実マシンと同じにすることで、理想的には第 1 の性質 (具体的なコード生成の容易さ) を満たすことができます。しかし、これにより、2 番目のプロパティを満たすことがはるかに難しくなります。したがって、抽象マシンは、ソース言語と特定の機械語コードの間の足がかりです。

### 3.1.1 例

G マシンコンパイラが動作している小さな例を次に示します。次の関数を考えます。

$$f\ g\ x = K\ (g\ x)$$

これは、一連の G コード命令にコンパイルされます。

```
Push 1
Push 1
Mkap
Pushglobal K
Mkap
Slide 3
Unwind
```

図 3.1 に、このコードがどのように実行されるかを示します。各ダイアグラムの左側にはスタックがあり、下に向かって成長します。各図の残りの部分はヒープです。アプリケーション ノードは @ 文字で表され、式は小文字でラベル付けされ、スーパーコンビネータは大文字でラベル付けされます。

図 3.1 の図 (a) では、 $f$  の一連の命令を実行する前のマシンの状態を示しています。テンプレート具体化マシンと同じように、スパインがほどかれています。スタックの上位 2 つの項目は関数適用ノードへのポインタであり、その右側の部分は  $g$  と  $x$  にバインドされる式です。

Push 命令は、スタックのトップに相対的なアドレス指定を使用します。スーパーコンビネータノード  $f$  へのポインタを無視すると、最初のスタック アイテムは 0、次は 1 というように番号が付けられます。次の図 (b) は、Push 1

命令を実行した後の変更されたスタックを示しています。これにより、式  $x$  へのポインターがスタックにプッシュされます。 $x$  は、スタックの 2 つ下のスタック アイテムです。

別の `Push 1` の後、スタックの一番上に  $g$  へのポインターがあります。前の命令が新しいポインターをスタックにプッシュしたため、これはスタックの 2 つ下のスタックアイテムです。新しい図は (c) です。

図 (d) は、`Mkap` 命令が実行されるとどうなるかを示しています。スタックから 2 つのポインターを取得し、それらから関数適用ノードを作成します。スタック上の結果へのポインターを残します。

図 (e) では、`Pushglobal K` 命令を実行し、 $K$  スーパーコンビネータへのポインターをプッシュします。

図 (f) に示すように、別の `Mkap` 命令が  $f$  の本体のインスタンス化を完了します。

元の式  $f\ g\ x$  を新しくインスタンス化されたボディ  $K\ (g\ x)$  に置き換えることができます。G マシンの最初のバージョン (これは *lazy* ではありません) では、本体をスタックの 3 つの場所にスライドさせ、そこにあった 3 つのポインターを破棄します。これは、図 (g) に示すように、`Slide 3` の指示を使用して実現されます。最終 `Unwind` 命令により、マシンは評価を続行します。



Figure 3.1: Execution of code for the  $f$  supercombinator

これで、G マシンの実行の概要を簡単に説明しました。

### 3.1.2 さらに最適化

これまでに説明したように、テンプレートをトラバースする際の解釈のオーバーヘッドをなくすことで、適度なパフォーマンスの向上を実現できます。ただし、コンパイルは、テンプレートインスタンス化マシンでは利用できない多くのショートカットと最適化への扉を開くことも判明しています。たとえば、次の定義を考えてみましょう。

$f\ x = x + x$

テンプレートインスタンス化マシンは  $x$  を 2 回評価します。2 回目の場合はもちろん、すでに評価されていることがわかります。コンパイルされた実装は、コンパイル時に  $x$  が既に評価されていることを発見し、評価ステップを省略できます。

### 3.2 テンプレートを構築するためのコード シーケンス

テンプレートインスタンス化マシンは次のように動作することを思い出してください。

- スタックの一番上にある単一のアイテムが整数へのポインタである場合、マシンは終了しました。
- そうでない場合は、スーパーコンビネータノードに到達するまで、遭遇した関数適用ノードを巻き戻します。次に、スーパーコンビネータ本体のコピーをインスタンス化し、その引数を置換します。

Mark 1 テンプレートインスタンス化マシンの中心にあるのは、59 ページと 60 ページで定義されている 2 つの関数 `scStep` と `instanceiate` です。`scStep` と `instantiate` の定義を見てみると、スーパーコンビネータをインスタンス化する操作について次のように説明できます。

1. 変数名のローカル環境をヒープ内のアドレスに構築します。
2. このローカル環境を使用して、ヒープ内にスーパーコンビネータ本体のインスタンスを作成します。変数はコピーされません。代わりに、対応するアドレスが使用されます。
3. 関数適用ノードとスーパーコンビネータノードへのポインタをスタックから削除します。
4. スーパーコンビネータの新しく作成されたインスタンスのアドレスをスタックにプッシュします。

テンプレートインスタンス化では、スーパーコンビネータのインスタンスを作成するには、スーパーコンビネータの本体である式のツリー構造をトラバースする必要があります。式は再帰的に定義されるため、ツリートラバース関数 `instantiate` は再帰的に定義されます。たとえば、`EAp e1 e2` の場合の `instantiate` の定義 (58 ページ) を参照してください。最初に `e1` に対して、次に `e2` に対して `instanceiate` を呼び出し、各サブ式のグラフのアドレスを保持します。最後に、グラフに関数適用ノードを作成して、2 つのアドレスを結合します。

命令の線形シーケンスをコンパイルして、式をインスタンス化する操作を実行したいと考えています。

### 3.2.1 算術の後置評価

式をインスタンス化する命令の線形シーケンスを構築したいという欲求は、算術式の後置評価を連想させます。G マシンに戻る前に、このアナロジーをさらに調べます。

算術式の言語は、数、加算、および乗算で構成されます。この言語を `AExpr` 型として表すことができます。

```
data AExpr = Num Int
           | Plus AExpr AExpr
           | Mult AExpr AExpr
```

言語は「明白な」意味を持つべきであることが意図されています。関数 `aInterpret` を使用してこれを与えることができます。

```
aInterpret :: AExpr -> Int
aInterpret (Num n)      = n
aInterpret (Plus e1 e2) = aInterpret e1 + aInterpret e2
aInterpret (Mult e1 e2) = aInterpret e1 * aInterpret e2
```

あるいは、式を演算子 (または命令) の後置シーケンスにコンパイルすることもできます。式を評価するには、コンパイルされた演算子と値のスタックを使用します。たとえば、算術式  $2 + 3 \times 4$  は、以下のシーケンスとして表されます。

[INum 2, INum 3, INum 4, IMult, IPlus]

後置マシンの命令は、`aInstruction` 型として指定できます。

```
data AInstruction = INum Int
                  | IPlus
                  | IMult
```

評価器の状態は、一連の演算子と数値のスタックであるペアです。コードシーケンスの意味は、次の遷移規則で与えられます。

(3.1)	$\Rightarrow$	<div style="display: flex; justify-content: space-between; align-items: center;"> <div style="border: 1px solid black; width: 20px; height: 20px; margin: 0 auto;"></div> <div style="margin-left: 10px;"> <math>[n]</math>  <math>n</math> </div> </div>
-------	---------------	---

(3.2)	$\Rightarrow$	<div style="display: flex; justify-content: space-between; align-items: center;"> <div style="margin-left: 10px;"> <math>\text{INum } n : i</math>  <math>i</math> </div> <div style="margin-left: 10px;"> <math>ns</math>  <math>n : ns</math> </div> </div>
-------	---------------	---



(3.3)	$\begin{array}{c} \text{IPlus } n : i \quad n_0 : n_1 : ns \\ \Rightarrow \quad i \quad (n_1 + n_0) : ns \end{array}$
-------	---

(3.4)	$\begin{array}{c} \text{IMult } n : i \quad n_0 : n_1 : ns \\ \Rightarrow \quad i \quad (n_1 \times n_0) : ns \end{array}$
-------	--

これらの遷移規則を Miranda に翻訳すると、次のようになります。

```
aEval :: ([AInstruction], [Int]) -> Int
aEval ([], [n]) = n
-- 遷移規則 (3.1)
aEval (INum n : is, s) = aEval (is, n : s)
-- 遷移規則 (3.2)
aEval (IPlus : is, n0 : n1 : s) = aEval (is, n1+n0 : s)
-- 遷移規則 (3.3)
aEval (IMult : is, n0 : n1 : s) = aEval (is, n1*n0 : s)
-- 遷移規則 (3.4)
```

式の後置コードのシーケンスを生成するには、コンパイラを定義する必要があります。これは式を取り、実行時に式の値を計算する一連の命令を配信します。

```
aCompile :: AExpr -> [AInstruction]
aCompile (Num n)      = [INum n]
aCompile (Plus e1 e2) = aCompile e1 ++ aCompile e2 ++ [IPlus]
aCompile (Mult e1 e2) = aCompile e1 ++ aCompile e2 ++ [IMult]
```

これからの重要なアイデアは、aCompile 関数の型によって与えられます。命令のリストを返します。

式の後置表現は、式ツリーをフラット化または線形化する方法であり、式をフラットな一連の演算子で表すことができます。

**演習 3.1.** 構造帰納法を使用するか、またはその他の方法で、算術式の後置評価が式のツリー評価と同じ結果になることを証明します。つまり、型 aExpr のすべての式 e に対して、

$$\text{aInterpret } e = \text{aEval } (\text{aCompile } e, [])$$

これは一致証明の例です。

演習 3.2. 関数 `aInterpret`、`aCompile`、および `aEval` を拡張して、`let` 式を処理します。`aExpr` 型の `e` のすべての式について、これらの新しい関数が次の関係を満たすことを証明します。

$$\text{aInterpret } e = \text{aEval } (\text{aCompile } e, [])$$

`letrec` 式など、さらに複雑な式に言語を拡張できますか？これらの拡張機能を正しく実装したことを証明できますか？

### 3.2.2 後置コードを使用してグラフを作成する

同じテクニックを使用して、スーパーコンピネータ本体のインスタンスを作成できます。この場合、スタック上の「値」は、インスタンス化される式の部分のアドレスになります。

テンプレート構築命令の操作は、上記の算術例で見たものとは異なります。通常、命令にはヒープにノードを割り当てるという副作用があります。例として、`Mkap` 命令の導入を検討してください。この命令は、スタックの上位 2 つのアドレスからヒープ内に関数適用ノードを作成します。完了すると、この新しいノードへのポインタがスタックに残されます。

テンプレートインスタンス化マシンには既にそのようなスタックがあるため、アドレスの新しい評価スタックを創案する理由はありません。ただし、このスタックを使用する場合は、覚えておくべき重要な点があります。

スタックからオブジェクトをポップおよびプッシュすると、変数名に対応するスタックの場所のマップが変化します。したがって、式をコンパイルするときは、これを追跡する必要があります。

スタック内のアイテムへのアクセスは、スタックの一番上を基準にしています。そのため、アイテムが追加されると、そのアイテムに到達するまでのオフセットが 1 増加します。同様に、アイテムがポップされると、オフセットは 1 だけ減少します。

### 3.2.3 インスタンス化が行われた後はどうなりますか？

スーパーコンピネータ本体のインスタンス化が完了したら、スタックを整理し、評価プロセスの継続を手配する必要があります。 $n$  個の引数を持つスーパーコンピネータの後置シーケンスの評価が完了すると、スタックは次の形式になります。

- 一番上には、新しくインスタンス化された本体  $e$  のヒープ内のアドレスがあります。
- 次に、 $n + 1$  個のポインタがあります。これらから、インスタンス化プロセスで使用される引数にアクセスできます。
- $n + 1$  個のポインタの最後は、インスタンス化したばかりの式のルートを指します。

これを図 3.2 に示します。




Figure 3.2: The stack layout for the Mark 1 machine

`redex` を新しくインスタンス化された本体に置き換え、`Slide` 命令を使用して  $n$  個のアイテムをスタックからポップする必要があります。次のスーパーコンピネータを見つけるには、`Unwind` 命令を使用して、再び巻き戻しを開始する必要があります。後置演算子シーケンスに整理と巻き戻しを行う操作を追加することで、テンプレートインスタンス化マシンを Mark 1 G マシンに変換しました。

関数  $f \ x_1 \ \dots \ x_n = e$  のコードは次のとおりです。

```
<code to construct an instance of e>
Slide n+1
Unwind
```

### 3.3 Mark 1: 最小限の G マシン

ここで、完全な G マシンとそのコンパイラのコードを示します。更新 (セクション 3.4 で紹介) や算術演算 (セクション 3.6 で紹介) は実行しません。

#### 3.3.1 全体構造

最上位では、G-machine はテンプレート インスタンスierer に非常に似ています。いつものように、システム全体が実行機能と一緒に編まれています。

```
-- The function run is already defined in gofers standard.prelude
runProg :: [Char] -> [Char]
runProg = showResults . eval . compile . parse
```

アクセスする必要があるため、パーサのデータ構造と関数が含まれています。

```
-- :a language.lhs -- parser data types
```

#### 3.3.2 データ型の定義

グラフ簡約実装手法の基本はグラフです。とりわけ、Appendix A で提供されているユーティリティの heap データ型を使用します。

```
-- :a util.lhs -- heap data type and other library functions
```

Mark 1 G マシンは、状態として 5 タプル gmState を使用します。gmState は、コンパイルされたプログラムの実行中に必要なすべての情報を保持します。

```
type GmState
  = (GmCode,      -- Current instruction stream
     GmStack,     -- Current stack
     GmHeap,      -- Heap of nodes
     GmGlobals,   -- Global addresses in heap
     GmStats)     -- Statistics
```

G マシンの説明では、ステート アクセス関数を使用して、ステートのコンポーネントにアクセスします。このアプローチの利点は、新しいコンポーネントに対応するために状態を変更するときに、作成した元のコードのほとんどを再利用できることです。接頭辞 get を使用して状態からコンポーネントを取得するアクセス関数を示し、接頭辞 put を使用して状態内のコンポーネントを置き換えます。

状態の 5 つのコンポーネントのそれぞれの型定義と、それらのアクセス関数を順に検討します。

- 命令ストリームは `GmCode` 型で、単なる `instruction` のリストです。

```
type GmCode = [Instruction]
```

コードへの便利なアクセスを取得するために、後で状態が追加のコンポーネントで拡張されるときに、`getCode` と `putCode` の 2 つの関数を定義します。

```
getCode :: GmState -> GmCode
getCode (i, stack, heap, globals, stats) = i

putCode :: GmCode -> GmState -> GmState
putCode i' (i, stack, heap, globals, stats)
    = (i', stack, heap, globals, stats)
```

最初は 6 つの命令しかありません。これらについては、サブセクション 3.3.3 で詳しく説明します。

```
data Instruction
    = Unwind
    | Pushglobal Name
    | Pushint Int
    | Push Int
    | Mkap
    | Slide Int
instance Eq Instruction
    where
        Unwind == Unwind = True
        Pushglobal a == Pushglobal b = a == b
        Pushint a == Pushint b = a == b
        Push a == Push b = a == b
        Mkap == Mkap = True
        Slide a == Slide b = a == b
        _ == _ = False
```

- G マシン スタック gmStack は、ヒープ内のアドレスのリストです。

```
type GmStack = [Addr]
```

スタックへの便利なアクセスを取得するために、後で状態が追加のコンポーネントで拡張されたときに、getStack と putStack の 2 つの関数を定義します。

```
getStack :: GmState -> GmStack
getStack (i, stack, heap, globals, stats) = stack

putStack :: GmStack -> GmState -> GmState
putStack stack' (i, stack, heap, globals, stats)
    = (i, stack', heap, globals, stats)
```

- テンプレート インスタシエーターの場合と同様に、utils のヒープデータ構造を使用してヒープを実装します。

```
type GmHeap = Heap Node
```

繰り返しますが、状態のこのコンポーネントにアクセスするには、アクセス関数を定義します。

```
getHeap :: GmState -> GmHeap
getHeap (i, stack, heap, globals, stats) = heap
putHeap :: GmHeap -> GmState -> GmState
putHeap heap' (i, stack, heap, globals, stats)
    = (i, stack, heap', globals, stats)
```

最小限の G マシンには、次の 3 種類のノードしかありません。数 NNum、関数適用 NAp、そしてグローバル NGlobal です。

```
data Node
    = NNum      Int          -- Numbers
    | NAp       Addr Addr    -- Applications
    | NGlobal   Int  GmCode   -- Globals
```

数 ノードには関連する番号が含まれます。関数適用 ノードは、最初のアドレスの関数を 2 番目のアドレスの式に適用します。NGlobal ノードには、グローバルが期待する引数の数と、そのときに実行されるコード シーケンスが含まれています。グローバルには十分な引数があります。これは、アリティとコードの代わりにテンプレートを保持していたテンプレート インスタンスエーターの NSupercomb ノードを置き換えます。

- 後で遅延実装を作成するため、各グローバルに対して 1 つのノードのみが存在することが重要です。グローバルのアドレスは、連想リスト gmGlobals でその値を検索することによって決定できます。これは、テンプレート マシンの tiGlobals コンポーネントに対応します。

```
type GmGlobals = ASSOC Name Addr
```

使用するアクセス関数は getGlobals です。Mark 1 マシンでは、このコンポーネントは一定であるため、対応する put 関数は必要ありません。

```
getGlobals :: GmState -> GmGlobals
getGlobals (i, stack, heap, globals, stats) = globals
```

- 状態の統計コンポーネントは、抽象データ型として実装されます。

```
statInitial  :: GmStats
statIncSteps :: GmStats -> GmStats
statGetSteps :: GmStats -> Int
```

gmStats の実装が提供されるようになりました。

```
type GmStats = Int
statInitial  = 0
statIncSteps s = s + 1
statGetSteps s = s
```

このコンポーネントにアクセスするには、getStats と putStats を定義します。

```

getStats :: GmState -> GmStats
getStats (i, stack, heap, globals, stats) = stats
putStats :: GmStats -> GmState -> GmState
putStats stats' (i, stack, heap, globals, stats)
    = (i, stack, heap, globals, stats')

```

### 3.3.3 評価器

G マシン評価器 `eval` は、状態のリストを生成するように定義されています。最初のは、コンパイラによって構築されたものです。最後の状態がある場合、評価の結果は最後の状態のスタック コンポーネントの一番上になります。

```

eval :: GmState -> [GmState]
eval state = state: restStates
    where
        restStates | gmFinal state = []
                   | otherwise = eval nextState
        nextState = doAdmin (step state)

```

関数 `doAdmin` は `statIncSteps` を使用して、状態の統計コンポーネントを変更します。

```

doAdmin :: GmState -> GmState
doAdmin s = putStats (statIncSteps (getStats s)) s

```

評価器の重要な部分は、関数 `gmFinal` とこれから見ていく `step` です。

#### 最終状態のテスト

実行中のコード シーケンスが空の場合、G マシン インタープリターは終了しています。この状態を `gmFinal` 関数で表現します。

```

gmFinal :: GmState -> Bool
gmFinal s = case (getCode s) of
    []      -> True
    otherwise -> False

```

#### ステップを進める

`step` 関数は、実行中の命令に基づいて状態遷移を行うように定義されています。



```

step :: GmState -> GmState
step state = dispatch i (putCode is state)
              where (i : is) = getCode state

```

現在の命令  $i$  を `dispatch` し、現在のコードシーケンスをコードシーケンス  $is$  に置き換えます。これは、実機でプログラムカウンタを進めることに相当します。

```

dispatch :: Instruction -> GmState -> GmState
dispatch (Pushglobal f) = pushglobal f
dispatch (Pushint n) = pushint n
dispatch Mkap           = mkap
dispatch (Push n)       = push n
dispatch (Slide n)      = slide n
dispatch Unwind         = unwind

```

ご覧のとおり、`dispatch` 関数は実行する状態遷移を選択するだけです。

後置命令の遷移規則を調べることから始めましょう。`instruction` 内の構文オブジェクトごとに 1 つ存在します。まず、`Pushglobal` 命令から始めます。この命令は、状態のグローバルコンポーネントを使用して、グローバル  $f$  を保持する `heap` 内の一意の `NGlobal` ノードを見つけます。見つからない場合は、適切なエラーメッセージが出力されます。

(3.5)	$\text{Pushglobal } f : i \quad s \quad h \quad m[f : a] \Rightarrow i \quad a : s \quad h \quad m$
-------	---

`pushglobal` 関数を使用してこのルールを実装します。

```

pushglobal :: Name -> GmState -> GmState
pushglobal f state
  = putStack (a: getStack state) state
    where a = aLookup (getGlobals state) f (error ("Undeclared global " ++ f))

```

残りの遷移は、スーパーコンビネータの本体を構築するためのものです。`Pushint` の遷移は、整数ノードをヒープに配置します。

(3.6)	$\text{PushInt } n : i \quad s \quad h \quad m \Rightarrow i \quad a : s \quad h[a : \text{NNum } n] \quad m$
-------	---

対応する関数は `pushint` です。数は、アドレス  $a$  の新しいヒープ `heap'` に配置されます。次に、ヒープとスタックを元の状態に戻します。

```

pushint :: Int -> GmState -> GmState
pushint n state
  = putHeap heap' (putStack (a: getStack state) state)
    where (heap', a) = hAlloc (getHeap state) (NNum n)

```

Mkap 命令は、スタックの最上部にある 2 つのアドレスを使用して、ヒープ内に関数適用ノードを構築します。次の遷移規則があります。

(3.7)	$\text{Mkap } a_1 : a_2 : i \quad s \quad h \quad m$ $\Longrightarrow \quad i \quad a : s \quad h[a : \text{NApp } a_1 \ a_2] \quad m$
-------	--

このトランジションが mkap になります。ここでも、heap' と a は、それぞれ新しいヒープと新しいノードのアドレスです。

```

mkap :: GmState -> GmState
mkap state
  = putHeap heap' (putStack (a:as') state)
    where (heap', a) = hAlloc (getHeap state) (NApp a1 a2)
          (a1:a2:as') = getStack state

```

Push 命令は、関数に渡された引数のコピーを取得するために使用されます。これを行うには、スタックからポイントされている関数適用ノードを「調べる」必要があります。また、スタック上にあるスーパーコンビネータ ノードをスキップすることも忘れないでください。

(3.8)	$\text{Push } n : i \quad a_0 : \dots : a_{n+1} : s \quad h[a_{n+1} : \text{NApp } a_n \ a'_n] \quad m$ $\Longrightarrow \quad i \quad a'_n : a_0 : \dots : a_{n+1} : s \quad h \quad m$
-------	--

```

push :: Int -> GmState -> GmState
push n state
  = putStack (a:as) state
    where as = getStack state
          a = getArg (hLookup (getHeap state) (as !! (n+1)))

```

これは、補助関数 getArg を使用して、アプリケーション ノードから必要な式を選択します。

```

getArg :: Node -> Addr
getArg (NApp a1 a2) = a2

```

スタック構造のため、Push 命令のアドレッシング モードを [Peyton Jones 1987] で使用されていたものから変更しました。

次に、スーパーコンビネータがインスタンス化された後、アンwindを  
 続行する前に発生するスタックの整理が、Slide 命令によって実行されます。

(3.9)	$\text{Slide } n : i \quad a_0 : \dots : a_n : s \quad h \quad m$ $\Rightarrow \quad i \quad a_0 : s \quad h \quad m$
-------	---

```
slide :: Int -> GmState -> GmState
slide n state
= putStack (a: drop n as) state
  where (a:as) = getStack state
```

Unwind は、テンプレート インスタンス化子の外側のループを置き換える  
 ため、最も複雑な命令です。Unwind 命令は、次のセクションで説明するよう  
 に、常にシーケンスの最後の命令です。構築された newState は、スタック  
 の一番上の項目に依存します。これは、スタックの一番上にあるアイテムに  
 も依存する、最初の遷移ルールに依存します。

```
unwind :: GmState -> GmState
unwind state
= newState (hLookup heap a)
  where
    (a:as) = getStack state
    heap = getHeap state
```

まず、スタックの一番上に数値がある場合を考えます。この場合、これで  
 終了です。G マシンが終了したことを示すために、コード コンポーネント  
 に [] を配置します。

(3.10)	$[\text{Unwind}] \quad a : s \quad h[a : \text{NNum } n] \quad m$ $\Rightarrow \quad [] \quad a : s \quad h \quad m$
--------	--

```
newState (NNum n) = state
```

スタックの一番上に関数適用ノードがある場合は、次のノードからアンw  
 indを続行する必要があります。

(3.11)	$[\text{Unwind}] \quad a : s \quad h[a : \text{NApp } a_1 a_2] \quad m$ $\Rightarrow \quad [\text{Unwind}] \quad a_1 : a : s \quad h \quad m$
--------	---

```
newState (NApp a1 a2) = putCode [Unwind] (putStack (a1:a:as) state)
```

最も複雑なルールは、スタックの一番上にグローバル ノードがある場合に  
 発生します。スーパーコンビネータの適用を減らすのに十分な引数があるか  
 どうかに応じて、考慮すべき 2 つのケースがあります。

第 1 に、スーパーコンビネータ アプリケーションを削減するのに十分な引数がない場合、プログラムの型が正しくありません。Mark 1 G マシン については、このケースを無視します。あるいは、十分な引数がある場合、スーパーコンビネータのコードに「ジャンプ」することで、スーパーコンビネータを減らすことができます。遷移規則では、これはスーパーコンビネータ コードをマシンのコード コンポーネントに移動することによって表現されます。

$SC[d]$ は、スーパーコンビネータ定義 $d$ の G マシン コードです。	
$SC[f\ x_1 \dots x_n = e] = \mathcal{R}[e]\ [x_1 \mapsto 0, \dots, x_n \mapsto n-1]\ n$	
$\mathcal{R}[e]\ \rho\ d$ は、アリティ $d$ のスーパーコンビネータのために、環境 $\rho$ で式 $e$ をインスタンス化するコードを生成し、結果のスタックの巻き戻しに進みます。	
$\mathcal{R}[e]\ \rho\ d = C\rho++[\text{Slide } d+1, \text{Unwind}]$	
$\mathcal{C}[e]\ \rho$ は、環境 $\rho$ で $e$ のグラフを構築するコードを生成し、それへのポインターをスタックの一番上に残します。	
$\mathcal{C}[f]\ \rho = [\text{Pushglobal } f]$	ここで、 $f$ はスーパーコンビネータです。
$\mathcal{C}[x]\ \rho = [\text{Push } (\rho\ x)]$	ここで、 $x$ はローカル変数です。
$\mathcal{C}[i]\ \rho = [\text{Pushint } i]$	
$\mathcal{C}[e_0\ e_1]\ \rho = \mathcal{C}[e_1]\ \rho ++ \mathcal{C}[e_0]\ \rho^{+1} ++ [\text{Mkap}]$	ここで、 $\rho^{+n}\ x = (\rho\ x) + n$ です。

### 3.3.4 プログラムのコンパイル

一連のコンパイル スキームを使用してコンパイラを説明します。各スーパーコンビネータの定義は、コンパイル スキーム  $SC$  を使用してコンパイルされます。各スーパーコンビネータ用に生成されたコンパイル済みコードは、図 3.3 に定義されています。コンパイル スキーム  $SC$ 、 $\mathcal{R}$ 、および  $\mathcal{C}$  に対応するのは、コンパイラ関数 `compileSc`、`compileR`、および `compileC` です。これらのそれぞれを順番に検討します。

`compile` 関数は、プログラムを G マシンの初期状態にします。最初のコードシーケンスは、グローバルな `main` を見つけて評価します。ヒープは、宣言された各グローバルのノードが含まれるように初期化されます。`globals` には、グローバル名からそれらに提供される `NGlobal` ノードへのマップが含まれます。

```
compile :: CoreProgram -> GmState
compile program
  = (initialCode, [], heap, globals, statInitial)
  where (heap, globals) = buildInitialHeap program
```

初期ヒープを構築し、定義された各グローバルのグローバル ノードのマッピングを提供するために、`buildInitialHeap` を使用します。これは、テンプレート マシンの場合とまったく同じです。

```
buildInitialHeap :: CoreProgram -> (GmHeap, GmGlobals)
buildInitialHeap program
  = mapAccum1 allocateSc hInitial compiled
    where compiled = map compileSc (preludeDefs ++ program) ++
                                compiledPrimitives
    --where
    --   compiled = map compileSc program
```

`buildInitialHeap` 関数は `mapAccum1` を使用して、コンパイルされた各グローバルにノードを割り当てます。型 `[gmCompiledSC]` を持つコンパイル済みで (必要に応じて) 発生するコンパイル。

```
type GmCompiledSC = (Name, Int, GmCode)
```

関数 `allocateSc` は、コンパイルされたスーパーコンビネータ引数に新しいグローバルを割り当て、新しいヒープとグローバルが格納されているアドレスを返します。

```
allocateSc :: GmHeap -> GmCompiledSC -> (GmHeap, (Name, Addr))
allocateSc heap (name, nargs, instns)
  = (heap', (name, addr))
    where (heap', addr) = hAlloc heap (NGlobal nargs instns)
```

初期状態では、マシンにプログラムの値を評価してもらいたい。これは単にグローバル `main` の値であることを思い出してください。

```
initialCode :: GmCode
initialCode = [Pushglobal "main", Unwind]
```

各スーパーコンビネータは、図 3.3 の *SC* スキームを実装する `compileSc` を使用してコンパイルされます。これは、スーパーコンビネータ名、スーパーコンビネータが簡約される前に必要な引数の数、およびスーパーコンビネータに関連付けられたコード シーケンスを含むトリプルを返します。

```
compileSc :: (Name, [Name], CoreExpr) -> GmCompiledSC -- SC スキーム
compileSc (name, env, body)
  = (name, length env, compileR body (zip2 env [0..]))
```

これは、図 3.3 の *R* スキームに対応する `compileR` を使用します。

```

compileR :: GmCompiler -- R スキーム
compileR e env = compileC e env ++ [Slide (length env + 1), Unwind]
--compileR e env = compileC e env ++ [Slide (length env + 1), Unwind]

```

各コンパイラ スキームには、同じ型 `gmCompiler` があります。

```

type GmCompiler = CoreExpr -> GmEnvironment -> GmCode

```

コンパイルスキームから写像  $\rho$  を連想リストとして表すことができるという事実を使用します。このリストから変数のオフセットを検索できるだけでなく、スタックにある引数の数を計算することもできます。これは、`Slide` 命令で絞り出すスタック要素の数を調べるために `compileR` で使用されます。リストのタイプは `gmEnvironment` で、次のように定義されています。

```

type GmEnvironment = ASSOC Name Int

```

これは、図 3.3 の `C` スキームに対応する `compileC` を使用して、スーパーコンビネータ本体のインスタンス化を構築します。

```

compileC :: GmCompiler -- C スキーム
compileC (EVar v) env
  | elem v (aDomain env) = [Push n]
  | otherwise = [Pushglobal v]
  where n = aLookup env v (error "Can't happen")
compileC (ENum n) env = [Pushint n]
compileC (EAp e1 e2) env = compileC e2 env ++ compileC e1 (argOffset 1 env) ++ [Mkap]

```

関数 `argOffset` を使用して、スタック オフセットを変更できます。env が  $\rho$  を実装する場合、`(argOffset n env)` は  $\rho^{+n}$  を実装します。

```

argOffset :: Int -> GmEnvironment -> GmEnvironment
argOffset n env = [(v, n + m) | (v, m) <- env]

```

## コンパイル例

K コンビネータのコンパイルを見てみましょう。この関数をコンパイルするとき、次の式を評価することから始めます。

```

compileSc ("K", ["x", "y"], EVar "x")

```

タプルの最初の要素は名前 (この場合は `K`) です。2 番目は引数リストです (この場合、`x` と `y` の 2 つの変数があります)。タプルの 3 番目のコンポーネントは、スーパーコンビネータの本体です (この例では、変数 `x` だけです)。

この式を書き直すと、次のようになります。

```
("K", 2, compileR (EVar "x") [("x", 0), ("y", 1)])
```

結果のトリプルは、名前 (K)、スーパーコンビネータを削減するために必要な引数の数 (この場合は 2 つ)、およびインスタンス化を実行するためのコードシーケンスで構成されます。この式を書き直すと、このスーパーコンビネータのコードシーケンスが生成されます。環境は式  $[("x", 0), ("y", 1)]$  で表されることに注意してください。これは、本体をインスタンス化すると、 $x$  へのポインターが引数スタックの一番上にあり、 $y$  へのポインターがスタックの  $x$  のすぐ下にあることを示しています。

```
("K", 2, compileC (EVar "x") [("x", 0), ("y", 1)] ++ [Slide 3, Unwind])
```

`compileR` 関数は、`compileC` を使用して本体をコンパイルし、最後に `Slide` および `Unwind` 命令を追加するように定義されています。

本体をコンパイルするには、 $x$  を検索し、それがスタックの一番上にあることを確認します。`Push 0` を使用して、スタックのトップのコピーを作成するコードを生成します。

```
("K", 2, [Push 0, Slide 3, Unwind])
```

**演習 3.3** プレリユードの定義から  $S$  コンビネータの変換の同等のシーケンスを書き出します。 $S$  が次のように定義されていることを思い出してください。

$$S f g x = f x (g x)$$

Appendix B に示す単純なプログラムのいずれかでコンパイラとマシンを実行して、最終結果を確認してください ( $S$  は標準プレリユードにあります)。

## プリミティブ

この最小限の  $G$  マシンにはプリミティブがないため、実装するものは何もありません。

```
compiledPrimitives :: [GmCompiledSC]
compiledPrimitives = []
```

### 3.3.5 結果の表示

多くの状態コンポーネントは抽象データ型である (したがって、直接出力できない) ため、マシンが生成する状態のプリティ プリンターを定義する必要があります。一度に表示すると出力が膨大になり、あまり参考にならないことも事実です。印刷は、showResults によって制御されます。スーパーコンビネータ コード シーケンス、状態遷移、および最終統計の 3 つの出力が生成されます。

```
showResults :: [GmState] -> [Char]
showResults states
  = iDisplay (iConcat [iStr "Supercombinator definitions", iNewline,
                        iInterleave iNewline (map (showSC s) (getGlobals s)),
                        iNewline, iNewline,
                        iStr "State transitions", iNewline, iNewline,
                        iLayn (map showState states),
                        iNewline, iNewline,
                        showStats (last states)])
  where (s:ss) = states
```

これらのそれぞれを順番に取り上げて、showSC から始めます。これにより、グローバルに関連付けられた一意のグローバル ヒープ ノードでスーパーコンビネータのコードが検出され、showInstructions を使用してコード シーケンスが出力されます。

```
showSC :: GmState -> (Name, Addr) -> Iseq
showSC s (name, addr)
  = iConcat [ iStr "Code for ", iStr name, iNewline,
              showInstructions code, iNewline, iNewline]
  where (NGlobal arity code) = (hLookup (getHeap s) addr)
```

次に、showInstructions を使用してコード シーケンスを出力します。

```
showInstructions :: GmCode -> Iseq
showInstructions is
  = iConcat [iStr " Code:{",
              iIndent (iInterleave iNewline (map showInstruction is)),
              iStr "}", iNewline]
```

個々の命令の出力は、showInstruction によって提供されます。

```
showInstruction :: Instruction -> Iseq
showInstruction Unwind      = iStr "Unwind"
```



```

showInstruction (Pushglobal f) = (iStr "Pushglobal ") 'iAppend' (iStr f)
showInstruction (Push n)      = (iStr "Push ")      'iAppend' (iNum n)
showInstruction (Pushint n)   = (iStr "Pushint ")    'iAppend' (iNum n)
showInstruction Mkap          = iStr "Mkap"
showInstruction (Slide n)     = (iStr "Slide ")      'iAppend' (iNum n)

```

出力の次の主要な部分は状態遷移です。これらは、showState を使用して個別に処理されます。

```

showState :: GmState -> Iseq
showState s
  = iConcat [showStack s, iNewline,
             showInstructions (getCode s), iNewline]

```

ダイアグラムに対応するために、印刷されたスタックの一番下にスタックの一番上を配置したいと考えています。この目的のために、スタックを逆にします。

```

showStack :: GmState -> Iseq
showStack s
  = iConcat [iStr " Stack:[",
             iIndent (iInterleave iNewline
                        (map (showStackItem s) (reverse (getStack s)))),
             iStr "]""]

```

スタック アイテムは、showStackItem を使用して表示されます。スタックに格納されているアドレスと、それが指すヒープ内のオブジェクトを出力します。

```

showStackItem :: GmState -> Addr -> Iseq
showStackItem s a
  = iConcat [iStr (showaddr a), iStr ": ",
             showNode s a (hLookup (getHeap s) a)]

```

関数 showNode は、グローバル名とヒープ アドレスの関連リストを逆にし、検出されたグローバル ノードを表示する必要があります。

```

showNode :: GmState -> Addr -> Node -> Iseq
showNode s a (NNum n) = iNum n
showNode s a (NGlobal n g) = iConcat [iStr "Global ", iStr v]
                                where v = head [n | (n,b) <- getGlobals s, a==b]
showNode s a (NAp a1 a2) = iConcat [iStr "Ap ", iStr (showaddr a1),
                                     iStr " ", iStr (showaddr a2)]

```

最後に、showStats を使用して、累積された統計を出力します。

```
showStats :: GmState -> Iseq
showStats s
    = iConcat [ iStr "Steps taken = ", iNum (statGetSteps (getStats s)) ]
```

以上で基本的な G マシンの説明を終わります。今後は、より洗練されたものにする方法を検討していきます。

### 3.3.6 Mark 1 G マシンの改良

演習 3.4 プログラム `main = S K K 3` を実行します。何ステップかかりますか？ テンプレートインスタンス化マシンで取得した値と異なるのはなぜですか？ 取られたステップを比較することは、マシンの公平な比較だと思いますか？

演習 3.5 Appendix B の他のプログラムを実行してみてください。この単純な機械には算術がないことを思い出してください。

演習 3.6 Pushglobal で使用したのと同じトリックを使用して Pushint を実装することができます：個別の数値ごとに、ヒープ内に一意のノードを作成します。たとえば、最初に Pushint 2 を実行するとき、ノード NNum 2 のヒープ内のアドレスに "2" を関連付けるように gmGlobals を更新します。

遷移規則では、 $n$  という名前のグローバルが既に存在する場合、このグローバル ノードを再利用できます。

(3.13)	$\text{Pushint } n : i \quad s \quad h \quad m[n : a]$ $\Rightarrow \quad i \quad a : s \quad h \quad m$
--------	--

そうでない場合は、新しいノードを作成してグローバル マップに追加します。

(3.14)	$\text{Pushint } n : i \quad s \quad h \quad m$ $\Rightarrow \quad i \quad a : s \quad h[a : \text{NNum } n] \quad m[n : a]$
--------	--

このスキームの利点は、Pushint が実行されるたびにヒープ内の同じ番号のノードを再利用できることです。Pushint 命令のこの新しい遷移 pushint を実装します。グローバル コンポーネントのアクセス関数を定義して、putGlobals と呼ぶ必要があります。

### 3.4 Mark 2: 遅延評価させる

Mark 1 G マシンを遅延評価させるために、いくつかの小さな変更を加えます。巻き戻す前に元の式のルートノードを上書きしないため、Mark 1 マシンは現時点では遅延評価していません。この更新については、セクション 2.1.5 で説明します。Mark 2 マシンでは、スーパーコンビネータの本体をインスタンス化した後、元の redex のルートを、新しく構築されたインスタンスを指す間接ノードで上書きするという考えです。その効果は、マシンが前回の redex 簡約の際にインスタンス化された値を「覚えている」ため、それを再計算する必要がないということです。

この変更を次のように実装します。Mark 1 マシンでは、各スーパーコンビネータのコードは `[Slide (n + 1); Unwind]` で終了していました。更新をキャプチャするには、これを `[Update n; Pop n; Unwind]` に置き換えます。これを次の図に示します。この図では、# を使用して間接ノードを表しています。




Figure 3.4: Mark 1 G-machine (executing Slide n+1)

図 3.4 は、Mark 1 マシンが Slide  $n + 1$  命令を実行する方法を示しています。



Figure 3.5: Mark 2 G-machine (executing [Update  $n$ , Pop  $n$ ])

図 3.5 では、Mark 2 マシンがシーケンス [Update  $n$ ; Pop  $n$ ] を実行しています。これは、[Slide  $n + 1$ ] の遅延置換として使用することを提案するシーケンスです。Update 命令は、スーパーコンピネータ本体の新しく作成されたインスタンスでルートノードを上書きする役割を果たします。Pop 命令は、引数が不要になったため、スタックから引数を削除するために使用されます。

まず、データ構造に必要な変更を考えてみましょう。

### 3.4.1 データ構造

前回生成した単一の命令 Slide  $n + 1$  の代わりに、一連の命令 [Update  $n$ , Pop  $n$ ] を生成します。したがって、これらの命令を新しい命令セットに含める必要があります。

```
data Instruction
  = Unwind
  | Pushglobal Name
  | Pushint Int
  | Push Int
  | Mkap
  | Update Int -- Mark2 で追加
  | Pop Int

instance Eq Instruction
  where
    Unwind      == Unwind      = True
    Pushglobal a == Pushglobal b = a == b
    Pushint    a == Pushint    b = a == b
```

```

Push      a == Push      b = a == b
Mkap      == Mkap        = True
Update    a == Update    b = a == b
_          == _           = False

```

演習 3.7 関数 `showInstruction` を変更して、新しい命令が表示されるようにします。

間接ノードを実装するには、ヒープに新しいノード タイプが必要です: 間接ノードに使用する `NInd` です。

```

data Node
  = NNum      Int          -- Numbers
  | NAp       Addr Addr    -- Applications
  | NGlobal   Int  GmCode  -- Globals
  | NInd      Addr          -- Indirections  -- Mark2 で追加
instance Eq Node
  where
    NNum      a  == NNum      b  = a == b -- needed to check conditions
    NAp       a b == NAp       c d = False -- not needed
    NGlobal   a b == NGlobal   c d = False -- not needed
    NInd      a  == NInd      b  = False  -- not needed

```

ここでも、データ型の拡張を反映するように、表示関数 `showNode` を再定義する必要があります。

演習 3.8 `showNode` に必要な変更を加えます。

2 つの新しい命令のセマンティクスはまだ与えていません。これは以下で行われます。

### 3.4.2 評価器

`Update n` 命令の効果は、 $n + 1$  番目のスタック項目を、スタックの一番上にある項目への間接指定で上書きすることです。このアドレッシングモードは、[Peyton Jones 1987] で使用されているものとは異なることに注意してください。この命令の意図した関数適用では、 $a_1, \dots, a_n$  はスパインを形成する  $n$  個の関数適用ノードであり、 $a_0$  は関数ノードです。

(3.15)	$\text{Update } n : i \quad a : a_0 : \dots : a_n : s \quad h \quad m$
$\Rightarrow$	$i \quad a_0 : \dots : a_n : s \quad h[a_n : \text{NInd } a] \quad m$

Pop  $n$  命令は、 $n$  個のスタック項目を単純に削除します。繰り返しますが、Mark 2 G マシンでは、 $a_1, \dots, a_n$  は redex のスパインを形成する関数適用ノードです。

(3.16)	$\text{Pop } n : i \quad a_1 : \dots : a_n : s \quad h \quad m$ $\Rightarrow \quad i \quad s \quad h \quad m$
--------	---

スタックアイテムのトップが間接の場合、Unwind の遷移も定義する必要があります。その結果、現在のスタックアイテムが間接参照が指すアイテムに置き換えられます。

(3.17)	$[\text{Unwind}] \quad a_0 : s \quad h[a_0 : \text{NInd } a] \quad m$ $\Rightarrow \quad [\text{Unwind}] \quad a : s \quad h \quad m$
--------	---

演習 3.9 Mark 1 マシンの dispatch 関数を変更して、新しい命令を組み込みます。新しい遷移規則を実装します。

### 3.4.3 コンパイラ

コンパイラに対する唯一の変更点は、 $\mathcal{R}$  スキームによって生成されたコードにあります。新しい定義を図 3.6 に示します。

$\mathcal{R}[[e]] \rho d$ は、アリティ $d$ のスーパーコンビネータのために、環境 $\rho$ で式 $e$ をインスタンス化するコードを生成し、次に、結果のスタックの巻き戻しに進みます。  $\mathcal{R}[[e]] \rho d = \mathcal{C}[[e]] \rho ++ [\text{Update } d, \text{Pop } d, \text{Unwind}]$
--

演習 3.10 compileR を変更して、新しい  $\mathcal{R}$  スキームを実装します。

演習 3.11 次のプログラムで遅延評価を実行します。

```
twice f x = f (f x)
id x = x
main = twice twice id 3
```

何ステップがかかりますか？ Mark1 マシンで取得した値と異なるのはなぜですか？ マシンのステップ数を比較するのは公正ですか？

### 3.5 Mark 3: let(rec) 式

本体に `let(rec)` バインド変数を含むスーパーコンビネータをコンパイラが受け入れるように、言語を拡張します。これらは、コンストラクタ `Elet` によってデータ型 `coreExpr` で表されます。3 つの引数を取ります。定義を再帰的に処理するかどうかを示すブールフラグ、定義自体、および定義を使用する式です。

ローカル定義を追加してマシンを拡張する前に、スタックをもう一度見てみましょう。特に、変数のより効率的なアクセス方法を定義しようとしています。効率の引数に加えて、ローカルにバインドされた変数へのアクセスを、関数パラメータのバインドに使用されるものと同じにしたいと考えています。

#### スタックからの引数アクセス

巻き戻しプロセスがスーパーコンビネータノード `f` に到達し、スーパーコンビネータが  $n$  個の引数を取るとします。Mark 1 マシンでは、スタックは図 3.7 の左側の図に示す状態になります。



Figure 3.7: Stack layout on entry to function `f`

スーパーコンビネータ `f` に到達したので、Mark 3 G マシンでは、スタックがわずかに変更されています。同等の Mark 3 スタックを図 3.7 の右側の図に示します。上位  $n$  要素は式 `e1 ... en` を直接指しています。ここで重要

な点は、変数へのアクセスが高速になることです (変数が少なくとも 1 回アクセスされる場合)。これは、右側の引数を取得するために、関数適用ノードを 1 回しか見ていないためです。

これにより、スーパーコンビネータで仮パラメータに代入される式へのアクセス効率が向上します。

Mark1 マシンに関して:

- Push 命令で関数 `getArg` はもう必要ありません。
- しかし、十分な引数でスーパーコンビネータを `Unwind` するときは、スタックを再配置する必要があります。

`Update` を実行できるように、`redex` のルートへのポインターを保持していることに注意してください。

#### 命令への影響

新しいスタックレイアウトを使用することを選択した場合、対処するために特定のマシン命令を変更する必要があります。影響を受ける命令は、`Push` と `Unwind` です。引数を取得するために関数適用ノードを「調べる」必要がないため、`Push` 命令を変更する必要があります。

(3.18)	$\text{Push } n : i \quad a_0 : \dots : a_n : s \quad h \quad m$
	$\Rightarrow \quad i \quad a_n : a_0 : \dots : a_n : s \quad h \quad m$

新しいスタックレイアウトに必要なその他の変更は、`Unwind` がスタックを再配置する必要があることです。十分な引数を持つスーパーコンビネータがスタックの一番上にある場合は常に、この再配置が必要です。`Unwind` の新しい遷移規則は次のとおりです。

(3.19)	$[\text{Unwind}] \quad a_0 : \dots : a_n : s \quad h$	$\left[ \begin{array}{l} a_0 : \text{NGlobal } n \ c \\ a_1 : \text{NAp } a_0 \ a'_1 \\ \dots \\ a_n : \text{NAp } a_{n-1} \ a'_n \end{array} \right]$	$m$
	$\Rightarrow \quad c \quad a'_1 : \dots : a'_n : a_n : s \quad h$		$m$

`Unwind` のこの定義は、 $n$  がゼロの場合に適切に機能することに注意してください。

演習 3.12 `dispatch` 関数と新しい命令セットの新しい遷移を書き直します。スタックを再配置するには、`rearrange` 関数を使用する必要があります。



```

rearrange :: Int -> GmHeap -> GmStack -> GmStack
rearrange n heap as
  = take n as' ++ drop n as
    where as' = map (getArg . hLookup heap) (tl as)

```

演習 3.13 Appendix B のいくつかのサンプルプログラムでコンパイラと新しい抽象マシンをテストし、実装が引き続き機能することを確認します。

### 3.5.1 ローカルにバインドされた変数

ここで、最初に非再帰的なケースを考慮して、`let(rec)` 式の実装に戻ります。式 `let  $x_1 = e_1; \dots; x_n = e_n$  in  $e$`  の変数  $x_1 \dots x_n$  は、式  $e_1 \dots e_n$  が作成されると、スーパーコンビネータへの引数と同じ方法で処理できます。つまり、変数  $x_1 \dots x_n$  にオフセットを介してスタックにアクセスし、環境を使用してそれらの位置を記録します。

ローカル定義を構築するためのコードが `Code` であると仮定すると、図 3.8 に示す一連のアクションが必要になります。最初に、スタックにはスーパーコンビネータへの引数へのポインタが含まれます。ローカル定義を構築するコードが実行されると、スタック上に  $n$  個の新しいポインタができます。 $x_i$  を  $e_i$  へのポインタにマップする新しい環境で、`let` 式の本体の作成に進むことができます。最後に、式  $e_1 \dots e_n$  へのポインタをスタックから捨てる必要があります。




Figure 3.8: Stack usage in non-recursive local definitions

$n$  個の新しい変数をスタック  $(x_1 \dots x_n)$  に追加したため、 $e$  をコンパイルするために使用する変数マップでこの事実には注意する必要があります。ローカルバインディングを構築するコード (`Code` と呼びます) は、単純に各式  $e_1 \dots e_n$  のグラフを作成し、グラフの断片のアドレスをスタックに残します。

変数  $x_1 \dots x_n$  のいずれかを使用する本体式  $e$  を作成した後、スタックから  $e_1 \dots e_n$  へのポインタを削除する必要があります。これは、Slide 命令を使用して達成されます。非再帰的なローカル定義をコンパイルするための完全なスキームは、図 3.10 (p.109) に示されています。

再帰的なローカル定義の状況はより複雑です。変数  $x_1 \dots x_n$  がスコープ内にあるように、式  $e_1 \dots e_n$  のそれぞれをコンパイルする必要があります。これを行うには、グラフに空のノードを作成し、それらへのポインタをスタックに残します。次に、各式  $e_1 \dots e_n$  は、非再帰ケースの本体のコンパイルに使用したのと同じ変数マップを使用してコンパイルされます。各式のコンパイル済みコードの最後に、空のノードを正しいグラフで上書きする Update 命令を配置します。これを行うには、 $n$  個の空のグラフノードを作成する新しい命令 Alloc  $n$  が 1 つ必要です。図 3.9 では、空のグラフノードは ? 記号で表されます。



Figure 3.9: Constructing a recursively defined expression:  $e_1$

図 3.9 に示すプロセスは、各式  $e_1 \dots e_n$  が処理されるまで繰り返す必要があります。本体  $e$  のコードのコンパイルは、非再帰的なローカル定義の前のケースと同じです。Mark3 マシン用の新しいデータ型を追加します。

### 3.5.2 データ構造

命令データ型には、新しい Alloc 命令と Mark1 マシンからの Slide 命令を含む、Mark2 マシンのすべての命令が含まれます。

演習 3.14 Alloc と Slide が含まれるように、データ型 instruction を変更します。また、これらの新しい命令に対応するために、関数 showInstruction を変更する必要があります。

### 3.5.3 評価器

Mark3 G マシンの場合、ヒープ内に  $n$  個の場所を作成する `Alloc` 命令を追加する必要があります。これらの場所を使用して、ローカルにバインドされた式を保存する場所をマークします。これらのノードは、無効なヒープアドレス `hNull` を指す間接ノードとして最初に作成されます。`Alloc` によって作成されたこれらのノードは上書きされるため、それらにどの値を割り当てるかは問題ではありません。

`Alloc` 命令の遷移関数である `alloc` を実装するには、補助関数 `allocNodes` を使用します。必要なノード数と現在のヒープを指定すると、変更されたヒープと間接ノードのアドレスのリストで構成されるペアが返されます。

```
allocNodes :: Int -> GmHeap -> (GmHeap, [Addr])
allocNodes 0 heap = (heap, [])
allocNodes (n+1) heap = (heap2, a:as)
    where (heap1, as) = allocNodes n heap
          (heap2, a) = hAlloc heap1 (NInd hNull)
```

演習 3.15 新しい命令のケースを使用して、`dispatch` 関数を拡張します。

`AllocNodes` を使用して、`Alloc` 命令の遷移関数である `alloc` を実装する必要があります。

### 3.5.4 コンパイラ

コンパイラに対する唯一の変更点は、`C` スキームがコードをコンパイルできるケースが 2 つ増えたことです。`compileC` への変更は簡単です。より広い範囲の `coreExpr` に対応できるようになりました。`compileLetrec` と `compileLet` という 2 つの新しい関数が必要です。

$C[e] \rho$  は、環境  $\rho$  で  $e$  のグラフを構築するコードを生成し、  
それへのポインターをスタックの一番上に残します。

$C[f] \rho$	$=$	$[Pushglobal\ f]$	ここで、 $f$ はスーパーコンピネータです。
$C[x] \rho$	$=$	$[Push\ (\rho\ x)]$	ここで、 $x$ はローカル変数です。
$C[i] \rho$	$=$	$[Pushint\ i]$	
$C[e_0\ e_1] \rho$	$=$	$C[e_1] \rho ++$ $C[e_0] \rho^{+1} ++ [Mkap]$	ここで、 $\rho^{+n} x = (\rho\ x) + n$ です。
$C[let\ x_1 = e_1; \dots; x_n = e_n\ in\ e] \rho$	$=$	$C[e_1] \rho^{+0} ++$ $\dots$ $C[e_n] \rho^{+(n-1)} ++$ $C[e] \rho' ++ [Slide\ n]$	ここで $\rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0]$ です。
$C[letrec\ x_1 = e_1; \dots; x_n = e_n\ in\ e] \rho$	$=$	$[Alloc\ n] ++$ $C[e_1] \rho' ++$ $[Update\ n-1] ++$ $\dots$ $C[e_n] \rho' ++$ $[Update\ 0] ++$ $C[e] \rho' ++ [Slide\ n]$	ここで $\rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0]$ です。

```

compileC :: GmCompiler
compileC (EVar v) args
  | elem v (aDomain args) = [Push n]
  | otherwise              = [Pushglobal v]
  where n = aLookup args v (error "")
compileC (ENum n)    env = [Pushint n]
compileC (EAp e1 e2) env = compileC e2 env ++
                             compileC e1 (argOffset 1 env) ++
                             [Mkap]
compileC (ELet recursive defs e) args
  | recursive = compileLetrec compileC defs e args
  | otherwise = compileLet    compileC defs e args

```

`compileLet` の定義は、図 3.10 の仕様に従います。引数として、本体  $e$  のコンパイルスキーム `comp`、定義 `defs`、および現在の環境 `env` を受け取ります。マシンの新しいバージョンでこの関数を書き直す必要がないように、コンパイラパラメータを提供しました。

```

compileLet :: GmCompiler -> [(Name, CoreExpr)] -> GmCompiler
compileLet comp defs expr env

```

```

= compileLet' defs env ++ comp expr env' ++ [Slide (length defs)]
  where env' = compileArgs defs env

```

新しい定義のコンパイルは、関数 `compileLet'` によって実行されます。

```

compileLet' :: [(Name, CoreExpr)] -> GmEnvironment -> GmCode
compileLet' [] env = []
compileLet' ((name, expr):defs) env
  = compileC expr env ++ compileLet' defs (argOffset 1 env)

```

`compileLet` はまた、`compileArgs` を使用して、本体のコンパイル用のスタックへのオフセットを変更します。

```

compileArgs :: [(Name, CoreExpr)] -> GmEnvironment -> GmEnvironment
compileArgs defs env
  = zip (map first defs) [n-1, n-2 .. 0] ++ argOffset n env
  where n = length defs

```

例

この例では、不動点コンビネータ `Y` のコードがどのようにコンパイルされるかを示します。使用する定義は次のとおりです。

```

Y f = letrec x = f x in x

```

これは、いわゆる「結び目」不動点コンビネータです。結果のコードを実行すると、なぜこの名前になっているのかがわかります。上記の定義がコンパイルされると、`compileSc` 関数はスーパーコンビネータのコードを作成する必要があります。

```

compileSc ("Y", ["f"], ELet True [("x", EAp (EVar "f") (EVar "x"))] (EVar "x"))

```

これは、変数 `f` の環境で `compileR` 関数を呼び出します。最初にスーパーコンビネータの名前 (`Y`) とその引数の数 (`1`) を作成しました。

```

("Y", 1, compileR e [("f", 0)])
where e = ELet True [("x", EAp (EVar "f") (EVar "x"))] (EVar "x")

```

便宜上、式の本体を `e` と呼びます。関数 `compileR` は `compileC` を呼び出し、最後に片付けコードを配置します。

```

("Y", 1, compileC e [("f", 0)] ++ [Update 1, Pop 1, Unwind])

```

図 3.10 のコンパイルスキームを参照すると、letrec をコンパイルするには、最初に新しい環境を作成することがわかります。図では、これは  $\rho'$  と呼ばれます。この例では、p と呼びます。これは初期環境の拡張であり、ローカル変数 x にスタックの場所も与えます。

```
("Y", 1, [Alloc1] ++
  compileC (EAp (EVar "f") (EVar "x")) p ++ [Update 0] ++
  compileC (EVar "x") p ++ [Slide 1] ++
  [Update 1, Pop 1, Unwind])
where p = [("x", 0), ("f", 1)]
```

コード生成は、コンパイルスキームと同じ方法でレイアウトされます。compileC を含む式を単純化すると、次のようになります。

```
("Y", 1, [Alloc 1] ++
  [Push 0, Push 2, Mkap] ++ [Update 0] ++
  [Push 0] ++ [Slide 1] ++
  [Update 1, Pop 1, Unwind])
```

これにより、次のコードシーケンスが得られます。

```
("Y", 1, [Alloc 1, Push 0, Push 2, Mkap, Update 0, Push 0,
  Slide 1, Update 1, Pop 1, Unwind])
```

このコードがどのように実行されるかは、図 3.11 で確認できます。Y スーパーコンビネータのこの定義は、5 番目の命令として Update 0 を実行するときにグラフに結び目を作るため、「結び目」と呼ばれます。演習 3.18 として残されているため、残りの手順は示していません。



Figure 3.11: Execution of code for Y

演習 3.16 letrec のコンパイルは、図 3.10 で定義されています。この操作を実行するには、関数 `compileLetrec` を実装します。

演習 3.17 新しいコンパイラと命令セットが適切に動作することを示すために、どのテスト プログラムを使用しますか？

演習 3.18 スーパーコンビネータ  $Y$  用に生成されたコードを実行するか、そうでない場合は、残りの状態遷移を図 3.11 のスタイルで描画します。

演習 3.19 スーパーコンビネータ  $Y$  の短い代替コードシーケンスを指定します。それはまだ「結び目」バージョンを構築する必要があります。

演習 3.20 言語に letrec 構文がない場合、不動点コンビネータ  $Y$  をどのように定義しますか？ この定義は、例で使った定義とどう違うのでしょうか？

### 3.6 Mark 4: プリミティブの追加

このセクションでは、基本的な操作を G マシンに追加します。これは便利です。基本演算とは、加算、乗算などの演算を意味します。このセクションでは、実行例として足し算を使用します。

加算命令は Add と呼ばれます。ヒープから 2 つの数値を追加し、結果をヒープ内の新しいノードに配置します。2 つの引数のアドレスはスタックの一番上にあります。これは、結果のアドレスがその後配置される場所です。次の遷移規則があります。

(3.21)	$\text{Add} : i \quad a_0 : a_1 : s \quad d \quad h[a_0 : \text{NNum } n_0, \quad a_1 : \text{NNum } n_1] \quad m$
$\Rightarrow$	$i \quad a : s \quad d \quad h[a : \text{NNum } (n_0 + n_1)] \quad m$

必要な残りの操作を実装するために、他の命令で G マシンを拡張し続けることができますが、その前に、ここで何か見逃していないかどうかを検討するために一時停止しましょう。問題は、スタックの一番上にある 2 つのオブジェクトが数値である場合にのみルールが適用されることです。遅延評価をサポートするマシンで作業しているため、これが常に当てはまると想定する正当な理由はありません。テンプレートマシンで Add は、その引数が評価されることを確認しました。G マシンでは、命令を単純に保ちたいので、引数が既に評価されていることが保証されている状況でのみ Add を使用します。

代わりに、Eval 命令を使用して命令セットをさらに拡張します。これは、次の制約を満たします。

次の状態にあるとします。

$\text{Eval} : i \quad a : s \quad h \quad m$
---

命令シーケンス  $i$  で実行が再開されるたびに、状態は次のようになります。

$i \quad a : s \quad h' \quad m$
----------------------------------

スタックの一番上にあるアイテムは WHNF になります。Eval が終了しない可能性もあります。これは、スタックの一番上から指されたノードに WHNF がない場合に当てはまります。

アドレスがスタックの一番上にあるノードが既に WHNF にある場合、Eval 命令は何もしません。実行するリダクションがある場合、Eval のアクション



は WHNF に対する評価を実行することです。この呼び出しが終了すると、ヒープ コンポーネント以外は何も変更されずに実行が再開されます。これは、プログラミング言語の実装で伝統的に使用されているサブルーチンの呼び出しと戻りの構造に似ています。この機能を実装する古典的な方法は、スタックを使用することです。スタックは、サブルーチン呼び出しが完了したときに再開できるように、マシンの現在のコンテキストを十分に保存します。

Mark 4 G マシンでは、このスタックはダンプと呼ばれ、最初のコンポーネントがコード シーケンスで、2 番目のコンポーネントがスタックであるペアのスタックです。これは、元のコード シーケンスと元のスタックを復元する必要があることを除いて、テンプレート マシンのダンプ (セクション 2.6 を参照) に似ています。したがって、両方のコンポーネントがダンプに保持されます。

### 3.6.1 データ構造

ダンプコンポーネントを追加して、G マシンの状態を拡張します。前に説明したように、これは評価器への再帰呼び出しを実装するために使用されます。

```
type GmState = ( GmCode, -- current Instruction
                 GmStack, -- current Stack
                 GmDump,  -- current Dump
                 GmHeap,  -- Heap of Nodes
                 GmGlobals, -- Global addresses in Heap
                 GmStats) -- Statistics
```

ダンプ自体は、dumpItem のスタックです。これらはそれぞれ、元の計算を再開するときに使用する命令ストリームとスタックで構成されるペアです。

```
type GmDump = [GmDumpItem]
type GmDumpItem = (GmCode, GmStack)
```

この新しいコンポーネントを追加するときは、以前に指定したすべてのアクセス機能を変更する必要があります。ダンプ用のアクセス機能も追加する必要があります。

```
getDump :: GmState -> GmDump
getDump (i, staxk, dump, heap, globals, stats) = dump

putDump :: GmDump -> GmState -> GmState
putDump dump' (i, stack, dump, heap, globals, stats)
    = (i, stack, dump', heap, globals, stats)
```

G マシンの状態でパターンマッチングを行ったのはアクセス関数のみであることに注意してください。新しいコンポーネントを状態に追加した結果としての他の関数への変更は、もはや必要ありません。

演習 3.21 他のアクセス機能に関連する変更を加えます。

状態の新しい定義に加えて、いくつかの新しい指示も必要です。Mark3 マシンからのすべての命令を再利用します。

```
data Instruction
  = Slide Int
  | Alloc Int
  | Update Int
  | Pop Int
  | Unwind
  | Pushglobal Name
  | Pushint Int
  | Push Int
  | Mkap
```

さらに、Eval 命令を含めます。

```
  | Eval
```

次の算術命令:

```
  | Add | Sub | Mul | Div | Neg
```

および次の比較命令:

```
  | Eq | Ne | Lt | Le | Gt | Ge
```

また、Cond 命令にはプリミティブ形式の条件分岐も含まれています。

```
  | Cond GmCode GmCode
```

演習 3.22 ケースを showInstruction に追加して、新しい指示をすべて出力します。

### 3.6.2 状態の表示

これを機に showState の定義を見直し、dump コンポーネントを表示するようにしました。

```

showState :: GmState -> Iseq
showState s
    = iConcat [showStack s, iNewline,
               showDump s, iNewline,
               showInstructions (getCode s), iNewline]

```

したがって、showDump を定義する必要があります。

```

showDump :: GmState -> Iseq
showDump s
    = iConcat [iStr " Dump:[",
               iIndent (iInterleave iNewline (map showDumpItem (reverse (getDump s)))),
               iStr "]" ]

```

これには、関数 showDumpItem が必要です。

```

showDumpItem :: GmDumpItem -> Iseq
showDumpItem (code, stack)
    = iConcat [iStr "<",
               shortShowInstructions 3 code, iStr ", ",
               shortShowStack stack, iStr ">"]

```

関数 shortShowInstructions を使用して、命令ストリームの最初の 3 つの命令のみをダンプ項目に出力します。これは通常、計算が再開される場所を示すのに十分です。

```

shortShowInstructions :: Int -> GmCode -> Iseq
shortShowInstructions number code
    = iConcat [iStr "{", iInterleave (iStr "; ") dotcodes, iStr "}"]
    where codes = map showInstruction (take number code)
          dotcodes | length code > number = codes ++ [iStr "..."]
                  | otherwise             = codes

```

同様に、ダンプ アイテムのスタック コンポーネントの完全な詳細も必要ないため、shortShowStack を使用します。

```

shortShowStack :: GmStack -> Iseq
shortShowStack stack
    = iConcat [iStr "[",
               iInterleave (iStr ", ") (map (iStr . showaddr) stack),
               iStr "]" ]

```

### 3.6.3 新しい命令遷移

#### 評価命令

実際には、ダンプを操作する命令はほとんどありません。まず、Eval 自体があり、スタックの最上位のノードが WHNF にならない場合は常に新しいダンプ項目を作成します。2 つ目は、評価が完了したときにダンプ項目をポップする Unwind 命令への変更です。

まず、新しい Unwind 命令について説明します。スタックに保持されている式が WHNF にある場合、Unwind はダンプから古いコンテキストを復元し、復元された古いスタックにスタック内の最後のアドレスを配置できます。これは、数値の場合の遷移で明確にわかります。<sup>7</sup>

(3.22)	$\begin{array}{c} [\text{Unwind}] \quad a : s \quad \langle i', s' \rangle : d \quad h[a : \text{NNum } n] \quad m \\ \Rightarrow \quad i' \quad a : s' \quad d \quad h \quad m \end{array}$
--------	--

アドレス  $a$  の式は整数であるため WHNF にあるため、古い命令シーケンス  $i'$  を復元すると、スタックはアドレス  $a$  を先頭にした古いスタック  $s'$  になります。Unwind の他のすべてのトランジションは、Mark 3 マシンの場合と同じままです (ダンプ コンポーネントがその状態にあることを除いて)。

これで、Eval のルールを指定できるようになりました。スタック  $s$  の残りとし残りの命令  $i$  をダンプ項目としてダンプに保存します。新しいコードシーケンスはちょうど巻き戻されており、新しいスタックにはシングルトン  $a$  が含まれています。

(3.23)	$\begin{array}{c} \text{Eval} : i \quad a : s \quad d \quad h \quad m \\ \Rightarrow \quad [\text{Unwind}] \quad [a] \quad \langle i, s \rangle : d \quad h \quad m \end{array}$
--------	--

#### 算術命令

すべての 2 項算術演算子には、次の一般的な遷移規則があります。実装したい算術演算子が  $\odot$  であると仮定しましょう。命令の遷移規則は次のとおりです。

(3.24)	$\begin{array}{c} \odot : i \quad a_0 : a_1 : s \quad d \quad h[a_0 : \text{NNum } n_0, \quad a_1 : \text{NNum } n_1] \quad m \\ \Rightarrow \quad i \quad a : s \quad d \quad h[a : \text{NNum } (n_0 \odot n_1)] \quad m \end{array}$
--------	---

<sup>7</sup>ルールは、ダンプが空でない場合にのみ適用されます。ダンプが空の場合、マシンは終了しています。

何が起きたかという、スタックの一番上にある 2 つの数値に 2 項演算子  $\odot$  が適用されています。ヒープに入力された結果のアドレスは、スタックに配置されます。Neg 命令は、スタックの一番上にある数値を否定するため、遷移規則があります。

(3.25)	$\text{Neg} : i \quad a : s \quad d \quad h[a : \text{NNum } n] \quad m$
$\Rightarrow$	$i \quad a' : s \quad d \quad h[a' : \text{NNum } (-n)] \quad m$

すべての二項演算がいかに類似しているかに注目してください。最初にヒープから 2 つの数値を抽出し、次に演算を実行し、最後に答えをヒープに戻します。これは、ヒープからの抽出 (値の「ボックス化解除」と呼びます) をパラメーター化する高階関数を作成する必要があることを示唆しています。そして、実行したい特定の操作とともに、ヒープに挿入します (これを値の「ボックス化」と呼びます)。

最初にボックス化操作を書きましょう。boxInteger は数値と初期状態を取り、数値がヒープに配置された新しい状態と、スタックの一番上に残されたこの新しいノードへのポインターを返します。

```
boxInteger :: Int -> GmState -> GmState
boxInteger n state
  = putStack (a: getStack state) (putHeap h' state)
    where (h', a) = hAlloc (getHeap state) (NNum n)
```

状態からアドレス  $a$  の整数を抽出するには、unboxInteger を使用します。

```
unboxInteger :: Addr -> GmState -> Int
unboxInteger a state
  = ub (hLookup (getHeap state) a)
    where ub (NNum i) = i
          ub n       = error "Unboxing a non-integer"
```

一般的なモナド演算子は、ボックス化関数 box、ボックス化解除関数 unbox、およびボックス化されていない値に対する演算子 op に関して指定できるようになりました。

```
primitive1 :: (b -> GmState -> GmState) -- boxing function
            -> (Addr -> GmState -> a) -- unbixing function
            -> (a -> b)                -- operator
            -> (GmState -> GmState)   -- state transition
primitive1 box unbox op state
```

```

= box (op (unbox a state)) (putStack as state)
  where (a:as) = getStack state

```

汎用二項演算子は、primitive2 を使用して同様の方法で実装できるようになりました。

```

primitive2 :: (b -> GmState -> GmState) -- boxing function
            -> (Addr -> GmState -> a) -- unbixing function
            -> (a -> a -> b)          -- operator
            -> (GmState -> GmState)   -- state transition

primitive2 box unbox op state
= box (op (unbox a0 state) (unbox a1 state)) (putStack as state)
  where (a0:a1:as) = getStack state

```

もっと明確に言うと、arithmetic1 はすべての単項算術を実装し、arithmetic2 はすべての二項算術を実装します。

```

arithmetic1 :: (Int -> Int)           -- arithmetic operator
            -> (GmState -> GmState) -- state transition
arithmetic1 = primitive1 boxInteger unboxInteger

arithmetic2 :: (Int -> Int -> Int)    -- arithmetic operation
            -> (GmState -> GmState) -- state transition
arithmetic2 = primitive2 boxInteger unboxInteger

```

注意深い読者が予想するように、これらの関数の一般性をこの章の後半で利用します。

**演習 3.23** マシンの新しい命令遷移をすべて実装します。dispatch 関数を変更して、新しい命令を処理します。高階関数の primitive1 と primitive2 を使用して、演算子を実装する必要があります。

**演習 3.24** Eval 命令の完了時に間接ノードがスタックの一番上に残らないのはなぜですか？

比較命令

すべての比較演算子には、次の一般的な遷移規則があります。実装したい比較演算子が  $\odot$  であると仮定すると、命令  $\odot$  の遷移規則は次のようになります。

(3.26)	$\begin{array}{ccccccc} \odot : i & a_0 : a_1 : s & d & h[a_0 : \text{NNum } n_0, & a_1 : \text{NNum } n_1] & m \\ \Rightarrow & i & a : s & d & h[a : \text{NNum } (n_0 \odot n_1)] & m \end{array}$
--------	---

スタックの一番上にある 2 つの数値には、2 項演算子  $\odot$  が適用されています。ヒープに入力された結果のアドレスは、スタックに配置されます。これは算術演算とほぼ同じです

違いは、たとえば `==` という操作は、整数ではなくブール値を返すことです。これを修正するために、次のルールを使用してブール値を整数に変換します。

- True を整数 1 で表します。
- False は整数 0 で表します。

Primitive2 の使用を可能にするために、`boxBoolean` を定義します。

```
boxBoolean :: Bool -> GmState -> GmState
boxBoolean b state
  = putStack (a: getStack state) (putHeap h' state)
    where (h',a) = hAlloc (getHeap state) (NNum b')
          b' | b          = 1
              | otherwise = 0
```

この定義を使用して、`comparison` と呼ばれる一般的な比較関数を書くことができます。この関数は、ブール値のボックス化関数、整数のボックス化解除関数 (`unboxInteger`)、および比較演算子を取ります。そして状態遷移を返します。

```
comparison :: (Int -> Int -> Bool) -> GmState -> GmState
comparison = primitive2 boxBoolean unboxInteger
```

最後に、`if` 関数のコンパイルに使用する `Cond` 命令を実装します。次の 2 つの遷移規則があります。

(3.27)	$\text{Cond } i_1 \quad i_2 : i \quad a : s \quad d \quad h[a : \text{NNum } 1] \quad m$ $\Longrightarrow \quad i_1 ++ i \quad s \quad d \quad h \quad m$
--------	---

最初のケース (スタックの一番上に数字の 1 がある場合) では、最初の分岐を取ります。これは、命令  $i$  の実行を続ける前に、命令  $i_1$  を実行することを意味します。

(3.28)	$\text{Cond } i_1 \quad i_2 : i \quad a : s \quad d \quad h[a : \text{NNum } 0] \quad m$ $\Longrightarrow \quad i_2 ++ i \quad s \quad d \quad h \quad m$
--------	---

あるいは、スタックの一番上の番号が 0 の場合、最初に命令シーケンス  $i_2$  を実行し、次にシーケンス  $i$  を実行します。

演習 3.25 比較命令と Cond 命令の遷移を実装します。

### 3.6.4 コンパイラ

これらの新しい命令を利用して算術式をコンパイルするには、最終的にコンパイラを変更する必要があります。現時点では、最小限の変更のみを行います。苦労して追加した算術命令を使用できるようになります。まず、compile 関数は、初期ダンプが空で、初期コード シーケンスがこれまでに使用したものとは異なる新しい初期状態を作成する必要があります。

```
compile :: CoreProgram -> GmState
compile program
  = (initialCode, [], [], heap, globals, statInitial)
    where (heap, globals) = buildInitialHeap program

initialCode :: GmCode
initialCode = [Pushglobal "main", Eval]
```

演習 3.26 最初の命令シーケンスが変更されたのはなぜですか？ 古いものを保持するとどうなりますか？

コンパイラを拡張する最も簡単な方法は、新しい組み込み関数ごとに G マシン コードを compiledPrimitives に追加することです。シーケンスの最初の 4 つの命令により、引数が整数に評価されることが保証されます。

```
compiledPrimitives :: [GmCompiledSC]
compiledPrimitives
  = [( "+", 2, [Push 1, Eval, Push 1, Eval, Add, Update 2, Pop 2, Unwind]),
      ("-", 2, [Push 1, Eval, Push 1, Eval, Sub, Update 2, Pop 2, Unwind]),
      ("*", 2, [Push 1, Eval, Push 1, Eval, Mul, Update 2, Pop 2, Unwind]),
      ("/", 2, [Push 1, Eval, Push 1, Eval, Div, Update 2, Pop 2, Unwind]),
```

否定関数も追加する必要があります。これは 1 つの引数しかとらないため、1 つの引数のみを評価します。

```
("negate", 1, [Push 0, Eval, Neg, Update 1, Pop 1, Unwind]),
```

比較操作は次のように実装されます。

```
("==", 2, [Push 1, Eval, Push 1, Eval, Eq, Update 2, Pop 2, Unwind]),
("~=", 2, [Push 1, Eval, Push 1, Eval, Ne, Update 2, Pop 2, Unwind]),
("<", 2, [Push 1, Eval, Push 1, Eval, Lt, Update 2, Pop 2, Unwind]),
("<=", 2, [Push 1, Eval, Push 1, Eval, Le, Update 2, Pop 2, Unwind]),
```



```
( ">", 2, [Push 1, Eval, Push 1, Eval, Gt, Update 2, Pop 2, Unwind]),  
( ">=", 2, [Push 1, Eval, Push 1, Eval, Ge, Update 2, Pop 2, Unwind]),
```

if 関数は、分岐に Cond を使用するようにコンパイルされます。

```
( "if", 3, [Push 0, Eval, Cond [Push 1] [Push 2],  
           Update 3, Pop 3, Unwind]])
```

演習 3.27 新しい命令とコンパイラが機能することを確認するために、Appendix B のどのテストプログラムを使用しますか?

### 3.7 Mark 5: 算術演算のより良い処理に向けて

現時点での G マシンの実装方法では、各演算子はコンパイルされたプリミティブの 1 つを経由して呼び出されます。この配置を改善するために、しばしば算術演算子を直接呼び出すことができることを観察することができます。例えば、次のような簡単なプログラムを考えてみよう。

```
main = 3+4*5
```

これにより、現在のコンパイラを使用した場合、以下のようなコードが生成されます。

```
[Pushint 5, Pushint 4, Pushglobal "*", Mkap, Mkap,  
Pushint 3, Pushglobal "+", Mkap, Mkap, Eval]
```

このコードを実行すると、33 ステップを要し、11 ヒープノードを使用することになります。最初に思いつくのは、関数 '+' と '\*' の呼び出しの代わりに、Add と Mul という命令を使えばいいのでは、ということでしょう。これは次のような改善されたコードにつながります。

```
[Pushint 5, Pushint 4, Mul, Pushint 3, Add]
```

これは、わずか 5 ステップで実行され、5 つのヒープノードを使用します

#### 3.7.1 問題点

次の例題のプログラムを考えると、考えられる問題が発生します。

```
main = K 1 (1/0)
```

これにより、以下のようなコードが生成されます。

```
[Pushint 0, Pushint 1, Pushglobal "/", Mkap, Mkap,  
Pushint 1, Pushglobal "K", Mkap, Mkap, Eval]
```

前の例のパターンでいくと、以下のようなコードを生成してみることができるかもしれません。

```
[Pushint 0, Pushint 1, Div,  
Pushint 1, Pushglobal "K", Mkap, Mkap, Eval]
```

問題は、K を簡約する前に除算演算子が適用され、その結果、ゼロ除算のエラーが発生することです。正しいコンパイラは、このようなエラーを出さないようなコードを生成しなければなりません。

何が起こったかという、私たちのコードが厳しすぎるのです。その結果、評価する必要のない式が評価され、本来は発生しないはずのエラーが発生しています。同じような問題は、終端でない式が不用意に評価されたときにも起こります。

### 3.7.2 解決策

この問題を解決するには、ある表現が出現する文脈を記録しておくことです。ここでは、2つのコンテキスト<sup>8</sup>を区別します。

Strict 式の値は WHNF で要求されます。

Lazy 式の値は、WHNF で要求されるかもしれないし、されないかもしれません。

各コンテキストに対応するコンパイルスキームがあり、式を G マシンの命令列にコンパイルするスキームがあります。Strict なコンテキストでは、このコンパイルスキームは  $\mathcal{E}$  スキームであり、Lazy なコンテキストでは、すでに見た  $\mathcal{C}$  スキームを使用することになります。

できるだけ多くの Strict なコンテキストを見つけたいです。なぜなら、これらのコンテキストはより良いコードを生成することを可能にするからです。スーパーコンビネータがインスタンス化される時は常に、その値を WHNF で評価したいからです。このことから、スーパーコンビネータの本体は常に Strict なコンテキストで評価することができると結論づけられます。また、式が WHNF で評価されれば、いくつかの部分式が WHNF で評価されることが分かっている式もあります。

Strict なコンテキストの式のクラスは、再帰的に記述することができます。

- スーパーコンビネータ定義の本体中の式は、Strict なコンテキストにある。
- $e_0 \odot e_1$  が Strict なコンテキストで出現し、ここで  $\odot$  が算術演算子または比較演算子である場合、式  $e_0$  と  $e_1$  も Strict なコンテキストにあります。
- もし  $\text{negate } e$  が Strict なコンテキストで発生した場合、 $e$  という式も Strict なコンテキストになる。

---

<sup>8</sup>より多くのコンテキストを区別することが可能です。投影法 [Wadler 1987] と評価変換器 [Burn 1991] はこのための2つの方法である。

- $\text{if } e_0 \ e_1 \ e_2$  という式が Strict なコンテキストで出現する場合、 $e_0, e_1, e_2$  という式も Strict なコンテキストになる。
- もし  $\text{let}(\text{rec}) \ \Delta \ \text{in } e$  が Strict なコンテキストで現れたら、式  $e$  も Strict なコンテキストにあることになります。

例として、次のスーパーコンビネータ  $f$  の本体を考えると、このことがよくわかるでしょう。

$$f \ x \ y = (x+y) + g \ (x*y)$$

$(x+y)$  と  $g \ (x*y)$  はどちらも Strict なコンテキストで評価されます。なぜならスーパーコンビネータの本体は Strict なコンテキストで評価されるからです。最初のケースでは、 $+$ が Strict なコンテキストを伝播するので、 $x$  と  $y$  は Strict なコンテキストで評価されます。2 番目の式では、ユーザー定義のスーパーコンビネータ  $g$  が存在するため、部分式  $x*y$  は評価されないものとしてコンパイルされる。

このことは、Strict コンテキストコンパイラ  $\mathcal{E}$  を再帰的な方法で実装できることを示唆しています。各スーパーコンビネータの本体は Strict コンテキストで評価されるので、 $\mathcal{R}$  スキームから  $\mathcal{E}$  スキームの関数を呼び出す必要があります。これは上記の第一の点を満たしています。また、コンテキスト情報を部分式に伝達するために、算術式に対して  $\mathcal{E}$  スキームを再帰的に呼び出します。

新しいコンパイラのスキームは図 3.12 に定義されています。

$\mathcal{R}[e] \rho d$  は、アリティ  $d$  のスーパーコンビネータに対して、式  $e$  を環境  $\rho$  でインスタンス化し、その結果のスタックを巻き戻すコードを生成します。

$\mathcal{R}[e] \rho d = \mathcal{E}[e] \rho ++ [\text{Update } d, \text{Pop } d, \text{Unwind}]$

$\mathcal{E}[e] \rho$  は、環境  $\rho$  において式  $e$  を WHNF に評価するコードをコンパイルし、スタックの先頭に式へのポインタを残します。

$$\begin{aligned} \mathcal{E}[i] \rho &= [\text{Pushint } i] \\ \mathcal{E}[\text{let } x_1 = e_1; \dots; x_n = e_n \text{ in } e] \rho &= \mathcal{C}[e_1] \rho^{+0} ++ \\ &\quad \dots \\ &\quad \mathcal{C}[e_n] \rho^{+(n-1)} ++ \\ &\quad \mathcal{E}[e] \rho' ++ [\text{Slide } n] \\ &\quad \text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\ \mathcal{E}[\text{letrec } x_1 = e_1; \dots; x_n = e_n \text{ in } e] \rho &= [\text{Alloc } n] ++ \\ &\quad \mathcal{C}[e_1] \rho' ++ [\text{Update } n-1] ++ \\ &\quad \dots \\ &\quad \mathcal{C}[e_n] \rho' ++ [\text{Update } 0] ++ \\ &\quad \mathcal{E}[e] \rho' ++ [\text{Slide } n] \\ &\quad \text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\ \mathcal{E}[e_0 + e_1] \rho &= \mathcal{E}[e_1] \rho ++ \mathcal{E}[e_0] \rho^{+1} ++ [\text{Add}] \\ &\quad \text{他の算術式や比較式についても同様。} \\ \mathcal{E}[\text{negate } e] \rho &= \mathcal{E}[e] \rho ++ [\text{Neg}] \\ \mathcal{E}[\text{if } e_0 \text{ } e_1 \text{ } e_2] \rho &= \mathcal{E}[e_0] \rho ++ [\text{Cond } (\mathcal{E}[e_1] \rho) (\mathcal{E}[e_2] \rho)] \\ \mathcal{E}[e] \rho &= \mathcal{C}[e] \rho ++ [\text{Eval}] \\ &\quad \text{デフォルトケース} \end{aligned}$$

新しいスキームを使ってコンパイルできる組み込み演算子のセットを簡単に拡張できるように、builtInDyadic を採用しました。

```
builtInDyadic :: ASSOC Name Instruction
builtInDyadic
  = [("+", Add), ("-", Sub), ("*", Mul), ("div", Div),
     ("==", Eq), ("~=", Ne), (">=", Ge),
     (">", Gt), ("<=", Le), ("<", Lt)]
```

演習 3.28 既存のコンパイラ関数 compileR と compileE を、図 3.12 の  $\mathcal{R}$  スキームと  $\mathcal{E}$  スキームを実装するように修正します。builtInDyadic を使用する必要があります。

時折、新しいコンパイラを使うとマシンが失敗することがあります。必要

なのは、Unwind 命令に対して新しいルールを導入することです。

(3.29)	$\frac{[\text{Unwind}] \quad [a_0, \dots, a_k] \quad \langle i, s \rangle : d \quad h[a_0 : \text{NGlobal } n \ c] \quad m}{\Rightarrow \quad i \quad a_k : s \quad d \quad h \quad m \ (k < n \text{ のとき})}$
--------	---

これにより、Eval を使って数値だけでなく、あらゆるオブジェクトを WHNF に評価することができるようになりました。

演習 3.29 Unwind の新しい遷移規則を実装してください。新しい Unwind の遷移規則がないと失敗するプログラムを書いてください。

演習 3.30 冒頭で使ったサンプルプログラムの実行を、Mark4 マシンでの実行と比較してみてください。

```
main = 3+4*5
```

Appendix B の他のプログラムも試してみてください。

Strict なコンテキストをコンパイラに実装する方法は、コンパイラの理論から継承される属性の簡単な例である。Strict コンテキストを式の属性と見なすと、部分式はその親となる式から Strict コンテキストを継承する。一般的な理論は [Aho et al. 1986] で述べられている。

残念ながら、ある式を Strict なコンテキストでコンパイルすべきかどうかをコンパイル時に判断することは、一般的には不可能です。したがって、妥協点を受け入れるしかありません。本書では、限られた式の集合である算術式だけを特別な方法で扱いました。この解析をより一般的な式に拡張する研究が盛んに行われています [Burn 1991]。

### 3.8 Mark 6: データ構造の追加

本節では、G マシンを拡張して、任意のデータ構造を扱えるようにします。第 1 章で述べたように、データ構造を含むプログラムには、コンストラクタ `EConstr` と、case 式 `ECase` という 2 つの新しいコア言語コンストラクタが必要です。Mark6 マシンの目標は、これらの式に対応したコードをコンパイルすることです。

#### 3.8.1 概要

1.1.4 節で、タグ `t` とアリティ `a` を持つコンストラクタはコア言語では `Pack{t, a}` と表現されることを見ました。例えば、通常のリストデータ型は 2 つのコンストラクタを持っています。 `Pack{1, 0}` と `Pack{2, 2}` です。これらはそれぞれ Miranda の `[]` と `(:)` に相当します。

Miranda では直接対応するものがない case 式は、コンストラクタに保持される値を検査するために使用されます。例えば、リストの長さを求める関数を次のように書きます。

```
length xs = case xs of
    <1>      -> 0;
    <2> y ys -> 1 + length ys
```

case 式の実行の仕方を見てみると、参考になることがあります。

1. case 式を評価するために、まず `xs` を WHNF に評価します。
2. この評価が行われると、どの選択肢を取るべきかがわかるようになります。評価された式のタグ（構造化データ・オブジェクトでなければなりません）が、どの選択肢を取るかを決定します。上の例では、長さについて
  - `xs` のコンストラクタのタグが 1 の場合、リストは空であり、最初の選択肢を取ります。従って、0 を返します。
  - タグが 2 であれば、リストは空でないことになります。今回はコンストラクタのコンポーネント（`y` と `ys`）があります。リストの長さは `ys` の長さより 1 つ多くなります。

ここでは、コンストラクタを解体しようとするときはいつでも、正しい数の引数に適用されていると仮定します。このような状態のコンストラクタは充足状態であるといえます。例として、セクション 1.1.3 で `Cons` は 2 つの引数を取るように定義されているので、2 つの式に適用されると充足状態になります。

また、コア言語プログラムは、構造化データオブジェクトである結果を返すことができるようになったことにも注目します。Mark5 G マシンは、構造化データオブジェクトを遅延評価スタイルで表示できるようにしなければなりません。まず、Mark5 マシンのデータ構造にどのような追加をする必要があるかを考えてみましょう。

### 3.8.2 データ構造

マシンが単なる数値ではない値を返すことができればいいのです。コンストラクタからなる値を返せるようにしたいのです。そのためには、構造体の構成要素を再帰的に評価し、その値を返す必要があります。そのためには、さらに別のコンポーネント、gmOutput を状態に追加する必要があります。これは、プログラムの結果を保持するためのものです。

```
type GmState =
  (GmOutput, -- Current Output
   GmCode,   -- Current Instruction Stream
   GmStack,  -- Current Stack
   GmDump,   -- The Dump
   GmHeap,   -- Heap of Nodes
   GmGlobals, -- Global addresses in Heap
   GmStats)  -- Statistics
```

このコンポーネントは、文字列として定義されます。

```
type GmOutput = [Char]
```

アクセス関数は、当たり前のようには書けばいいのです。

```
getOutput :: GmState -> GmOutput
getOutput (o, i, stack, dump, heap, globals, stats) = o

putOutput :: GmOutput -> GmState -> GmState
putOutput o' (o, i, stack, dump, heap, globals, stats)
  = (o', i, stack, dump, heap, globals, stats)
```

演習 3.31 残りのアクセス機能を適切に変更します。

ヒープ内のコンストラクタノードをサポートするために、node 型を NConstr で拡張します。これは、タグを表す正の数と、コンポーネントのリストを取ります。コンポーネントのリストはヒープ内のノードのアドレスのリストとして表されます。



```

data Node
  = NNum Int -- Numbers
  | NAp Addr Addr -- Applications
  | NGlobal Int GmCode -- Globals
  | NInd Addr
  | NConstr Int [Addr]
instance Eq Node
  where
    NNum a      == NNum b      = a == b -- needed to check conditions
    NAp a b     == NAp cd      = False  -- not needed
    NGlobal a b == NGlobal c d = False  -- not needed
    NInd a      == NInd b      = False  -- not needed
    NConstr a b == NConstr c d = False  -- not needed

```

### 3.8.3 結果の表示

表示したい新しい状態コンポーネントがあるため、関数 `showState` を再定義する必要があります。

```

showState :: GmState -> Iseq
showState s
  = iConcat [showOutput s,          iNewline,
             showStack s,          iNewline,
             showDump s,           iNewline,
             showInstructions (getCode s), iNewline]

```

出力コンポーネントはすでに文字列であるため、`showOutput` 関数は簡単です。

```

showOutput :: GmState -> Iseq
showOutput s = iConcat [iStr "Output:\n", iStr (getOutput s), iStr "\n"]

```

他の唯一の変更(新しい命令セットの `showInstruction` の変更を除く)は、コンストラクタノードを含めるようにデータ型を拡張したため、`showNode` で発生します。

```

showNode :: GmState -> Addr -> Node -> Iseq
showNode s a (NNum n)      = iNum n
showNode s a (NGlobal n g) = iConcat [iStr "Global ", iStr v]
                                where v = head [n | (n,b) <- getGlobals s, a==b]
showNode s a (NAp a1 a2)   = iConcat [iStr "Ap ", iStr (showaddr a1),
                                       iStr " ", iStr (showaddr a2)]

```

```

showNode s a (NInd a1)      = iConcat [iStr "Ind ", iStr (showaddr a1)]
showNode s a (NConstr t as)
  = iConcat [iStr "Cons ", iNum t, iStr " [",
             iInterleave (iStr ", ") (map (iStr.showaddr) as), iStr "]" ]

```

### 3.8.4 命令セット

新しい命令セットが定義されました。Mark4 マシンに 4 つの新しい命令を追加するだけです。

```

data Instruction
  = Slide Int
  | Alloc Int
  | Update Int
  | Pop Int
  | Unwind
  | Pushglobal Name
  | Pushint Int
  | Push Int
  | Mkap
  | Eval
  | Add | Sub | Mul | Div
  | Neg
  | Eq | Ne | Lt | Le | Gt | Ge
  | Cond GmCode GmCode

```

マシンに追加された 4 つの新しい命令は次のとおりです。

```

  | Pack Int Int
  | Casejump [(Int, GmCode)]
  | Split Int
  | Print

```

**演習 3.32** 新しい命令セットに一致するように showInstruction を拡張します。

Pack 命令は単純です。充足したコンストラクタを構築するのに十分な引数がスタックにあると想定しています。存在する場合は、充足したコンストラクタの作成に進みます。十分な引数がない場合、命令は未定義です。

(3.30)	$  \begin{array}{lcl}  o \text{ Pack } t \ n : i \ a_1 : \dots : a_n : s \ d \ h & & m \\  \Rightarrow o \quad i \quad a : s \ d \ h[a : \text{NConstr } t \ [a_1, \dots, a_n]] & & m  \end{array}  $
--------	---

Casejump の遷移ルールは、

- (a) スタックの一番上のノードが WHNF にあること
- (b) ノードが構造化データ オブジェクトであること

を想定しています。このオブジェクトのタグを使用して、選択枝の命令シーケンスの 1 つを選択すると、現在の命令ストリームの前に、選択された特定の選択枝のコードが追加されます。

(3.31)	$\begin{array}{l} o \text{ Casejump } [\dots, t \rightarrow i', \dots] : i \quad a : s \quad d \quad h[a : \text{NConstr } t \text{ ss} \quad m \\ \Rightarrow \quad o \quad i' ++ i \quad a : s \quad d \quad h \quad m \end{array}$
--------	---

これは、多方向へのジャンプと結合を指定する簡単な方法です。つまり、選択枝コード  $i'$  のコードによって現在のコード  $i$  を事前に修正することで、最初に選択枝のコードを実行し、次に主な式の残りのコードが必要とするもので再開するという効果を達成します。<sup>9</sup>

各選択枝のコードは、Split  $n$  命令で始まり、Slide  $n$  命令で終了します。 $n$  の値は、コンストラクタ内のコンポーネントの数によって決まります。Split 命令は、コンストラクタのコンポーネントにアクセスするために使用されます。

length 関数用に生成されたコードシーケンスを考えてみましょう。

```
[Push 0, Eval,
 Casejump [1 -> [Pushint 0]
           2 -> [Split 2, Push 1, Pushglobal "length", Mkap,
                  Eval, Pushint 1, Add, Slide 2]],
 Update 1,
 Pop 1,
 Unwind]
```

このパターンの実行を図 3.13 に示します。ここでは、現在のローカルバインディングのセットを拡張するために、Slide および Split 命令が一時的に使用されていることがわかります。

<sup>9</sup>Casejump を使用したコードシーケンスはフラットではないことに注意してください。ただし、各選択枝にラベルを付けて、ラベル付けされたコードアドレスにジャンプすることにより、必要なフラットコードシーケンスを構築できます。コード生成が不必要に複雑になるため、これは行っていません。



Figure 3.13: Running compiled code for alternatives

`length` 関数が `nil` 以外のノードに適用されたと仮定すると、`Casejump` 命令を実行するときに、2 というラベルの付いた選択肢が取られます。これが最初の図 (a) です。

`Split 2` 命令は、コンストラクタノードをスタックに「unpack」します。これを図 (b) に示します。

選択肢の本体、つまりコードシーケンス

```
[Push 1, Pushglobal "length", Mkap, Eval, Pushint 1, Add]
```

を完成させた後、この `length` の呼び出しに対するリスト引数の長さがスタックの一番上になります。図 (c) で 1 とラベル付けされています。

実行を完了するには、`head` と `tail` へのポインタを削除します。これを図 (d) に示します。

`Split` の遷移規則は簡単です。

(3.32)	$o$	<code>Split <math>n : i</math></code>	$a : s$	$d$	$h[a : \text{NConstr } t [a_1, \dots, a_n]]$	$m$
$\Rightarrow$	$o$	$i$	$a_1 : \dots : a_n : s$	$d$	$h$	$m$

次に、`Print` の遷移規則について説明します。`Print` には 2 つの遷移規則があります。コンストラクタと数値にそれぞれ 1 つずつです。

(3.33)	$o$	<code>Print : <math>i</math></code>	$a : s$	$d$	$h[a : \text{NNum } n]$	$m$
$\Rightarrow$	$o ++ [n]$	$i$	$s$	$d$	$h$	$m$

コンストラクタのルールは、コンストラクタの各コンポーネントをプリントするように調整する必要があるため、より複雑です。簡単にするために、コンポーネントのみをプリントします。

(3.34)	$\begin{array}{c} o \text{ Print} : i \quad a : s \quad d \quad h[a : \text{NConstr } t [a_1, \dots, a_n]] \quad m \\ \Rightarrow o \quad i' ++ i \quad a_1 : \dots : a_n : s \quad d \quad h \quad m \end{array}$
--------	--

コード  $i'$  は次のとおりです。

$$\underbrace{[\text{Eval}, \text{Print}, \dots, \text{Eval}, \text{Print}]}_n$$

最後に、Unwind の新しいルールを追加する必要があります。これは、NNum のルールと同様に、NConstr を巻き戻すときに戻るように指示します。

(3.35)	$\begin{array}{c} o \text{ [Unwind]} \quad a : s \quad \langle i', s' \rangle : d \quad h[a : \text{NConstr } n \text{ as}] \quad m \\ \Rightarrow o \quad i' \quad a : s' \quad d \quad h \quad m \end{array}$
--------	---

演習 3.33 新しい遷移規則を実装し、dispatch 関数を変更します。

### 3.8.5 コンパイラ

図 3.14 では、 $\mathcal{E}$  および  $\mathcal{C}$  コンパイルスキームの新しいケースが示されています。

$\mathcal{E}[[e]] \rho$ は、式 $e$ を環境 $\rho$ の WHNF に評価するコードをコンパイルし、式へのポインタをスタックの一番上に残します。
$\begin{aligned} \mathcal{E}[\text{case } e \text{ of } \text{alts}] \rho &= \mathcal{E}[[e]] \rho ++ [\text{Casejump } \mathcal{D}[\text{alts}] \rho] \\ \mathcal{E}[\text{Pack}\{t, a\} e_1 \dots e_a] \rho &= \mathcal{C}[[e_a]] \rho^{+0} ++ \dots ++ \mathcal{C}[[e_1]] \rho^{+(a-1)} ++ [\text{Pack } t \ a] \end{aligned}$
$\mathcal{C}[[e]] \rho$ は、環境 $\rho$ で $e$ のグラフを構築するコードを生成し、それへのポインタをスタックの一番上に残します。
$\mathcal{C}[\text{Pack}\{t, a\} e_1 \dots e_a] \rho = \mathcal{C}[[e_a]] \rho^{+0} ++ \dots ++ \mathcal{C}[[e_1]] \rho^{+(a-1)} ++ [\text{Pack } t \ a]$
$\mathcal{D}[\text{alts}] \rho$ は、case 式の選択枝のリストをコンパイルします。
$\mathcal{D}[\text{alt}_1 \dots \text{alt}_n] \rho = [\mathcal{A}[\text{alt}_1] \rho, \dots, \mathcal{A}[\text{alt}_n] \rho]$
$\mathcal{A}[\text{alt}] \rho$ は、case 式の各選択枝のコードをコンパイルします。
$\mathcal{A}[\langle t \rangle x_1 \dots x_n \rightarrow \text{body}] \rho = t \rightarrow [\text{Split } n] ++ \mathcal{E}[\text{body}] \rho' ++ [\text{Slide } n]$ <p style="text-align: center;">ここで <math>\rho' = \rho^{+n} [x_1 \mapsto 0, \dots, x_n \mapsto n-1]</math> です。</p>

case 式で選択される可能性のある選択枝を処理するために、補助的なコンパイルスキーム  $\mathcal{D}$  および  $\mathcal{A}$  が必要です。関数 `compileAlts` ( $\mathcal{D}$  スキームに対応) は、現在の環境を使用して選択枝のリストをコンパイルし、タグ付き

コードシーケンスのリストを生成します。また、`comp` 引数 ( $A$  に対応) を使用して、各選択肢の本体をコンパイルします。現時点では、この引数は常に `compileE'` になります。

```
compileAlts :: (Int -> GmCompiler) -- compiler for alternative bodies
            -> [CoreAlt]           -- the list of alternatives
            -> GmEnvironment      -- the current environment
            -> [(Int, GmCode)]    -- list of alternative code sequences

compileAlts comp alts env
  = [(tag, comp
        (length names)
        body
        (zip names [0..] ++ argOffset (length names) env))
    | (tag, names, body) <- alts]
```

`compileE'` スキームは、`compileE` スキームを少し変更したものです。通常の `compileE` スキームによって生成されたコードの周りに `Split` と `Slide` を配置するだけです。

```
compileE' :: Int -> GmCompiler
compileE' offset expr env
  = [Split offset] ++ compileE expr env ++ [Slide offset]
```

演習 3.34 関連する変更を `compile` に加え、`initialCode` を変更して最後に `Print` 命令を追加します。

演習 3.35 新しいケースをコンパイラ関数 `compileE` および `compileC` に追加します。

演習 3.36 出力を「構造化された形式」で出力するには、どのような変更が必要ですか。これは、整数だけでなく、コンストラクタと括弧を出力コンポーネント `gmOutput` に配置することを意味します。

### 3.8.6 比較における新しい論理表現の使用

このセクションでは、構築した `Mark6` マシンを変更して、ブール値の新しい表現を使用する方法を示します。最初に、ブール値を構造化データオブジェクトとして実装できることを確認します。`True` と `False` は、アリティがゼロでタグが 2 と 1 のコンストラクタとして表されます。

条件式をどのように実装しますか？ これは、`if` のプログラムに新しい定義を追加することで実行できます。最初の引数に応じて、2 番目または 3 番目の引数のいずれかを返します。

```

if c t f = case c of
    <1> -> f;
    <2> -> t

```

必要な最初の変更は、比較操作にあります。これらには、次の一般的な遷移規則があります。

(3.36)	$\frac{o \odot i \quad a_0 : a_1 : s \quad d \quad h[a_0 : \text{NNum } n_0, \quad a_1 : \text{NNum } n_1] \quad m}{\Rightarrow o \quad i \quad a : s \quad d \quad h[a : \text{Constr } (n_0 \odot n_1) \quad []] \quad m}$
--------	--

たとえば、Eq 命令では、2 つの数値  $n_0$  と  $n_1$  が同じ場合は 2(True のタグ) を返し、そうでない場合は 1(False のタグ) を返す関数に置き換えます。

Mark4 マシン用に開発したコードの一部を再利用することで、遷移規則をすばやく実装できます。セクション 3.6.3 では、一般的な算術演算子と比較演算子を表す方法を見てきました。実際、比較関数の定義を構築した方法により、ブール値の新しい表現をほとんどすぐに使用できます。

ボックス化関数 `boxBoolean` は、比較演算と、スタックの一番上に 2 つの整数がある状態を取ります。スタックの一番上にある 2 つの整数を比較したブール値の結果がある新しい状態を返します。

```

boxBoolean :: Bool -> GmState -> GmState
boxBoolean b state
    = putStack (a: getStack state) (putHeap h' state)
      where (h',a) = hAlloc (getHeap state) (NConstr b' [])
            b' | b = 2          -- 2 is tag of True
                | otherwise = 1 -- 1 is tag of False

```

**演習 3.37** Appendix B のサンプルプログラムをいくつか実行します。たとえば、階乗プログラムを試してください。

```

fac n = if (n==0) 1 (n * fac (n-1))

```

### 3.8.7 使用可能な言語の拡大

賢明な読者はお気づきかもしれませんが、コンパイラが失敗する `ECase` と `EConstr` に関するいくつかの正当な表現があります。コンパイルできない法的な表現は、次の 2 つのクラスに分類されます。

1. Strict でないコンテキスト (つまり、 $\mathcal{C}$  スキームによってコンパイルされた式) での ECase の出現。
2. 適用される引数が少なすぎる式での EConstr の出現。

どちらの問題も、プログラム変換手法を使用して解決できます。ECase の解決策は、問題のある式をスーパーコンビネータにして、自由変数に適用することです。たとえば、次のプログラム

```
f x = Pack{2,2} (case x of <1> -> 1; <2> -> 2) Pack{1,0}
```

は以下の同等のプログラムに変換できます。

```
f x = Pack{2,2} (g x) Pack{1,0}
g x = case x of <1> -> 1; <2> -> 2
```

EConstr の解決策は、コンストラクタごとにスーパーコンビネータを作成することです。これは、コンストラクタを充足させるのに十分な自由変数で生成されます。ここに例があります。

```
prefix p xs = map (Pack{2,2} p) xs
```

これは次のように変換されます。

```
prefix p xs = map (f p) xs
f p x = Pack{2,2} p x
```

この問題を解決する別の方法は、Pushglobal 命令を変更して、「Pack{t,a}」という形式の名前を持つ関数に対して機能するようにすることです。そうすれば、上記の例の f のようなコンストラクタ関数を、状態のグローバルコンポーネントから探すだけでよいのです。関数がまだ存在しない場合は、その関数に関連付ける新しいグローバルノードを作成することができます。このノードは特に単純な構造持っているからです。

```
NGlobal a [Pack t a, Update 0, Unwind]
```

新しい遷移規則は、最初に、関数が既に存在する場合:

(3.37)	$\begin{array}{l} o \quad \text{Pushglobal Pack}\{t,n\} : i \quad s \quad d \quad h \quad m[\text{Pack}\{t,n\} : a] \\ \Rightarrow \quad o \quad \quad \quad \quad \quad i \quad a : s \quad d \quad h \quad m \end{array}$
--------	---

第二に、それがまだ存在しない場合:

(3.38)	$\begin{array}{l} o \quad \text{Pushglobal Pack}\{t,n\} : i \quad s \quad d \quad h \quad m \\ \Rightarrow \quad o \quad \quad \quad \quad \quad i \quad a : s \quad d \quad h[a : \text{gNode}_{t,n}] \quad m[\text{Pack}\{t,n\} : a] \end{array}$
--------	---



ここで  $\text{gNode}_{t,n}$  は以下の通りです。

$$\text{NGlobal } n \text{ [Pack } t \text{ } n, \text{ Update } 0, \text{ Unwind}]$$

その後、コンパイラは、未充足のコンストラクタノードを含む式のコードを直接生成できます。これは、不充足コンストラクタに対して次のコードを生成することによって行われます。

$$\mathcal{C}[\text{Pack}\{t, a\}] \rho = [\text{Pushglobal } \text{"Pack}\{t, a\}"]$$

演習 3.38 `Pushglobal` 命令の `pushglobal` 関数に拡張機能を実装し、コンパイラを変更します。

### 3.9 Mark 7: さらなる改善

Mark 5 コンパイラの開発時に見たサンプルプログラムをもう一度考えてみましょう。

```
main = 3+4*5
```

これにより、Mark 6 コンパイラを使用した場合、以下のようなコードが生成されます。

```
[Pushint 5, Pushint 4, Mul, Pushint 3, Add]
```

このコードを実行すると、ヒープノードが1つ使用されます。これをさらに減らすことは可能でしょうか？

答えはイエスです。演算の中間値を表す数値のスタックを使用することで、演算のためのヒープアクセス数をさらに減らすことができます。Mark 7 マシンでは、これらの値はVスタックと呼ばれる新しいステートコンポーネントに保持されます。問題は、実機では数値をヒープに入れたり取り出したりするのは高くつく操作であることです。マシンのレジスタセットやスタックを利用する方がはるかに効率的です。Mark 7 G マシンではスタックを使うので、レジスタが足りなくなる心配はありません。

#### プログラム

```
main = 3+4*5
```

の新しいコードは、前回生成したコードと非常によく似ています。

```
[Pushbasic 5, Pushbasic 4, Mul, Pushbasic 3, Add, Mkint]
```

最初の命令 `Pushbasic 5` は、5 を V スタックの一番上に置きます。次に 4 を V スタックにプッシュし、その後に乗算を行います。この命令は、その引数が V スタックにあることを想定しています。これは、同様に V スタックに答えを配置します。次の 2 つの命令は、V スタックの一番上にある値に 3 を足します。最後の命令 `Mkint` は、V スタックの一番上の値を取って、ヒープ内の数値ノードに入れ、この新しいノードへのポインタを S スタックの一番上に残します。

#### 3.9.1 V スタックによる階乗関数の実行

まず、Mark7 マシンを例にとって調べてみます。ここでは、次のように定義された階乗関数の実行を見ることにします。

```
fac n = if (n==0) 1 (n * fac (n-1))
```

Mark 7 コンパイラを使って、スーパーコンピネータ本体に以下のようなコード列を生成します。

```
[Pushbasic 0, Push 0, Eval, Get, Eq,
Cond [Pushint 1, Update 1, Pop 1, Unwind]
      [Pushint 1, Push 1, Pushglobal "-", Mkap, Mkap, Pushglobal "fac", Mkap,
        Eval, Get, Push 0, Eval, Get, Mul, Mkint, Update 1, Pop 1, Unwind]
```

このコードが実行されるとき、V スタックは空であり、通常のスタックには 1 つのアイテムがあります。この 2 種類のスタックを区別するために、以後後者を S スタックと呼ぶことにします。




Figure 3.15: Mark 7 machine running `fac`

図 3.15 の (a) 図から、S スタックの上に `fac` の引数へのポインタがある初期状態を見ることができます。図 (b) では、`Pushbasic` 命令が実行されたときに、整数が V スタックにプッシュされる様子を示しています。図 (c) では、`fac` への引数が評価され、図 (d) では、`Get` 命令の効果が示されています。ヒープ内のノードから値を取り出し、V スタックに格納したものです。




Figure 3.16: Mark 7 machine running `fac`

図(e)(図 3.16) は、Eq 命令が実行された後の状態を示しています。V スタックの2つの項目を比較した結果、両者が等しくないことを発見しました。Mark 7 G-machine は、V スタックのブール値 False を 1 として表現しています。図(f) では、Cond 命令がこの値を検査し、どの分岐を実行するかを選択しています。




Figure 3.17: Mark 7 machine running fac

図(g)(図 3.17) は fac (1-1) を構築し評価した後の状態です。次の命令は、新しく評価された値を V スタックにフェッチする Get です。図(i) では、ノード 1 から評価し、V スタックに値をフェッチしていることがわかります。図(j) では、Mul 命令によって V スタック内の 2 つの値が乗算され、その結果がスタックに戻されています。図(k) では、Mkint 命令により、この結果を V スタックからヒープに移動し、S スタックに新しく生成されたノードのアドレスを記録しています。

ヒープにオブジェクトを作成しアクセスすることが、スタックにオブジェクトを保持することと比較して性能上のペナルティがあるマシンでは、V-stack の使用は改善されると期待されます。V スタックがどのように機能するかを見た後、Mark 7 マシンを実装するために G マシンに小さな修正を加えます。

### 3.9.2 データ構造

V スタックを使用することで、G マシンの各状態には、新しい状態コンポーネント gmVStack がその状態に追加されます。

```

type GmState = (GmOutput, -- Current output
                GmCode,    -- Current instruction stream
                GmStak,    -- Current stack
                GmDump,    -- Current dump
                GmVStack,  -- Current V-stack
                GmHeap,    -- Heap of nodes
                GmGlobals, -- Global addresses in heap
                GmStats)   -- Statistics

```

すでに述べたように、この新しいコンポーネントは数のスタックとして動作します。

```

type GmVStack = [Int]

```

このコンポーネントのアクセス関数を追加します。

```

getVStack :: GmState -> GmVStack
getVStack (o, i, stack, dump, vstack, heap, globals, stats) = vstack

```

```

putVStack :: GmVStack -> GmState -> GmState
putVStack vstack' (o, i, stack, dump, vstack, heap, globals, stats)
    = (o, i, stack, dump, vstack', heap, globals, stats)

```

演習 3.39 その他のアクセス関数を変更します。

状態を表示する

関数 `showState` は、V-stack コンポーネントを表示するように変更されています。

```

showState :: GmState -> Iseq
showState s
    = iConcat [showOutput s, iNewline,
               showStack s, iNewline,
               showDump s, iNewline,
               showVStack s, iNewline,
               showInstructions (getCode s), iNewline]

```

そのために、関数 `showVStack` を使用します。

```

showVStack :: GmState -> Iseq
showVStack s
    = iConcat [iStr "Vstack:[",
               iInterleave (iStr ", ") (map iNum (getVStack s)),
               iStr "]" ]

```

### 3.9.3 命令セット

最初の要件は、各算術演算の遷移が、通常のスタックの代わりに、V スタックから値を取得し、その結果を返すように修正されなければならないことであることは明らかです。まず、ダイアディックプリミティブの場合を考えてみましょう。この操作  $\odot$  の一般的な遷移は以下の通りです。これは V-stack から 2 つの引数を取り、演算  $\odot$  の結果を V-stack に戻します。

(3.39)	$\begin{array}{l} o \quad \odot : i \quad s \quad d \quad n_0 : n_1 : v \quad h \quad m \\ \Rightarrow o \quad i \quad s \quad d \quad n_0 \odot n_1 : v \quad h \quad m \end{array}$
--------	---

Neg 命令は、V スタックの先頭の数とその-1 倍に置き換えるだけで、次のように遷移します。

(3.40)	$\begin{array}{l} o \quad \text{Neg} : i \quad s \quad d \quad n : v \quad h \quad m \\ \Rightarrow o \quad i \quad s \quad d \quad (-n) : v \quad h \quad m \end{array}$
--------	---

また、ヒープと V スタック間で値を移動させる命令も必要です。まず、整数  $n$  を V スタックにプッシュする Pushbasic から始めます。

(3.41)	$\begin{array}{l} o \quad \text{Pushbasic } n : i \quad s \quad d \quad v \quad h \quad m \\ \Rightarrow o \quad i \quad s \quad d \quad n : v \quad h \quad m \end{array}$
--------	---

V スタックからヒープに値を移動するには、2 つの命令を使用します。Mkbool と Mkint です。これらは、V スタックの上にある整数をそれぞれブール値と整数値として扱います。まず、Mkbool から。

(3.42)	$\begin{array}{l} o \quad \text{Mkbool} : i \quad s \quad d \quad t : v \quad h \quad m \\ \Rightarrow o \quad i \quad a : s \quad d \quad v \quad h[a : \text{NConstr } t \quad \square] \quad m \end{array}$
--------	--

Mkint の遷移規則は、ヒープに新しい整数ノードを作成することを除けば、Mkbool に似ています。

(3.43)	$\begin{array}{l} o \quad \text{Mkint} : i \quad s \quad d \quad n : v \quad h \quad m \\ \Rightarrow o \quad i \quad a : s \quad d \quad v \quad h[a : \text{NNum } n] \quad m \end{array}$
--------	--

逆演算を行うには、Get を使用します。これは 2 つの遷移規則で定義されています。最初の遷移規則では、Get がスタックの一番上にあるブール値をどのように扱うかを見えています。

(3.44)	$\begin{array}{c} o \quad \text{Get} : i \quad a : s \quad d \quad v \quad h[a : \text{NConstr } t \quad \square] \quad m \\ \Rightarrow \quad o \quad i \quad s \quad d \quad t : v \quad h \quad m \end{array}$
--------	---

2 つ目は、Get が数をどのように扱うかを見ることです。

(3.45)	$\begin{array}{c} o \quad \text{Get} : i \quad a : s \quad d \quad v \quad h[a : \text{NNum } n] \quad m \\ \Rightarrow \quad o \quad i \quad s \quad d \quad n : v \quad h \quad m \end{array}$
--------	--

最後に、V スタック上のブーリアン演算を利用するために、V スタックを検査してどの命令ストリームを使用するかを決定する単純化された Casejump 命令を使用します。この新しい命令は Cond と呼ばれ、次の 2 つの遷移規則で定義されます。最初の遷移規則では、V-stack の一番上の値が true のとき、最初のコードシーケンス  $t$  を選択します。

(3.46)	$\begin{array}{c} o \quad \text{Cond } t \quad f : i \quad s \quad d \quad 2 : v \quad h \quad m \\ \Rightarrow \quad o \quad t ++ i \quad s \quad d \quad v \quad h \quad m \end{array}$
--------	---

2 番目の遷移規則では、V スタックの最上位の値が false であるため、Cond はその 2 番目の符号列  $f$  を選択することがわかります。

(3.47)	$\begin{array}{c} o \quad \text{Cond } t \quad f : i \quad s \quad d \quad 1 : v \quad h \quad m \\ \Rightarrow \quad o \quad f ++ i \quad s \quad d \quad v \quad h \quad m \end{array}$
--------	---

**演習 3.40** 命令データ型の拡張、showInstruction 関数の再定義、新しい命令遷移の実装、dispatch 関数の修正。

次に、コンパイラの改造（かなり大掛かりなもの）について考えてみましょう。

### 3.9.4 コンパイラ

状態コンポーネントが追加されたため、compile 関数は V スタックコンポーネントを空に初期化する必要があります。

```
compile :: CoreProgram -> GmState
compile program
  = ([], initialCode, [], [], [], heap, globals, statInitial)
  where (heap, globals) = buildInitialHeap program
```

厳密に言えば、これだけでマシンは動くのですが、V スタックを導入したのは、算術関数を「インライン」でコンパイルできるようにするためで、私たちのコードはこれを意図しているのです。

```

buildInitialHeap :: CoreProgram -> (GmHeap, GmGlobals)
buildInitialHeap program
  = mapAccuml allocateSc hInitial compiled
    where compiled = map compileSc (preludeDefs ++ program ++ primitives)

```

プリミティブ命令の遷移規則を変更したため、コンパイルしたプリミティブごとにコードを変更する必要があります。Mark 6 マシンで行ったように、このコードを手作業でコンパイルする代わりに、コンパイラにこの仕事を任せることができます。もちろんこれは、コンパイラが生成するコードを最適化できるほど賢ければの話ですが、そうでなければ、Add 命令は生成されません。

```

primitives :: [(Name,[Name],CoreExpr)]
primitives
  = [( "+", ["x","y"], (EAp (EAp (EVar "+") (EVar "x")) (EVar "y"))),
      ("-", ["x","y"], (EAp (EAp (EVar "-") (EVar "x")) (EVar "y"))),
      ("*", ["x","y"], (EAp (EAp (EVar "*") (EVar "x")) (EVar "y"))),
      ("/", ["x","y"], (EAp (EAp (EVar "/") (EVar "x")) (EVar "y"))),

```

また、ネゲーション関数を追加する必要があります。

```

("negate", ["x"], (EAp (EVar "negate") (EVar "x"))),

```

比較関数は、二項演算関数とほぼ同じです。

```

("==", ["x","y"], (EAp (EAp (EVar "==") (EVar "x")) (EVar "y"))),
("~= ", ["x","y"], (EAp (EAp (EVar "~= ") (EVar "x")) (EVar "y"))),
(">=", ["x","y"], (EAp (EAp (EVar ">=") (EVar "x")) (EVar "y"))),
(">", ["x","y"], (EAp (EAp (EVar ">") (EVar "x")) (EVar "y"))),
("<=", ["x","y"], (EAp (EAp (EVar "<=") (EVar "x")) (EVar "y"))),
("<", ["x","y"], (EAp (EAp (EVar "<") (EVar "x")) (EVar "y"))),

```

最後に、条件付き関数と、ブール値を表現するスーパーコンビネータを入れます。

```

("if", ["c","t","f"],
  (EAp (EAp (EAp (EVar "if") (EVar "c")) (EVar "t")) (EVar "f"))),
("True", [], (EConstr 2 0)),
("False", [], (EConstr 1 0))]

```

コンパイルスキーム  $\mathcal{B}$



$\mathcal{B}[\![e]\!] \rho$  は、環境  $\rho$  において式  $e$  を WHNF に評価するコードをコンパイルし、結果を V スタックに残します。

$$\begin{aligned}
\mathcal{B}[\![i]\!] \rho &= [\text{Pushbasic } i] \\
\mathcal{B}[\![\text{let } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho &= \mathcal{C}[\![e_1]\!] \rho^{+0} ++ \\
&\dots \\
&\mathcal{C}[\![e_n]\!] \rho^{+(n-1)} ++ \\
&\mathcal{B}[\![e]\!] \rho' ++ [\text{Pop } n] \\
&\text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\
\mathcal{B}[\![\text{letrec } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho &= [\text{Alloc } n] ++ \\
&\mathcal{C}[\![e_1]\!] \rho' ++ [\text{Update } n-1] ++ \\
&\dots \\
&\mathcal{C}[\![e_n]\!] \rho' ++ [\text{Update } 0] ++ \\
&\mathcal{B}[\![e]\!] \rho' ++ [\text{Pop } n] \\
&\text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\
\mathcal{B}[\![e_0 + e_1]\!] \rho &= \mathcal{B}[\![e_1]\!] \rho ++ \mathcal{B}[\![e_0]\!] \rho^{+1} ++ [\text{Add}] \\
&\text{他の算術式についても同様。} \\
\mathcal{B}[\![e_0 == e_1]\!] \rho &= \mathcal{B}[\![e_1]\!] \rho ++ \mathcal{B}[\![e_0]\!] \rho^{+1} ++ [\text{Eq}] \\
&\text{他の比較式についても同様。} \\
\mathcal{B}[\![\text{negate } e]\!] \rho &= \mathcal{B}[\![e]\!] \rho ++ [\text{Neg}] \\
\mathcal{B}[\![\text{if } e_0 \ e_1 \ e_2]\!] \rho &= \mathcal{B}[\![e_0]\!] \rho ++ [\text{Cond } (\mathcal{B}[\![e_1]\!] \rho) (\mathcal{B}[\![e_2]\!] \rho)] \\
\mathcal{B}[\![e]\!] \rho &= \mathcal{E}[\![e]\!] \rho ++ [\text{Get}] \\
&\text{デフォルトケース}
\end{aligned}$$

図 3.18 に示す  $\mathcal{B}$  スキームは、別のタイプのコンテキストを構成しています。 $\mathcal{B}$  スキームによってコンパイルされるには、式は WHNF への評価が必要であることが分かっているだけでなく、整数型かブール型の式でなければなりません。以下の式は  $\mathcal{B}$  スキームを介して伝搬されます。

- $\text{let}(\text{rec}) \Delta \text{ in } e$  が  $\mathcal{B}$ -strict コンテキストで現れた場合、式  $e$  も  $\mathcal{B}$ -strict コンテキストになります。
- $\text{if } e_0 \ e_1 \ e_2$  という式が  $\mathcal{B}$ -strict コンテキストで現れたら、 $e_0, e_1, e_2$  という式も  $\mathcal{B}$ -strict コンテキストで現れることになります。
- $e_0 \odot e_1$  が  $\mathcal{B}$ -strict のコンテキストで出現し、 $\odot$  が比較または算術演算子である場合、式  $e_0$  と  $e_1$  も  $\mathcal{B}$ -strict のコンテキストで出現します。
- もし  $\text{negate } e$  が  $\mathcal{B}$ -strict のコンテキストで発生した場合、式  $e$  も同様となる。

式の特種なケースを認識できない場合、コンパイルされたコードは  $\mathcal{E}$  スキームを使用して式を評価し、その後 `Get` 命令を実行します。`Get` 命令は、 $\mathcal{E}$  スキームによってスタックの一番上に残された値をアンボックスし、 $V$  スタックに移動します。

このため、ある式が最初は  $\mathcal{B}$ -strict コンテキストにあることをどうやって知るかは未解決でした。通常の状態では、通常の  $\mathcal{E}$ -strict なコンテキストから、値が整数型またはブーリアン型であるという追加の知識とともに、 $\mathcal{B}$ -strict なコンテキストを生成します。

演習 3.41  $\mathcal{B}$  コンパイラスキームを実装する。

コンパイラスキーム  $\mathcal{E}$

$\mathcal{E}[\![e]\!] \rho$  は、式  $e$  を環境  $\rho$  の WHNF に評価するコードをコンパイルし、式へのポインタをスタックの一番上に残します。

$$\begin{aligned}
\mathcal{E}[\![i]\!] \rho &= [\text{Pushint } i] \\
\mathcal{E}[\![\text{let } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho &= \mathcal{C}[\![e_1]\!] \rho^{+0} ++ \\
&\dots \\
&\mathcal{C}[\![e_n]\!] \rho^{+(n-1)} ++ \\
&\mathcal{E}[\![e]\!] \rho' ++ [\text{Slide } n] \\
&\text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\
\mathcal{E}[\![\text{letrec } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho &= [\text{Alloc } n] ++ \\
&\mathcal{C}[\![e_1]\!] \rho' ++ [\text{Update } n-1] ++ \\
&\dots \\
&\mathcal{C}[\![e_n]\!] \rho' ++ [\text{Update } 0] ++ \\
&\mathcal{E}[\![e]\!] \rho' ++ [\text{Slide } n] \\
&\text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n-1, \dots, x_n \mapsto 0] \text{ です。} \\
\mathcal{E}[\![\text{case } e \text{ of } alts]\!] \rho &= \mathcal{E}[\![e]\!] \rho ++ [\text{Casejump } \mathcal{D}_{\mathcal{E}}[\![alts]\!] \rho] \\
\mathcal{E}[\![\text{Pack}\{t, a\} \ e_1 \dots e_a]\!] \rho &= \mathcal{C}[\![e_a]\!] \rho ++ \\
&\dots \\
&\mathcal{C}[\![e_1]\!] \rho ++ \\
&[\text{Pack } t \ a] \\
\mathcal{E}[\![e_0 + e_1]\!] \rho &= \mathcal{B}[\![e_0 + e_1]\!] \rho ++ [\text{Mkint}] \\
&\text{他の算術式についても同様。} \\
\mathcal{E}[\![e_0 == e_1]\!] \rho &= \mathcal{B}[\![e_0 == e_1]\!] \rho ++ [\text{Mkbool}] \\
&\text{他の比較式についても同様。} \\
\mathcal{E}[\![\text{negate } e]\!] \rho &= \mathcal{B}[\![\text{negate } e]\!] \rho ++ [\text{Mkint}] \\
\mathcal{E}[\![\text{if } e_0 \ e_1 \ e_2]\!] \rho &= \mathcal{B}[\![e_0]\!] \rho ++ [\text{Cond } (\mathcal{E}[\![e_1]\!] \rho) (\mathcal{E}[\![e_2]\!] \rho)] \\
\mathcal{E}[\![e]\!] \rho &= \mathcal{C}[\![e]\!] \rho ++ [\text{Eval}] \\
&\text{デフォルトケース}
\end{aligned}$$

$\mathcal{D}_{\mathcal{E}}[\![alts]\!] \rho \ d$  は、case 式の選択枝のリストをコンパイルします。

$$\mathcal{D}_{\mathcal{E}}[\![alt_1 \dots alt_n]\!] \rho \ d = [\mathcal{A}_{\mathcal{E}}[\![alt_1]\!] \rho \ d, \dots, \mathcal{A}_{\mathcal{E}}[\![alt_n]\!] \rho \ d]$$

$\mathcal{A}_{\mathcal{E}}[\![alt]\!] \rho \ d$  は、case 式の各選択枝のコードをコンパイルします。

$$\begin{aligned}
\mathcal{A}_{\mathcal{E}}[\![\langle t \rangle \ x_1 \dots x_n \rightarrow body]\!] \rho \ d &= t \rightarrow [\text{Split } n] ++ \mathcal{E}[\![body]\!] \rho' ++ [\text{Slide } n] \\
&\text{ここで } \rho' = \rho^{+n} [x_1 \mapsto 0, \dots, x_n \mapsto n-1] \text{ です。}
\end{aligned}$$

$\mathcal{E}$  スキーム (図 3.19) は、算術関数と比較関数の呼び出しを特別扱いすることを定めたものです。Mark6 マシンで使ったバージョンと異なるのは、

算術演算と比較演算に  $B$  スキームを使用するためです。また、特殊なケースがない場合、式をコンパイルするためにデフォルトのメソッドを使用しなければなりません。これは、 $C$  スキームを用いてグラフを構築し、コードストリームに `Eval` 命令を配置するだけです。これにより、グラフが WHNF に評価されることが保証されます。

演習 3.42 新しい  $\mathcal{E}$  コンパイラスキームを実装する。

コンパイルスキーム  $\mathcal{R}$

また、この機会に  $\mathcal{R}$  スキームを改善します。まず、 $B$  スキームが使われる機会を作りたいと考えています。また、関数のコード列の末尾で実行される命令の数を減らすことも試みています。  $\mathcal{R}$  スキームの新しいコンパイルスキームを図 3.20 に示します。

$\mathcal{R}[\![e]\!] \rho d$  は、アリティ  $d$  のスーパーコンビネータのために、  
環境  $\rho$  で式  $e$  をインスタンス化するコードを生成し、  
その後、結果のスタックの巻き戻しに進みます。

$$\begin{aligned}\mathcal{R}[\![\text{let } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho d \\ &= \mathcal{C}[\![e_1]\!] \rho^{+0} ++ \\ &\quad \dots \\ &\quad \mathcal{C}[\![e_n]\!] \rho^{+(n-1)} ++ \\ &\quad \mathcal{R}[\![e]\!] \rho' (n + d) \\ &\quad \text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n - 1, \dots, x_n \mapsto 0] \text{ です。}\end{aligned}$$

$$\begin{aligned}\mathcal{R}[\![\text{letrec } x_1 = e_1; \dots; x_n = e_n \text{ in } e]\!] \rho d \\ &= [\text{Alloc } n] ++ \\ &\quad \mathcal{C}[\![e_1]\!] \rho' ++ [\text{Update } n - 1] ++ \\ &\quad \dots \\ &\quad \mathcal{C}[\![e_n]\!] \rho' ++ [\text{Update } 0] ++ \\ &\quad \mathcal{R}[\![e]\!] \rho' (n + d) \\ &\quad \text{ここで } \rho' = \rho^{+n} [x_1 \mapsto n - 1, \dots, x_n \mapsto 0] \text{ です。}\end{aligned}$$

$$\begin{aligned}\mathcal{R}[\![\text{if } e_0 \ e_1 \ e_2]\!] \rho d &= \mathcal{B}[\![e_0]\!] \rho ++ [\text{Cond } [\mathcal{R}[\![e_1]\!] \rho d, \mathcal{R}[\![e_2]\!] \rho d]] \\ \mathcal{R}[\![\text{case } e \text{ of } \text{alts}]\!] \rho d &= \mathcal{E}[\![e]\!] \rho ++ [\text{Casejump } \mathcal{D}_{\mathcal{R}}[\![\text{alts}]\!] \rho d] \\ \mathcal{R}[\![e]\!] \rho d &= \mathcal{E}[\![e]\!] \rho ++ [\text{Update } d, \text{Pop } d, \text{Unwind}] \\ &\quad \text{デフォルトケース}\end{aligned}$$

$\mathcal{D}_{\mathcal{R}}[\![\text{alts}]\!] \rho d$  は、case 式の選択肢のリストをコンパイルします。

$$\mathcal{D}_{\mathcal{R}}[\![\text{alt}_1 \dots \text{alt}_n]\!] \rho = [\mathcal{A}_{\mathcal{R}}[\![\text{alt}_1]\!] \rho d, \dots, \mathcal{A}_{\mathcal{R}}[\![\text{alt}_n]\!] \rho d]$$

$\mathcal{A}_{\mathcal{R}}[\![\text{alt}]\!] \rho d$  は、case 式の各選択肢のコードをコンパイルします。

$$\begin{aligned}\mathcal{A}_{\mathcal{R}}[\![\langle t \rangle x_1 \dots x_n \rightarrow \text{body}]\!] \rho d &= t \rightarrow [\text{Split } n] ++ \mathcal{R}[\![\text{body}]\!] \rho' (n + d) \\ &\quad \text{ここで } \rho' = \rho^{+n} [x_1 \mapsto 0, \dots, x_n \mapsto n - 1] \text{ です。}\end{aligned}$$

Mark6 マシンで使ったバージョンから拡張され、コンテキストのように動作するようになりました。このコンテキストを  $\mathcal{R}\text{-strict}$  と呼ぶことにします。 $\mathcal{R}\text{-strict}$  コンテキストでコンパイルされる式は WHNF に評価され、現在の  $\text{redex}$  を上書きするために使われます。これは以下の伝搬のルールに従います。

- スーパーコンビネータ定義の本体である式は、 $\mathcal{R}\text{-strict}$  のコンテキストにあります。
- もし  $\text{let}(\text{rec}) \ \Delta \text{ in } e$  が  $\mathcal{R}\text{-strict}$  コンテキストで出現したら、式  $e$  も  $\mathcal{R}\text{-strict}$  コンテキストになります。
- もし式  $\text{if } e_0 \ e_1 \ e_2$  が  $\mathcal{R}\text{-strict}$  のコンテキストで現れたら、式  $e_1$  と  $e_2$

も  $\mathcal{R}$ -strict のコンテキストにある。(式  $e_0$  は今度は  $\mathcal{B}$ -strict のコンテキストに現れる)。

- もし  $\text{case } e \text{ of } alts$  が  $\mathcal{R}$ -strict コンテキストで出現するならば、 $e$  という式は strict なコンテキストにあることになる。さらに、各選択肢の式部分は  $\mathcal{R}$ -strict コンテキストで出現する。

演習 3.43 図 3.20 の  $\mathcal{R}$  スキームを実装してください。compileAlts 関数の一般性を利用して、 $\mathcal{A}_{\mathcal{R}}$  スキームと  $\mathcal{A}_{\mathcal{E}}$  スキームの両方を実装できることに注意してください。

Mark7 マシンについて特筆すべき点は、現在の V スタックをダンプに保存する Eval 命令を定義していないことです。これは、[Peyton Jones 1987] で説明された G マシンとは対照的です。これを行うかどうかは、本書で作成した抽象的なマシンでは、本当に好みの問題です。実際のマシン用にコードをコンパイルする場合、我々はスタックの数を最小にしようとすることが多いでしょう。このような場合、Eval の代替遷移規則として次のようなものを使いたいです。

(3.48)	$\begin{array}{ccccccc} o & \text{Eval} : i & a : s & & d & v & h \quad m \\ \Rightarrow & o & [\text{Unwind}] & [a] & \langle i, s, v \rangle : d & [] & h \quad m \end{array}$
--------	--

演習 3.44 Eval のためにこの代替遷移規則を実装します。この新しい遷移規則を可能にするために変更する必要がある他の命令は何ですか？

演習 3.45 ブール演算子  $\&$ ,  $|$ , not の最適化されたコードを生成するために、どのようなコンパイルルールを変更しますか？

演習 3.46 以下の遷移規則による Return 命令の追加。

(3.49)	$\begin{array}{ccccccc} o & [\text{Return}] & [a_0, \dots, a_k] & \langle i, s \rangle : d & v & h & m \\ \Rightarrow & o & i & a_k : s & & d & v \quad h \quad m \end{array}$
--------	--

上記の遷移規則では、WHNF にあることが分かっているスタックの一番上のアイテムを捨ててしまうことになる。

以下の様な遷移規則になるのでは？

(3.49)	$\begin{array}{ccccccc} o & [\text{Return}] & [a_0, \dots, a_k] & \langle i, s \rangle : d & v & h & m \\ \Rightarrow & o & i & a_0 : s & & d & v \quad h \quad m \end{array}$
--------	--

これは、スタックの一番上のアイテムが WHNF にあることが分かっているときに、 $\lambda$  スキームによって Unwind の代わりに使用されます。この新しい命令を生成するために compileR を修正します。

演習 3.47 UpdateInt  $n$  の遷移規則を書いてください。この命令はシーケンス [Mkint, Update  $n$ ] と同じアクションを実行します。この遷移規則と UpdateBool  $n$  のための同様の遷移規則を実装してください。compileR を修正して、元のシーケンスの代わりにこれらの命令を生成するようにしてください。なぜ新しい命令が元のシーケンスより好ましいのでしょうか？(ヒント：いくつかのサンプルプログラムからの統計値を利用すること)

### 3.10 結論

この章で採用したアプローチは、大きなソフトウェアを設計する際に非常に役立ちます。最初は非常に単純なものから始め、徐々に変更を加えることで、非常に大規模で複雑なソフトウェアを作成します。これは、小さな漸進的な変更のプロセスによって常に可能であると主張するのは誤解を招く可能性があります。実際、Mark1 マシンの一部として提示された材料は、Mark7 マシンを念頭に置いて特別に設計されました。

しかし、現時点では、かなり効率的なマシンを製造しています。次の章では、TIM について説明します。



```
module Tim where
import Utils
import Language
```

## 4 TIM: 3 命令マシン

3つの命令マシンであるTIMは、最初は、これまでに見たものとは非常に異なる形のリダクションマシンのように見えます。それでも、一連の比較的単純な手順でGマシンをTIMに変換できることがわかりました。この章では、これらの手順について説明し、それによってTIMがどのように機能するかを示し、完全な最小TIMコンパイラと評価器を定義します。そして、それに対する一連の改善と最適化を開発します。

TIMはFairbairnとWrayによって発明されたもので、彼らの元の論文[Fairbairn and Wray 1987]は読む価値があります。この章で採用したアプローチとはまったく異なる方法でTIMを説明します。ただし、この章で作成された資料は、FairbairnとWrayの研究をはるかに超えているため、あまり知られていないアイデアが議論され、実装されている後のセクションで、詳細レベルが高められています。提示された新しいアイデアの多くはGuy Argoによるものであり、彼のFPCA論文[Argo 1989]と博士論文[Argo 1991]に提示されています。

### 4.1 背景: TIMの仕組み

次の関数定義を検討してください。

$$f\ x\ y = g\ E1\ E2$$

ここで、E1とE2は任意の(そしておそらく複雑な)式であり、gはその他の関数です。テンプレートインスタンス化マシン(第2章)とGマシン(第3章)の両方が、次の簡約を実行します。

@	reduces to	@
/ \		/ \
@ y		@ E2
/ \		/ \
f x		g E1

Gマシンはこれを行うためにかなりの数の(単純な)命令を必要としますが、テンプレートマシンは1つの(複雑な)ステップでそれを行います。ですが、最終的な結果は同じです。

この図で、E1とE2は式E1とE2のグラフです。たとえば、E1が $(x+y)*(x-y)$ の場合、gの最初の引数は $(x+y)*(x-y)$ のグラフになります。このグラフは、(Cコンパイルスキームによって生成されたコードによって)ヒープに苦労して構築する必要があります。残念ながら、gは最初の引数を使用せずに

破棄する可能性があるため、これは無駄な作業になる可能性があります。関数への引数に対して行われるグラフ作成の量を制限する何らかの方法を見つけないと考えています。

#### 4.1.1 平坦化

変換のステップ 1 はまさにこれを行います。f の定義を次の新しい定義に置き換えるとします。

```
f x y = g (c1 x y) (c2 x y)
c1 x y = E1
c2 x y = E2
```

私たちは、c1 と c2 という 2 つの補助関数を発明しました。この定義は明らかに古いものと同等ですが、E1 がどれほど大きく複雑であっても、f の簡約中に行われる唯一の作業は、(c1 x y) のグラフを作成することです。

さらに良いことに、G マシンの実装には、自動的に得られるさらなる利点があります。最初の定義では、E1 は C スキームによってコンパイルされます。算術式をコンパイルするときに、E スキームに存在する最適化を利用することはできません。しかし、2 番目の定義では、式 E1 はスーパーコンビネーターの右辺になり、これらすべての最適化が適用されます。この方法で  $(x+y)*(x-y)$  をより効率的に評価できます。

もちろん、E1 と E2 自体に C スキームでコンパイルされる大きな式が含まれている可能性があるため（たとえば、E2 が  $(h\ E3\ E4)$  であるとします）、c1 と c2 の右辺に変換を再度適用する必要があります。その結果、ネストされた構造を持つ式がないため、平坦化されたプログラムと呼ばれます。

#### 4.1.2 タプル化

次の注目点は、c1 と c2 の両方が x と y の両方に適用されることです。そのため、g を呼び出す前に (c1 x y) と (c2 x y) のグラフを作成する必要があります。c1 と c2 に引数が 2 つだけではなく多数ある場合、グラフはかなり大きくなる可能性があります。2 つのグラフは互いに非常に似ているため、これらの引数グラフがいくつかの共通部分を共有できるかどうかを尋ねるのは自然なことです。重複を避け、それによってヒープ割り当てを減らします。このアイデアは、2 番目の変換で表現できます。


```
f x y = let tup = (x,y)
        in g (c1 tup) (c2 tup)
c1 (x,y) = E1
c2 (x,y) = E2
```

最初にその引数をタプルにパッケージ化し、次にこの単一のタプルを  $c1$  と  $c2$  に渡すという考え方です。 $f$  をこのように定義すると、 $f$  の簡約は次のようになります。



#### 4.1.3 スパインレスであること

前の図を見ると、スーパーコンビネータとタプル  $tup$  の場合、スパインが指す引数は常に  $(c \text{ } tup)$  の形式であることがわかります。簡約中に、これらの引数へのポインタのスタックを構築します。しかし、それらはすべて同じ形式になっているので、代わりに引数 (のルート) 自体を積み重ねることができます! したがって、 $f$  簡約後、スタックは次のようになります。



スパインスタックの各アイテムは、コードポインタとタプルへのポインタのペアになりました。このペアは、タプルに適用される関数を定義するコードである関数適用ノードと考えることができます。 $f$  へのエントリでは、引数  $x$  と  $y$  (のルート) がスタック上にあったため、 $x$  と  $y$  のタプルは実際にはコードポインタ/タプルポインタのペアのタプルです。

コードポインタ/タプルポインタのペアはクロージャと呼ばれ、そのようなクロージャのタプルはフレームと呼ばれます。フレームへのポインタをフレームポインタと呼びます。ヒープにスパインがなくなっていることに注意してください。スタックは、評価される式のスパインです。TIM はスパインレスマシンです。

#### 4.1.4 例

TIM プログラムがどのように機能するか例を示します。次のように定義された関数 `compose2` を考えてみましょう。

```
compose2 f g x = f (g x x)
```

`compose2` の「平坦化された」形式は次のようになります。

```
compose2 f g x = f (c1 g x)
```

```
c1 g x = g x x
```

`compose2` に入ると、次のように 3 つの引数がスタックの一番上に置かれます。

```

      |          |          |
      |-----|
x | x-code | x-frm |
      |-----|
g | g-code | g-frm |
      |-----|
f | f-code | f-frm |
      |-----|

```

最初に行うことは、これら 3 つの引数のタプル (フレーム) をヒープに形成することです。その後、それらをスタックから削除できます。フレームポインタと呼ばれる特別なレジスタに新しいフレームへのポインタを保持します。これは、命令によって行われます。

Take 3

マシンの状態は次のようになります。

```

      |          |          |
      |-----|
                                     -----
Frame ptr -----> f | f-code | f-frm |

```

```

|-----|
g | g-code | g-frm |
|-----|
x | x-code | x-frm |
-----

```

次に、f の引数を準備する必要があります。(g x x) という 1 つのみがあり、そのクロージャをスタックにプッシュします。クロージャのフレームポインタは現在のフレームポインタレジスタにすぎないため、命令はコードラベルを提供するだけで済みます。

```
Push (Label "c1")
```

最後に、f にジャンプします。f はグローバルなスーパーコンビネータではなく、compose の引数であるため、f は、現在のフレームのクロージャによって表されます。私たちがしなければならないことは、クロージャをフェッチし、そのフレームポインタをフレームポインタレジスタにロードし、そのコードポインタをプログラムカウンタにロードすることです。これは次の命令によって行われます。

```
Enter (Arg 1) -- f is argument 1
```

この命令の後、マシンの状態は次のようになります。

```

|      |      | | | |
|-----|      |-----|
| c1    | ----|-----> f | f-code| f-frm |
|-----|      |-----|
|      |      |      g | g-code| g-frm |
|-----|      |-----|
Frame ptr:  f-frm      |-----|
Program ctr: f-code    x | x-code| x-frm |
|-----|

```

それだ！compose2 の本体は、次の 3 つの命令だけで構成されています。

```

compose2:  Take 3          -- 3 arguments
           Push (Label "c1") -- closure for (g x x)
           Enter (Arg 1)    -- f is argument 1

```

ただし、ラベル c1 を処理する必要があります。(g x x) のクロージャが必要な場合は、Enter 命令で入力されるため、プログラムカウンタは c1 を指し、f、g、および x を含む元のフレームへのフレームポインタ。この時点で必要なことは、g の引数、つまり x を準備し、g を入力することだけです。

```

c1:      Push (Arg 3)          -- x is argument 3
        Push (Arg 3)          -- x again
        Enter (Arg 2)         -- g is argument 2

```

Push (Arg 3) 命令は、現在のフレームから  $x$  のクロージャのコピーをフェッチし、それをスタックにプッシュします。次に、Enter (Arg 2) 命令が、現在スタックにある引数<sup>10</sup>に  $g$  を適用します。

#### 4.1.5 状態遷移規則によるマシンの定義

3 つの命令マシンと呼ばれる理由がわかります。Take、Push、Enter の 3 つの主要な命令があります。いくつかの点で、3 つの命令しかないと主張するのはかなり楽観的です。なぜなら、Push と Enter には両方ともいくつかの「アドレス指定モード」があり、さらに、やがてかなりの数の新しい命令を発明する必要があるからです。それにしても素敵な名前ですね。

いつものように、状態遷移規則を使用して、各命令の正確な作用を表現します。まず、マシンの状態を定義する必要があります。これは、

*(instructions, framepointer, stack, heap, codestore)*

または略して  $(i, f, s, h, c)$  の五つ組です。コードストアは、まだ説明されていない唯一の項目です。コードのコレクションが含まれており、それぞれにラベルが付いています。実際には、コードストアにはコンパイル済みのスーパーコンピネータの定義が含まれており、それぞれにスーパーコンピネータの名前が付けられていますが、原則として、有用であることが判明した場合は、他のラベル付けされたコードフラグメントも含めることができます。

次に、各命令の遷移規則を作成します。Take  $n$  は、スタックの上位  $n$  個の要素を新しいフレームに形成し、現在のフレームポインタがそれを指すようにします。

(4.1)	$\text{Take } n : i \quad f \quad c_1 : \dots : c_n : s \quad h \quad c$
	$\Rightarrow \quad i \quad f' \quad s \quad h[f' : \langle c_1, \dots, c_n \rangle] \quad c$

次に、Push と Enter のルールについて説明します。これらの 2 つの命令はまったく同じアドレッシングモード (Arg、Label など) を持ち、それらの間には非常に明確な関係があります。

<sup>10</sup>( $g \ x \ x$ ) クロージャに入ったときにスタックが空でなかった場合、2 つ以上になる可能性があります。

Push/Enter の関係。Push *arg* 命令がクロージャ (*i*, *f*) をスタックにプッシュする場合、Enter *arg* は *i* をプログラムカウンタにロードし、*f* を現在のフレームポインタにロードします。

Push (Arg *n*) 命令は、現在のフレームから *n* 番目のクロージャをフェッチし、それをスタックにプッシュします。

(4.2)	$\text{Push (Arg } k) : i \quad f \quad s \quad h[f : \langle (i_1, f_1), \dots, (i_k, f_k), \dots, (i_n, f_n) \rangle] \quad c$
	$\Rightarrow i \quad f \quad (i_k, f_k) : s \quad h \quad c$

Push (Label *l*) は、コードストアでラベル *l* を検索し、このコードポインタと現在のフレームポインタで構成されるクロージャをプッシュします。

(4.3)	$\text{Push (Label } l) : i \quad f \quad s \quad h \quad c[l : i']$
	$\Rightarrow i \quad f \quad (i', f) : s \quad h \quad c$

compose の例では、任意のラベル *c1* を作成する必要がありました。これらのラベルを考え出さなければならないのは面倒です。代わりに、プッシュ命令の新しい形式、Push (Code *i*) を追加するだけです。これにより、ターゲットコードシーケンス *i* が命令自体の一部になります。したがって、

Push (Label "c1")

の代わりに

Push (Code [Push (Arg 3), Push (Arg 3), Enter (Arg 2)])

と書くことが出来ます。適切な状態遷移規則は次のとおりです。

(4.4)	$\text{Push (Code } i') : i \quad f \quad s \quad h \quad c$
	$\Rightarrow i \quad f \quad (i', f) : s \quad h \quad c$

これまでのところ、Arg、Code、Label の 3 つの「アドレス指定モード」があります。整数定数用に IntConst をもう 1 つ追加する必要があります。たとえば、呼び出し (f 6) は次のコードにコンパイルされます。

Push (IntConst 6)

Enter (Label "f")

Push 命令は常にクロージャ (つまり、コードポインタ/フレームポインタのペア) をスタックにプッシュしますが、整数定数の場合は、どのクロージャをプッシュするかはまったく明確ではありません。整数自体を格納する場所が必要なので、その目的のためにフレームポインタスロットを「スチール」し



ましょう。<sup>11</sup>この決定は、次の規則につながります。ここで、intCode は整数クロージャの (まだ未定の) コードシーケンスです。

(4.5)	$\text{Push (IntConst } n) : i \quad f \quad s \quad h \quad c$ $\Rightarrow \quad i \quad f \quad (\text{intCode}, n) : s \quad h \quad c$
-------	---

intCode は何をすべきか? 今のところ、私たちのマシンは演算を行わないので、簡単な解決策は intCode を空のコードシーケンスにすることです。

intCode = [] -- Mark1 TIM を写経する際は、これを忘れずに!

整数クロージャが入力されると、マシンは空のコードシーケンスにジャンプし、実行が停止します。これにより、Mark 1 には十分な整数を返すプログラムを書くことができます。

Push 命令については以上です。Enter 命令のルール (アドレス指定ごとに 1 つ) モードでは、Push/Enter 関係から直接に従います。

(4.6)	$[\text{Enter (Label } l)] \quad f \quad s \quad h \quad c[l : i]$ $\Rightarrow \quad i \quad f \quad s \quad h \quad c$
-------	--

(4.7)	$[\text{Enter (Arg } k)] \quad f \quad s \quad h[f : \langle (i_1, f_1), \dots, (i_k, f_k), \dots, (i_n, f_n) \rangle] \quad c$ $\Rightarrow \quad i_k \quad f_k \quad s \quad h \quad c$
-------	---

(4.8)	$[\text{Enter (Code } i)] \quad f \quad s \quad h \quad c$ $\Rightarrow \quad i \quad f \quad s \quad h \quad c$
-------	--

(4.9)	$[\text{Enter (IntConst } n)] \quad f \quad s \quad h \quad c$ $\Rightarrow \quad \text{intCode} \quad n \quad s \quad h \quad c$
-------	---

#### 4.1.6 コンパイル

これで、各 TIM 命令が何を行うかについて正確な説明ができました。プログラムを TIM 命令に変換する方法を説明する必要があります。これは、以前と同様に、一連のコンパイルスキームを使用して行います。各スーパーコンビネーターは、図 4.1 に示す SC スキームでコンパイルされます。SC に渡される初期環境は、各スーパーコンビネータ名をそのラベルアドレッシングモードにバインドします。SC スキームは単に Take 命令を生成し、R スキームを呼び出して、各引数に使用するアドレッシングモードを指定するバインドによって強化された環境を渡します。

<sup>11</sup>整数はフレームポインタより大きくないという暗黙の仮定を行っていますが、これは通常実際には当てはまります。

$SC[e] \rho$ は、環境 $\rho$ におけるスーパーコンビネータ定義 $def$ の TIM コードです。	
$SC[f \ x_1 \ \dots \ x_n = e] \rho = \text{Take } n : \mathcal{R}[e] \rho[x_1 \mapsto \text{Arg } 1, \dots, x_n \mapsto \text{Arg } n]$	
$\mathcal{R}[e] \rho$ は、環境 $\rho$ の式 $e$ の値をスタック上の引数に適用する TIM コードです。	
$\mathcal{R}[e_1 \ e_2] \rho = \text{Push } (\mathcal{A}[e_2] \rho) : \mathcal{R}[e_1] \rho$	
$\mathcal{R}[a] \rho = \text{Enter } (\mathcal{A}[a] \rho)$	
ここで、 $a$ は整数、変数、またはスーパーコンビネータです。	
$\mathcal{A}[e] \rho$ は、環境 $\rho$ における式 $e$ の TIM アドレッシングモードです。	
$\mathcal{A}[x] \rho = \rho \ x$	ここで、 $x$ は $\rho$ によって束縛されます。
$\mathcal{A}[n] \rho = \text{IntConst } n$	ここで、 $n$ は整数です。
$\mathcal{A}[e] \rho = \text{Code } (\mathcal{R}[e] \rho)$	その他

$\mathcal{R}$  スキーム (図 4.1) は、変数またはスーパーコンビネータが見つかるまで、単純に引数をスタックにプッシュします。 $\mathcal{A}$  方式を使用して正しいアドレッシングモードを生成します。セクション 4.1.1 で説明した平坦化プロセスがこれらのルールによって「オンザフライ」で実行されることに注意してください。

ここでは、算術、データ構造、ケース分析、および `let(rec)` 式を省略します。それらはすべて後で追加されます。

#### 4.1.7 更新

これまでのところ、更新についての言及はありません。これは、スパインが消滅したため、更新するスパインノードがないためです。実際、これまで説明してきたマシンはツリー簡約マシンです。共有引数は繰り返し評価される場合があります。更新を適切に行うことは、スパインレス実装のアキレス腱です。そうしないと無制限の量の作業が複製される可能性があるため、これは絶対に必要ですが、更新しないバージョンの優雅さと速度 (複製はさておき) の一部を失う複雑さを追加します。

後でセクション 4.5 で更新に戻りますが、それまでは非更新バージョンを実装するだけで十分です。

## 4.2 Mark 1: 最小限の TIM

このセクションでは、演算、データ構造、または更新を使用しない、最小限で完全な TIM 実装を開発します。これらは、後続のセクションで追加されます。

### 4.2.1 全体構造

構造は、テンプレートインスタンス化インタープリタとほとんど同じです。run 関数は、従来通り、parse、compile、eval、showResults の 4 つの関数で構成されています。解析のタイプは第 1 章で与えられます。他の 3 つのタイプを以下に示します。

```
runProg      :: [Char] -> [Char]
compile      :: CoreProgram -> TimState
eval         :: TimState -> [TimState]
showResults  :: [TimState] -> [Char]
```

```
runProg = showResults . eval . compile . parse
```

多くの場合、すべての中間状態を表示すると便利のため、showFullResults を使用して各状態を表示する fullRun も提供します。

```
fullRun :: [Char] -> [Char]
fullRun = showFullResults . eval . compile . parse
```

language モジュールをインポートする必要があります。

### 4.2.2 データ型の定義

TIM 命令のデータ型は、これまでに紹介した命令に直接対応しています。

```
data Instruction = Take Int
                 | Enter TimAMode
                 | Push  TimAMode
```

アドレッシングモードのタイプである timAMode は、Push と Enter の関係を強調するために、別個のデータタイプとして分離されています。

```
data TimAMode = Arg      Int
               | Label   [Char]
               | Code     [Instruction]
               | IntConst Int
```

TIM マシンの状態は、次の定義によって与えられます。

```
type TimState = ([Instruction], -- The current instruction stream
                FramePtr,      -- Address of current frame
                TimStack,       -- Stack of arguments
                TimValueStack,  -- Value stack (not used yet)
                TimDump,        -- Dump (not used yet)
                TimHeap,        -- Heap of frames
                CodeStore,      -- Labelled blocks of code
                TimStats)       -- Statistics
```

値スタックとダンプは、この章の後半でのみ必要になりますが、それらのブレースホルダーをすぐに追加する方が便利です。

これらの各コンポーネントの表現を順番に検討します。

- 現在の命令ストリームは、命令のリストで表されます。実際のマシンでは、これはプログラムメモリと共にプログラムカウンタになります。
- 通常、フレームポインタはヒープ内のフレームのアドレスですが、他に2つの可能性があります。整数値を保持するために使用されるか、初期化されていない可能性があります。マシンは、これら3つの可能性のどれを期待するかを常に「認識」していますが、`framePtr` の代数データ型を使用してそれらを区別する実装が便利です。

```
data FramePtr = FrameAddr Addr -- The address of a frame
              | FrameInt Int    -- An integer value
              | FrameNull       -- Uninitialised
```

これを行わないと、アドレスを数値として使用しようとすると、Miranda は (正当に) 型エラーを報告します。さらに、初期化されていない状態の `FrameNull` のコンストラクタがあるということは、初期化されていない値を有効なアドレスとして誤って使用しようとした場合に、インタープリタが検出することを意味します。

- スタックにはクロージャが含まれており、それぞれがコード ポインタとフレーム ポインタを含むペアです。スタックをリストとして表します。

```
type TimStack = [Closure]
type Closure = ([Instruction], FramePtr)
```

- 値スタックとダンプは最初からまったく使用されていないため、nullary コンストラクターが 1 つだけあるダミーの代数データ型でそれぞれを表します。後で、これらの定義をより興味深いものに置き換えます。

```
data TimValueStack = DummyTimValueStack
data TimDump = DummyTimDump
```

- ヒープにはフレームが含まれており、それぞれがクロージャのタプルです。フレームのデータ型は、独自の抽象データ型に値するほど重要です。

```
type TimHeap = Heap Frame

fAlloc  :: TimHeap -> [Closure] -> (TimHeap, FramePtr)
fGet    :: TimHeap -> FramePtr -> Int -> Closure
fUpdate :: TimHeap -> FramePtr -> Int -> Closure -> TimHeap
fList   :: Frame -> [Closure] -- Used when printing
```

これらの操作により、フレームを構築し、コンポーネントを抽出して更新できます。fGet と fUpdate のために、fAlloc に与えられるリストの最初の要素には 1 の番号が付けられます。これは、リストに基づく簡単な実装です。

```
type Frame = [Closure]

fAlloc heap xs = (heap', FrameAddr addr)
  where
    (heap', addr) = hAlloc heap xs

fGet heap (FrameAddr addr) n = f !! (n-1)
  where
    f = hLookup heap addr

fUpdate heap (FrameAddr addr) n closure
  = hUpdate heap addr new_frame
  where
    frame = hLookup heap addr
    new_frame = take (n-1) frame ++ [closure] ++ drop n frame
```

```
fList f = f
```

- ラベルごとに、コードストアは対応するコンパイル済みコードを提供します。

```
type CodeStore = ASSOC Name [Instruction]
```

この機会に、失敗した場合にエラーメッセージを生成するラベルのルックアップ関数を提供します。

```
codeLookup :: CodeStore -> Name -> [Instruction]
codeLookup cstore l
    = aLookup cstore l (error ("Attempt to jump to unknown label "
                               ++ show l))
```

- いつものように、統計を簡単に追加できる抽象データ型にします。

```
statInitial :: TimStats
statIncSteps :: TimStats -> TimStats
statGetSteps :: TimStats -> Int
```

ステップ数のみをカウントする最初の実装はかなり単純です。

```
type TimStats = Int -- The number of steps
statInitial = 0
statIncSteps s = s+1
statGetSteps s = s
```

最後に、ヒープとスタックのコードが必要です。

```
-- :a util.lhs -- heap data type and other library functions
```

#### 4.2.3 プログラムのコンパイル

`compile` は、与えられたプログラムから初期マシン状態を作成する、テンプレートインスタンス化のコンパイラと非常によく似た働きをします。主な違いは、コンパイル関数 `compileSC` にあります。これは各スーパーコンビネータに適用されます。

```

compile program
= ([Enter (Label "main")], -- Initial instructions
  FrameNull,               -- Null frame pointer
  initialArgStack,         -- Argument stack
  initialValueStack,       -- Value stack
  initialDump,             -- Dump
  hInitial,                -- Empty heap
  compiled_code,           -- Compiled code for supercombinators
  statInitial)             -- Initial statistics
where
  sc_defs = preludeDefs ++ program
  compiled_sc_defs = map (compileSC initial_env) sc_defs
  compiled_code = compiled_sc_defs ++ compiledPrimitives
  initial_env = [(name, Label name) | (name, args, body) <- sc_defs] ++
    [(name, Label name) | (name, code) <- compiledPrimitives]

```

今のところ、引数スタックは空に初期化されています。

```
initialArgStack = []
```

今のところ、値スタックとダンプはダミー値に初期化されています。後でこれらの定義を変更します。

```

initialValueStack = DummyTimValueStack
initialDump = DummyTimDump

```

コンパイルされたスーパーコンビネータ `compiled_sc_defs` は、`compileSC` を使用して、プログラム内の各スーパーコンビネータをコンパイルすることによって取得されます。`compileSC` に渡される初期環境は、各スーパーコンビネータに適したアドレス指定モードを提供します。コードストア、`compiled_code` は、`compiled_sc_defs` と `compiledPrimitives` を組み合わせることによって取得されます。後者は、組み込みのプリミティブ用にコンパイルされたコードを含むことを目的としていますが、現時点では空です:

```
compiledPrimitives = []
```

テンプレートマシンや G マシンとは異なり、初期ヒープは空です。これらの場合に空でない初期ヒープの理由は、CAF(つまり、引数のないスーパーコンビネータ - セクション 2.1.6) の共有を保持するためでした。TIM マシンのこの初期バージョンでは、CAF 用にコンパイルされた TIM コードが呼び出されるたびに実行されるため、CAF を評価する作業は共有されません。この問題については、後でセクション 4.7 で説明します。

コンパイラの核心は、コンパイルスキーム  $SC$ 、 $R$ 、および  $A$  をそれぞれ関数 `compileSC`、`compileR`、および `compileA` に直接変換することです。環境  $\rho$  は、名前をアドレッシングモードにバインドする連想リストによって表されます。G マシンのコンパイラは、名前からスタックオフセットへのマッピングを使用していましたが、アドレッシングモードを使用することで得られる特別な柔軟性はかなり有用であることがわかりました。

```
type TimCompilerEnv = [(Name, TimAMode)]
```

これで `compileSC` を定義する準備が整いました:

```
compileSC :: TimCompilerEnv -> CoreScDefn -> (Name, [Instruction])
compileSC env (name, args, body)
  = (name, Take (length args) : instructions)
  where
    instructions = compileR body new_env
    new_env = (zip2 args (map Arg [1..])) ++ env
```

`compileR` は式と環境を受け取り、命令のリストを提供します。

```
compileR :: CoreExpr -> TimCompilerEnv -> [Instruction]
compileR (EAp e1 e2) env = Push (compileA e2 env) : compileR e1 env
compileR (EVar v)      env = [Enter (compileA (EVar v) env)]
compileR (ENum n)      env = [Enter (compileA (ENum n) env)]
compileR e              env = error "compileR: can't do this yet"

compileA :: CoreExpr -> TimCompilerEnv -> TimAMode
compileA (EVar v) env = aLookup env v (error ("Unknown variable " ++ v))
compileA (ENum n) env = IntConst n
compileA e env = Code (compileR e env)
```

#### 4.2.4 評価器

次に、評価器が実際にどのように機能するかを定義する必要があります。`eval` の定義は、テンプレートインスタンス化マシンとまったく同じです。

```
eval state
  = state : rest_states
  where
    rest_states | timFinal state = []
                | otherwise      = eval next_state
    next_state = doAdmin (step state)
```



```
doAdmin state = applyToStats statIncSteps state
```

timFinal 関数は、状態が最終状態であることを示します。Stop 命令を発明することもできますが、コードシーケンスが空の場合に終了したと言うのは簡単です。

```
timFinal ([], frame, stack, vstack, dump, heap, cstore, stats) = True
timFinal state                                                    = False
```

applyToStats 関数は、状態の統計コンポーネントに関数を適用するだけです。

```
applyToStats stats_fun (instr, frame, stack, vstack,
                        dump, heap, cstore, stats)
    = (instr, frame, stack, vstack, dump, heap, cstore, stats_fun stats)
```

ステップを進める

step は、単一の命令を受け取って実行するケース分析を行います。Take 式は、対応する状態遷移規則 (4.1) を単純に訳したものです。

```
step ((Take n:instr), fptr, stack, vstack, dump, heap, cstore, stats)
    | length stack >= n = (instr, fptr', drop n stack, vstack, dump,
                        heap', cstore, stats)
    | otherwise         = error "Too few args for Take instruction"
    where (heap', fptr') = fAlloc heap (take n stack)
```

Enter と Push の方程式は、timAMode をクロージャに変換する共通関数 amToClosure を使用して、Push/Enter の関係を利用します。

```
step ([Enter am], fptr, stack, vstack, dump, heap, cstore, stats)
    = (instr', fptr', stack, vstack, dump, heap, cstore, stats)
    where (instr', fptr') = amToClosure am fptr heap cstore
```

```
step ((Push am:instr), fptr, stack, vstack, dump, heap, cstore, stats)
    = (instr, fptr, amToClosure am fptr heap cstore : stack,
        vstack, dump, heap, cstore, stats)
```

amToClosure は、最初の引数であるアドレッシングモードによってアドレス指定されたクロージャを提供します。

```

amToClosure :: TimAMode -> FramePtr -> TimHeap -> CodeStore -> Closure
amToClosure (Arg n)      fptr heap cstore = fGet heap fptr n
amToClosure (Code il)    fptr heap cstore = (il, fptr)
amToClosure (Label l)    fptr heap cstore = (codeLookup cstore l, fptr)
amToClosure (IntConst n) fptr heap cstore = (intCode, FrameInt n)

```

#### 4.2.5 結果の表示

テンプレートインスタンス化バージョンと同様に、結果を適切な方法で出力するには、かなり退屈な関数のコレクションが必要です。スーパーコンビネータの定義を出力すると便利な場合が多いため、最初の状態の定義を使用して、`showResults` を出力することから始めます。

```

showFullResults states
= iDisplay (iConcat [
    iStr "Supercombinator definitions", iNewline, iNewline,
    showSCDefns first_state, iNewline, iNewline,
    iStr "State transitions", iNewline,
    iLayn (map showState states), iNewline, iNewline,
    showStats (last states)
])
where
    (first_state:rest_states) = states

```

`showResults` は、最後の状態といくつかの統計を表示するだけです。

```

showResults states
= iDisplay (iConcat [
    showState last_state, iNewline, iNewline, showStats last_state
])
where last_state = last states

```

残りの機能は簡単です。`showSCDefns` は、各スーパーコンビネータのコードを表示します。

```

showSCDefns :: TimState -> Iseq
showSCDefns (instr, fptr, stack, vstack, dump, heap, cstore, stats)
    = iInterleave iNewline (map showSC cstore)

showSC :: (Name, [Instruction]) -> Iseq
showSC (name, il)
    = iConcat [

```

```

        iStr "Code for ", iStr name, iStr ":", iNewline,
        iStr " ", showInstructions Full il, iNewline, iNewline
    ]

```

showState は、TIM マシンの状態を表示します。

```

showState :: TimState -> Iseq
showState (instr, fptr, stack, vstack, dump, heap, cstore, stats)
    = iConcat [
        iStr "Code: ", showInstructions Terse instr, iNewline,
        showFrame heap fptr,
        showStack stack,
        showValueStack vstack,
        showDump dump,
        iNewline
    ]

```

showFrame は状態のフレーム コンポーネントを表示し、showClosure を使用してその内部の各クロージャを表示します。

```

showFrame :: TimHeap -> FramePtr -> Iseq
showFrame heap FrameNull = iStr "Null frame ptr" `iAppend` iNewline
showFrame heap (FrameAddr addr)
    = iConcat [
        iStr "Frame: <",
        iIndent (iInterleave iNewline
            (map showClosure (fList (hLookup heap addr))))),
        iStr ">", iNewline
    ]
showFrame heap (FrameInt n)
    = iConcat [ iStr "Frame ptr (int): ", iNum n, iNewline ]

```

showStack は引数スタックを表示し、showClosure を使用して各クロージャを表示します。

```

showStack :: TimStack -> Iseq
showStack stack
    = iConcat [ iStr "Arg stack: [",
        iIndent (iInterleave iNewline (map showClosure stack)),
        iStr "]", iNewline
    ]

```

現時点では、値スタックとダンプを表示する `showValueStack` と `showDump` は、状態のこれらのコンポーネントを使用していないため、現時点ではスタックです。

```
showValueStack :: TimValueStack -> Iseq
showValueStack vstack = iNil
```

```
showDump :: TimDump -> Iseq
showDump dump = iNil
```

`showClosure` はクロージャを表示し、`showFramePtr` を使用してフレームポインタを表示します。

```
showClosure :: Closure -> Iseq
showClosure (i,f)
  = iConcat [ iStr "(", showInstructions Terse i, iStr ", ",
              showFramePtr f, iStr ")"
            ]
```

```
showFramePtr :: FramePtr -> Iseq
showFramePtr FrameNull      = iStr "null"
showFramePtr (FrameAddr a) = iStr (show a)
showFramePtr (FrameInt n)  = iStr "int " 'iAppend' iNum n
```

`showStats` は、蓄積された統計を出力します。

```
showStats :: TimState -> Iseq
showStats (instr, fptr, stack, vstack, dump, heap, code, stats)
  = iConcat [ iStr "Steps taken = ", iNum (statGetSteps stats), iNewline,
              iStr "No of frames allocated = ", iNum (hSize heap),
              iNewline
            ]
```

## 命令の表示

命令と命令シーケンスを表示する必要があります。一連の命令が1つの長い行として出力されると、かなり読みにくくなるため、それらをきれいに出力するコードを作成する価値があります。

実際、命令シーケンスのコード全体 (たとえば、スーパーコンビネータの定義を出力する場合) を出力できるようにするか、一部の省略形だけを出力できるようにしたいと考えています。後者の例は、スタックの内容を表示す

るときに発生します。各クロージャのコードの一部を確認することは役に立ちますが、すべてを確認する必要はありません。したがって、各関数に追加の引数 *d* を与えて、どの程度完全に出力するかを伝えます。この引数の値は、Full、Terse、または None のいずれかです。

```
data HowMuchToPrint = Full | Terse | None
```

`showInstructions` は命令のリストを `iseq` に変換します。*d* が None の場合、省略記号のみが出力されます。*d* が Terse の場合、命令はすべて 1 行に出力され、ネストされた命令は *d* を None として出力されます。*d* が Full の場合、命令は 1 行に 1 つずつレイアウトされ、完全に出力されます。

```
showInstructions :: HowMuchToPrint -> [Instruction] -> Iseq
showInstructions None il = iStr "{..}"
showInstructions Terse il
  = iConcat [iStr "{", iIndent (iInterleave (iStr ", ") body), iStr "}"]
  where
    instrs = map (showInstruction None) il
    body | length il <= nTerse = instrs
          | otherwise          = (take nTerse instrs) ++ [iStr ".."]
showInstructions Full il
  = iConcat [iStr "{ ", iIndent (iInterleave sep instrs), iStr " }"]
  where
    sep = iStr ", " `iAppend` iNewline
    instrs = map (showInstruction Full) il
```

`showInstruction` は、単一の命令を `iseq` に変換します。

```
showInstruction d (Take m) = (iStr "Take ") `iAppend` (iNum m)
showInstruction d (Enter x) = (iStr "Enter ") `iAppend` (showArg d x)
showInstruction d (Push x) = (iStr "Push ") `iAppend` (showArg d x)

showArg d (Arg m) = (iStr "Arg ") `iAppend` (iNum m)
showArg d (Code il) = (iStr "Code ") `iAppend` (showInstructions d il)
showArg d (Label s) = (iStr "Label ") `iAppend` (iStr s)
showArg d (IntConst n) = (iStr "IntConst ") `iAppend` (iNum n)
```

`nTerse` は、シーケンスのいくつかの命令を簡潔な形式で出力する必要があるかを示します。

```
nTerse = 3
```

演習 4.1 `main` の次の定義を使用してマシンを実行します。

```
main = S K K 4
```

S K K は恒等関数であるため、main は 4 に減少し、マシンが停止します。もう少し手の込んだものにしてみてください。例えば

```
id = S K K ;
id1 = id id ;
main = id1 4
```

演習 4.2 さらに性能測定を追加します。例えば：

- 実行時間を測定し、Take を除く命令ごとに 1 つの時間単位をカウントします。Take は、フレームに含まれる要素と同じ数の時間単位をカウントする必要があります。
- 実行で割り当てられたヒープの合計量を出力して、ヒープ使用量を測定します。フレームのサイズを数えて、結果をテンプレートインスタンス化バージョンの結果と直接比較できるようにします。
- 最大スタック深さを測定します。

演習 4.3  $n = 0$  の場合、Take  $n$  は何も役に立ちません。CAF の Take 命令を完全に省略することで、compileSC の定義を調整して、この最適化を見つけます。

#### 4.2.6 ガベージコレクション †

ヒープベースのシステムと同様に、TIM にはガベージコレクタが必要ですが、少し洗練されたものも必要です。いつものように、ガベージコレクタはマシンの状態から開始して、すべてのライブデータを検出します。つまり、スタックとフレームポインタからです。スタック上の各クロージャはフレームを指しており、明確に保持する必要があります。しかし、そのフレームには、さらに別のフレームへのポインタが含まれています。問題が発生します：特定のフレームが与えられた場合、その中のどのフレームポインタを再帰的に追跡する必要がありますか？

安全な答えは「それらすべてに従う」ことですが、これには必要以上のデータを保持するリスクがあります。たとえば、セクション 4.1.4 の compose2 の例の  $(g \ x \ x)$  のクロージャには、 $f$ 、 $g$ 、および  $x$  を含むフレームへのポインタがありますが、 $g$  と  $x$  のクロージャのみが必要です。ナイーブなガベージコレクタは、 $f$  のクロージャからのフレームポインタもたどり、データを不必

要に保持する可能性があります。この不要な保持はスペースリークと呼ばれ、ガベージコレクションが他の場合よりも頻繁に発生する可能性があります。

ただし、この特定のスペースリークは、かなり面倒ではありますが、解消するのは簡単です。各クロージャは、フレームポインタとペアになったコードポインタで構成されます。コードは、どのフレーム要素が必要になるかを「認識」しており、この情報をコードに記録して、ガベージコレクターが調査できるようにします。たとえば、コードポインタと呼んでいるものは、実際には、コードで使用されるスロット番号のリストとコード自体で構成されるペアを指している可能性があります。(実際の実装では、リストはビットマスクとしてエンコードされる場合があります。) 有用なスロットのリストはどのように導出できますか? それは簡単です: コンパイルされている式の自由変数を見つけ、環境を使用してそれらをスロット番号にマップするだけです。

### 4.3 Mark 2: 算術演算の追加

このセクションでは、マシンに算術演算を追加します。

#### 4.3.1 概要: 算術演算の仕組み

オリジナルの Fairbairn と Wray の TIM マシンは、算術を行うためのかなり悪質なスキームを持っていました。彼らの主な動機は、マシンを最小限に抑えることでしたが、彼らのアプローチは非常に理解しにくく、効率的な実装を行うにはかなりの修正が必要です。

代わりに、G マシンの V スタックとまったく同じように TIM を変更します (セクション 3.9)。(評価され、ボックス化されていない) 整数のスタックである値スタックを導入することにより、状態を変更します。値スタックの最上位要素で算術演算  $op$  を実行し、値スタックの最上位に結果を残す命令  $Op\ op$  ファミリーで命令セットを拡張します。たとえば、 $Op\ Sub$  命令は、値スタックの上位 2 つの要素を削除し、それらを減算して、結果を値スタックにプッシュします。

(4.10)	$Op\ Sub : i \quad f \quad s \quad n_1 : n_2 : v \quad h \quad c$
	$\implies i \quad f \quad s \quad (n_1 - n_2) : v \quad h \quad c$

このようにして、 $Op\ Add$ 、 $Op\ Sub$ 、 $Op\ Mult$ 、 $Op\ Div$ 、 $Op\ Neg$  などの算術命令の完全なファミリを簡単に定義できます。

次に、以下の関数 `sub` を考えてみましょう。

`sub a b = a - b`

`sub` 用にどのコードを生成する必要がありますか? 次の手順を実行する必要があります。

1. 引数をフレームに形成する通常の `Take 2`。
2. `b` を評価し、その値を値スタックに置きます。
3. 同様に `a` を評価します。
4.  $Op\ Sub$  命令を使用して `a` の値から `b` の値を引き、その結果を値スタックの一番上に残します。
5. 「呼び出し元」に「戻る」。



まず、a と b の評価を考えます。それらはクロージャによって表され、現在のフレームに保持されます。クロージャに対してできることは、それを入力することだけです。おそらく a を評価するには、a のクロージャを入力する必要がありますが、整数値のクロージャを入力するとはどういう意味でしょうか？ これまで関数を入力しただけで、整数は関数ではありません。重要なアイデアは次のとおりです。

整数不変: 整数値のクロージャが入力されると、整数の値が計算され、それが値スタックにプッシュされ、引数スタックの一番上のクロージャが入力されます。

引数スタックの一番上にあるクロージャは、整数の評価が完了したら次に何をすべきかを示しているため、継続と呼ばれます。継続は、整数の評価が完了したときに何をすべきかを示す命令シーケンスと、現在のフレームポインタ (整数の評価によって妨害された場合) で構成されます。つまり、継続は完全に通常のクロージャです。

したがって、sub のコードは次のようになります。

```
sub: Take 2
      Push (Label L1)  -- Push the continuation
      Enter (Arg 2)    -- Evaluate b

L1:   Push (Label L2)  -- Push another continuation
      Enter (Arg 1)    -- Evaluate a

L2:   Op Sub           -- Compute a-b on value stack
```

Return 命令は何をすべきですか？ sub によって返される値は整数であり、Op Sub 命令の後、この整数は値スタックの一番上にあるため、Return がしなければならないことは、引数スタックの一番上にクロージャをポップして入力することだけです。

(4.11)	$\begin{array}{c} \text{[Return]} \quad f \quad (i', f') : s \quad v \quad h \quad c \\ \Rightarrow \quad \quad \quad i' \quad f' \quad \quad \quad s \quad v \quad h \quad c \end{array}$
--------	--

ラベルを使用して、sub のコードを記述しました。これが唯一の方法ではありません。別の方法として、Push Code 命令を使用する方法があります。これにより、面倒な新しいラベルを作成する必要がなくなります。このスタイルでは、sub のコードは次のようになります。

```
sub: Take 2
```

```

Push (Code [ Push (Code [Op Sub, Return]),
               Enter (Arg 1)
            ])
Enter (Arg 2)

```

このように書かれていると、ラベルを使用するよりも何が起きているかを確認するのが簡単ではないため、コードの断片を理解しやすくするために説明では引き続きラベルを使用しますが、コンパイラでは Push Code パージョンを使用します。

ここで、整数定数の問題に戻る必要があります。(sub 4 2) という式を考えてみましょう。以下のコードにコンパイルされます

```

Push (IntConst 2)
Push (IntConst 4)
Enter (Label "sub")

```

sub のコードはすぐにクロージャ(IntConst 2) に入り、フレームポインタに整数 2 を配置して intCode にジャンプします。現在、intCode は空のコードシーケンスです(整数を入力するとマシンが停止します)が、これを変更する必要があります。intCode は今何をすべきか? 答えは整数不変式によって与えられます: 整数を値スタックにプッシュして返す必要があるため、次のようになります。

```
intCode = [PushV FramePtr, Return]
```

PushV FramePtr は、現在フレームポインタとして偽装されている数値を値スタックの一番上にプッシュする新しい命令です。

(4.12)	$\text{PushV FramePtr} : i \quad n \quad s \quad v \quad h \quad c$
$\Rightarrow$	$i \quad n \quad s \quad n : v \quad h \quad c$

#### 4.3.2 実装への単純な算術演算の追加

これで、実装を変更する準備が整いました。各算術関数のコードを compiledPrimitives に追加することで、変更を最小限に抑えます。プログラムで (たとえば)p-q と書くと、パーサはそれを次のように変換することを思い出してください。

```
EAp (EAp (EVar "-") (EVar "p")) (EVar "q")
```

プリミティブに適したコードを作成し、このコードをコードストアに追加するだけです。コンパイラは、他のスーパーコンビネータと同じように-

扱います。最後に、必要な-のコードは、前のセクションで作成した `sub` とまったく同じであり、他の算術演算についても同様のコードを簡単に作成できます。

したがって、必要な手順は次のとおりです。

- 次の型定義と値スタックの初期化を追加します。

```
type TimValueStack = [Int]
initialValueStack = []
```

- 新しい命令 `PushV`、`Return`、および `Op` を `instruction` 型に追加します。この機会に、`Cond` という命令を 1 つ追加します。これについてはまだ説明していませんが、後の演習の対象となります。TIM はもはや 3 命令マシンではありません！

```
data Instruction = Take Int
                | Push TimAMode
                | PushV ValueAMode
                | Enter TimAMode
                | Return
                | Op Op
                | Cond [Instruction] [Instruction]

data Op = Add | Sub | Mult | Div | Neg
        | Gr | GrEq | Lt | LtEq | Eq | NotEq
        deriving (Eq) -- KH
```

これまでのところ、`PushV` 命令の引数は `FramePtr` のみですが、リテラル定数を値スタックにプッシュできるようにする 2 番目の形式をすぐに追加します。したがって、`valueAMode` の代数データ型を宣言する価値があります。

```
data ValueAMode = FramePtr
                | IntVConst Int
```

この追加構造を処理するには、`showInstruction` 関数を変更する必要があります。

- `step` 関数を変更して、追加の命令を実装します。これは、状態遷移規則を `Miranda` に翻訳する問題です。

- +、-などの適切な定義を `compiledPrimitives` に追加します。
- `intCode` が空でなくなったので、スタックを初期化して、`main` が戻るための適切な継続 (戻りアドレス) を持たせる必要があります。これを行う方法は、`initialArgStack` を再定義することにより、`compile` にクロージャ `([], FrameNull)` でスタックを初期化させることです。

```
initialArgStack = [([], FrameNull)]
```

この継続には空のコードシーケンスがあるため、マシンは結果を値スタックの上に置いて停止します。

演習 4.4 これらの変更をプロトタイプに実装します。簡単な例で試してみてください。例えば

```
four = 2 * 2
main = four + four
```

演習 4.5 まだ条件式がなく、条件式がなければ再帰を使用できないため、まだ「興味深い」プログラムを実行できません。簡単な解決策は、新しい命令 `Cond i1 i2` を追加することです。この命令は、値スタックの一番上から値を削除し、それがゼロかどうかをチェックし、ゼロである場合は命令シーケンス `i1` を続行し、そうでない場合は `i2` を続行します。その状態遷移規則は次のとおりです。

(4.13)	$\Rightarrow$	$\begin{array}{ccccccc} [\text{Cond } i_1 & i_2] & f & s & 0 : v & h & c \\ & & i_1 & f & s & v & h & c \end{array}$
(4.13)	$\Rightarrow$	$\begin{array}{ccccccc} [\text{Cond } i_1 & i_2] & f & s & n : v & h & c \\ & & i_2 & f & s & v & h & c \end{array}$ <p>ここで <math>n \neq 0</math></p>

最初のルールは、ゼロが値スタックの一番上にある場合に一致します。それ以外の場合は、2 番目の規則が適用されます。次のように動作するプリミティブ `if` も追加する必要があります。

```
if 0 t f = t
if n t f = f
```

Cond 命令を使用して if の TIM コードを作成し、それを `compiledPrimitives` に追加する必要があります。最後に、階乗関数を使用して改善されたシステムをテストできます。

```
factorial n = if n 1 (n * factorial (n-1))
main = factorial 3
```

#### 4.3.3 算術演算のコンパイルスキーム

G マシンの場合と同様に、現在行っているマシン用のコンパイルよりもはるかに優れた仕事を行うことができます。次のような関数を考えてみましょう

```
f x y z = (x+y) * z
```

現状では、これは次のように解析されます

```
f x y z = * (+ x y) z
```

`f` のコードがコンパイルされ、標準関数 `*` および `+` が呼び出されます。しかし、これよりもはるかに良いことができます! たとえば、`(+ x y)` のクロージャを作成して `*` に渡す代わりに、次の手順を使用してインラインで操作を行うことができます。

1. `x` を評価する
2. `y` を評価する
3. それらを加算する
4. `z` を評価する
5. 乗算する
6. 戻る

クロージャを構築する必要はなく、ジャンプを行う必要もありません (`x`、`y`、`z` を評価するために必要なものを除く)。

この改善を表現するために、値が整数である式を処理する新しいコンパイルスキーム、*B* スキームを導入します。次のように定義されます: 値が整数である任意の式  $e$  と、任意のコードシーケンス  $cont$  に対して、

$(B[e] \ \rho \ cont)$  は、現在のフレームが  $\rho$  で記述されているようにレイアウトされた状態で実行されると、式  $e$  の値を値スタックにプッシュしてから、コードシーケンス  $cont$  を実行するコードシーケンスです。

コンパイルスキームは継続渡しスタイルを使用します。このスタイルでは、 $cont$  引数が、値が計算された後に何をすべきかを示します。

図 4.2 は、改訂された  $\mathcal{R}$  および  $\mathcal{A}$  スキームと共に、 $\mathcal{B}$  コンパイルスキームを示しています。 $\mathcal{R}$  が算術式である式を見つけると、それをコンパイルするために  $\mathcal{B}$  を呼び出します。 $\mathcal{B}$  には、継続を明示的にプッシュすることを回避する、定数および算術演算子の適用に関する特殊なケースがあります。特別に処理できない式に遭遇した場合は、継続をプッシュして  $\mathcal{R}$  を呼び出すだけです。

$\mathcal{R}[e] \ \rho$  は、環境  $\rho$  の式  $e$  の値をスタック上の引数に適用する TIM コードです。

$$\mathcal{R}[e] \ \rho = \mathcal{B}[e] \ \rho \ [\text{Return}]$$

ここで、 $e$  は  $e_1 + e_2$  などの算術式、または数値です。

$$\mathcal{R}[e_1 \ e_2] \ \rho = \text{Push}(\mathcal{A}[e_2] \ \rho) : \mathcal{R}[e_1] \ \rho$$

$$\mathcal{R}[a] \ \rho = \text{Enter}(\mathcal{A}[a] \ \rho)$$

ここで、 $a$  は整数、変数、またはスーパーコンビネータです。

$\mathcal{A}[e] \ \rho$  は、環境  $\rho$  における式  $e$  の TIM アドレッシングモードです。

$$\mathcal{A}[x] \ \rho = \rho \ x \quad \text{ここで、} x \text{ は } \rho \text{ によって束縛されます。}$$

$$\mathcal{A}[n] \ \rho = \text{IntConst } n \quad \text{ここで、} n \text{ は整数定数です。}$$

$$\mathcal{A}[e] \ \rho = \text{Code}(\mathcal{R}[e] \ \rho) \quad \text{その他}$$

$\mathcal{B}[e] \ \rho$  は、環境  $\rho$  で式  $e$  を評価して、その値 (整数でなければならない) を値スタックの一番上に置き、コードシーケンス  $cont$  を続行する TIM コードです。

$$\mathcal{B}[e_1 + e_2] \ \rho \ cont = \mathcal{B}[e_2] \ \rho (\mathcal{B}[e_1] \ \rho \ (\text{Op Add} : cont))$$

... および他の算術プリミティブの同様の規則

$$\mathcal{B}[n] \ \rho \ cont = \text{PushV}(\text{IntVConst } n) : cont \quad \text{ここで、} n \text{ は数値です。}$$

$$\mathcal{B}[e] \ \rho = \text{Push}(\text{Code } cont) : \mathcal{R}[e] \ \rho \quad \text{その他}$$

必要な新しい命令が 1 つあります。これは、 $\mathcal{B}$  が定数をコンパイルするように要求されたときに使用されます。その時には、整数定数を値スタックにプッシュする命令  $\text{PushV}(\text{IntVConst } n)$  が必要です。その遷移規則は非常に単純です。

(4.14)	$\text{PushV (IntVCont } n) : i \quad f \quad s \quad v \quad h \quad c$
	$\implies i \quad f \quad s \quad n : v \quad h \quad c$

演習 4.6 改善されたコンパイル スキームを実装します。実装のパフォーマンスを以前のパフォーマンスと比較します。

演習 4.7  $\mathcal{R}$  スキームに新しいルールを追加して、`if` の (完全な) アプリケーションに一致させます。`if` プリミティブを呼び出して取得するよりもはるかに優れたコードを生成できるはずです。変更を実装し、パフォーマンスの改善を測定します。

演習 4.8 次のように、より一般的な算術比較を処理するために条件を一般化するとします。

```
fib n = if (n < 2) 1 (fib (n-1) + fib (n-2))
```

必要なのは、値スタックの上位 2 つの項目をポップし、それらと比較し、比較の結果に応じて値スタックに 1 または 0 をプッシュする新しい命令 `Op Lt` です。次に、`Cond` 命令がこの結果を検査します。

このような比較命令のファミリを実装し、他の算術演算子とまったく同じ方法で、それらの特殊なケースを  $\mathcal{B}$  スキームに追加します。あなたの改善をテストします。

演習 4.9 前の練習問題で、追加の比較モードを持つように `Cond` 命令を変更しなかった理由を疑問に思ったかもしれません。次に、このモードに従って値スタックの上位 2 つの項目を比較し、それに応じて動作します。なぜこれをしなかったのですか？ ヒント: このようなプログラムはどうなるでしょうか？

```
multipleof3 x = ((x / 3) * 3) == x
f y = if (multipleof3 y) 0 1
```

このセクションの内容は [Argo 1989] で説明されており、[Peyton Jones 1987] の第 20 章で説明されている改良された G マシンコンパイルスキームに正確に対応しています。

#### 4.4 Mark 3: let(rec) 式

現在、コンパイラはlet(rec) 式を処理できません。これは、このセクションで解決する問題です。2つの主要な新しいアイデアが導入されました。

- Take 命令を変更して、余分なスペースをフレームに割り当てて、let(rec) 束縛変数と仮パラメータを含めます。
- 間接クロージャの考え方を紹介します。

##### 4.4.1 let 式

let 式をコンパイルするとき、定義の右側の新しいクロージャを構築するコードを生成する必要があります。これらの新しいクロージャはどこに配置する必要がありますか？let(rec) で束縛された名前を引数名と同じように扱うには、それらを現在のフレームに配置する必要があります<sup>12</sup>。これには、ランタイム機構に2つの変更が必要です。

- Take 命令は、スーパーコンビネータの実行中に発生するすべてのlet 定義のクロージャを含むのに十分な大きさのフレームを割り当てる必要があります。Take 命令は、Take  $t\ n$  の形式に変更する必要があります。ここで、 $t \geq n$  です。この命令は、サイズ  $t$  のフレームを割り当て、スタックの一番上から  $n$  個のクロージャを取得し、それらをフレームの最初の  $n$  個の場所に配置します。
- 新しいクロージャ  $a$  を現在のフレームのスロット  $i$  に移動するには、新しい命令 Move  $i\ a$  が必要です。ここで  $a$  は、Push および Enter と同様に timAMode 型です。

たとえば、次の定義です。

```
f x = let y = f 3 in g x y
```

これは次のコードにコンパイルされます。

```
[ Take 2 1,  
  Move 2 (Code [Push (IntConst 3), Enter (Label "f")]),  
  Push (Arg 2),  
  Push (Arg 1),  
  Enter (Label "g")  
]
```

<sup>12</sup>このセクションの資料のヒントは [Wakeling and Dix 1989] にありますが、完全には解明されていません。



もう少し複雑な例を次に示します。

```
f x = let y = f 3
      in
      g (let z = 4 in h z) y
```

これは次のコードを生成します：

```
[ Take 3 1,
  Move 2 (Code [Push (IntConst 3), Enter (Label "f")]),
  Push (Arg 2),
  Push (Code [Move 3 (IntConst 4), Push (Arg 3), Enter (Label "h")]),
  Enter (Label "g")
]
```

スーパーコンピネータ本体のクロージャが必要とするすべてのスロットに、最初の Take がスペースを割り当てる方法に注目してください。

演習 4.10 新しい Take および Move 命令の状態遷移規則を記述します。

次に、新しい Take 命令と Move 命令を生成するようにコンパイラを変更する必要があります。let 式に遭遇した場合、フレーム内の空きスロットを各バインド変数に割り当てる必要があるため、フレーム内のどのスロットが使用中で、どのスロットが空きであるかを追跡する必要があります。これを行うには、追加のパラメータ  $d$  を各コンパイル スキームに追加して、 $d + 1$  以降のフレームスロットが空いていることを記録しますが、1 から  $d$  までのスロットは占有される可能性があることを記録します。

残りの複雑な点は、最初の Take 命令で十分な大きさのフレームを割り当てることができるように、 $d$  が取り得る最大値を発見する必要があることです。これには、各コンパイルスキームが、コンパイルされたコードと  $d$  によって取得される最大値のペアを返す必要があります。新しいコンパイルスキームを図 4.3 に示します。(この図とその後で、 $is_1 ++ is_2$  という表記を使用して、命令シーケンス  $is_1$  と  $is_2$  の連結を示します。)

$SC\llbracket e \rrbracket \rho$  は、環境  $\rho$  でコンパイルされたスーパーコンビネータ定義  $def$  の TIM コードです。

$$SC\llbracket f \ x_1 \ \dots \ x_n = e \rrbracket \rho = \text{Take } d' \ n : is$$

ここで

$$(d', is) = \mathcal{R}\llbracket e \rrbracket \rho[x_1 \mapsto \text{Arg } 1, \dots, x_n \mapsto \text{Arg } n] \ n$$

です。

$\mathcal{R}\llbracket e \rrbracket \rho \ d$  は、ペア  $(d', is)$  です。

$is$  は、環境  $\rho$  内の式  $e$  の値をスタック上の引数に適用する TIM コードです。  
このコードは、フレームの最初の  $d$  スロットが占有されていると想定しており、  
スロット  $(d + 1 \dots d')$  を使用します。

$$\mathcal{R}\llbracket e \rrbracket \rho \ d = \mathcal{B}\llbracket e \rrbracket \rho \ d [\text{Return}]$$

ここで、 $e$  は  $e_1 + e_2$  などの算術式、または数値です。

$$\begin{aligned} \mathcal{R}\llbracket \text{let } x_1 = e_1; \dots; x_n = e_n \text{ in } e \rrbracket \rho \ d \\ = (d', [\text{Move } (d + 1) \ am_1, \dots, \text{Move } (d + m) \ am_n] \ ++ \ is) \end{aligned}$$

ここで

$$\begin{aligned} (d_1, am_1) &= \mathcal{A}\llbracket e_1 \rrbracket \rho \ (d + n) \\ (d_2, am_1) &= \mathcal{A}\llbracket e_2 \rrbracket \rho \ d_1 \\ &\dots \\ (d_n, am_n) &= \mathcal{A}\llbracket e_n \rrbracket \rho \ d_{n-1} \ ++ \\ \mathcal{R}\llbracket e \rrbracket \rho' \ d_n \\ \rho' &= \rho^{+n} [x_1 \mapsto \text{Arg } (d + 1), \dots, x_n \mapsto \text{Arg } (d + n)] \end{aligned}$$

です。

$$\mathcal{R}\llbracket e_1 \ e_2 \rrbracket \rho \ d = (d_2, \text{Push } am : is)$$

ここで

$$\begin{aligned} (d_1, am) &= \mathcal{A}\llbracket e_2 \rrbracket \rho \ d \\ (d_2, is) &= \mathcal{R}\llbracket e_1 \rrbracket \rho \ d_1 \end{aligned}$$

です。

$$\mathcal{R}\llbracket a \rrbracket \rho \ d = (d', [\text{Enter } am])$$

ここで、 $a$  は整数、変数、またはスーパーコンビネータであり、  
 $(d', am) = \mathcal{A}\llbracket a \rrbracket \rho \ d$  です。

$\mathcal{A}\llbracket e \rrbracket \rho$  は、ペア  $(d', am)$  です。

ここで、 $am$  は環境  $\rho$  における式  $e$  の TIM アドレッシングモードです。  
このコードは、フレームの最初の  $d$  スロットが占有されていると想定しており、  
スロット  $(d + 1 \dots d')$  を使用します。

$$\begin{aligned} \mathcal{A}\llbracket x \rrbracket \rho \ d &= (d, \rho \ x) && \text{ここで、} x \text{ は } \rho \text{ によって束縛されます。} \\ \mathcal{A}\llbracket n \rrbracket \rho \ d &= (d, \text{IntConst } n) && \text{ここで、} n \text{ は整数です。} \\ \mathcal{A}\llbracket e \rrbracket \rho \ d &= (d', \text{Code } is) && \text{その他} \end{aligned}$$

ここで、 $(d', is) = \mathcal{R}\llbracket e \rrbracket \rho \ d$  です。

SC スキームでは、スーパーコンビネータ本体のコンパイルから返された最大フレームサイズ  $d'$  を使用して、実行する Take の大きさを決定する方法を確認できます。 $\mathcal{R}$  スキームの let 式の場合、定義ごとに命令 `Move  $i$   $a$`  を生成します。ここで、 $i$  は現在のフレームの空きスロットの番号であり、 $a$  は  $e$  を  $\mathcal{A}$  スキームでコンパイルした結果です。各右辺のコンパイルに、前の右辺によって占有されていた最後のスロットのインデックスが与えられる方法に注意してください。これにより、すべての右辺が異なるスロットを使用することが保証されます。

演習 4.11 このセクションで説明する変更を実装します。新しい命令を `instruction` 型に追加し、新しいケースを `step` と `showInstruction` に追加してそれら进行处理し、新しいコンパイルスキームを実装します。

演習 4.12 以下のプログラムを検討します。

```
f x y z = let p = x+y in p+x+y+z
main = f 1 2 3
```

let 式がない場合は、次のように補助関数を使用して記述する必要があります。

```
f' p x y z = p+x+y+z
f x y z = f' (x+y) x y z
main = f 1 2 3
```

これら 2 つのプログラムによって生成されたコードを比較し、消費されたストアと実行されたステップの違いを測定します。let 式を直接実装することで得られる主な節約は何ですか？

#### 4.4.2 letrec 式

letrec 式も処理するにはどうすればよいでしょうか？ 最初はとても簡単に思えます。 $\mathcal{R}$  スキームの letrec ケースは、 $am_i$  の定義で  $\rho$  を  $\rho'$  に置き換える必要があることを除いて、let ケースとまったく同じです。これは、 $x_i$  がそれぞれの右辺のスコープ内にあるためです。

演習 4.13 この追加のケースを `compileR` に実装し、プログラム上で試してみます。

```

f x = letrec p = if (x==0) 1 q ;
      q = if (x==0) p 2
      in p+q
main = f 1

```

生成されたコードを必ず理解し、テストしてください。

残念ながら、この実装には微妙なバグがあります。以下から生成されたコードを考えてみましょう。

```

f x = letrec a = b ;
      b = x
      in a

```

それは次のとおりです。

[Take 3 1, Move 2 (Arg 3), Move 3 (Arg 1), Enter (Arg 2)]

b のクロージャは、2 番目の Move によって割り当てられる前に、最初の Move によってコピーされます。

ここから抜け出す方法は 2 つあります。まず最初に、これは愚かなプログラムであると宣言します。b のバインディングのスコープ内で b を x に置き換えるだけです。しかし、後で有益になり、上記のようなばかげたプログラムでも動作できるようにする、より興味深いアプローチがあります。代わりに、最初の Move に対して次のコードを生成するとします。

Move 2 (Code [Enter (Arg 3)])

これですべてがうまくいきます。Enter (Arg 3) が実行される前にスロット 3 が割り当てられます。実際、クロージャ ([Enter (Arg 3)], f) はフレーム f のスロット 3 への間接指定と考えることができます。

このコードは、 $\mathcal{R}$  コンパイルスキームの let(rec) ケースを変更することで取得できます。これにより、let(rec) によってバインドされた変数ごとに環境内の間接アドレス指定モードが記録されます。図 4.3 を参照すると、必要な変更は  $\mathcal{R}$  スキームの let のルールに対するもので、 $\rho'$  の定義は次のようになります。

$\rho' = \rho[x_1 \mapsto \mathcal{I}[[d+1]], \dots, x_n \mapsto \mathcal{I}[[d+n]]]$  ここで  $\mathcal{I}[[d]] = \text{Code} [\text{Enter} (\text{Arg } d)]$

もちろん、これはかなり保守的です。多くの場合、その必要がない場合に間接参照を返しますが、結果として得られるコードは、効率は劣るものの、引き続き正常に動作します。

演習 4.14 このアイデアを実装するために `compileR` を変更し、上記の例に対して正しいコードが生成されることを確認します。新しい環境で間接アドレス指定モードを生成するには、`compileR` を変更する際に、補助関数 `mkIndMode` ( $\mathcal{I}$  スキームに対応) を使用します。

```
mkIndMode :: Int -> TimAMode
mkIndMode n = Code [Enter (Arg n)]
```

#### 4.4.3 フレームスロットの再利用 †

現在、スーパーコンビネータ定義の右側にあるすべての定義は、フレーム内に独自のプライベート スロットを取得します。場合によっては、異なる `let(rec)` 間でスロットを安全に共有できることがわかるかもしれません。たとえば、次の定義を考えてみましょう。

```
f x = if x (let ... in ...) (let ... in ...)
```

2 つの `let` 式のうち 1 つだけが評価できることは明らかなので、それらの定義に同じスロットを使用しても完全に安全です。

同様に、式  $e_1 + e_2$  では、 $e_1$  の評価中に使用された `let(rec)` スロットは、 $e_2$  が評価されるまでに終了します (または、 $+$  がその引数を逆の順序で評価した場合はその逆になります)。 $e_1$  内の `let(rec)` バインドされた変数は、 $e_2$  内の変数とスロットを共有できます。

演習 4.15 これを見つけて利用できるようにコンパイラを変更してください。

#### 4.4.4 ガベージ コレクション †

セクション 4.2.6 では、スペース リークを回避できるように、コード シーケンスによってどのフレーム スロットが使用されたかを記録することが望ましいと述べました。Take が割り当てた追加のフレーム スロットを初期化しない場合、ガベージ コレクターがそのコンテンツを処理する危険があります。これらの初期化されていないスロットは有効なポインターとして扱われ、予測できない結果が生じます。最も簡単な解決策はすべてのスロットを初期化することですが、これには非常にコストがかかります。セクション 4.2.6 の解決策を採用し、保持する必要があるスロットのリストを各コード シーケンスとともに記録する方が良いでしょう。初期化されていないスロットは、ガベージ コレクターによって参照されることはありません。

## 4.5 Mark 4: 更新

これまでのところ、共有 Redexes を繰り返し評価するため、グラフ削減ではなくツリー削減を実行してきました。これを修正する時期が来ました。TIM 更新を実行するさまざまな方法がどのように機能するかを正確に理解するのは少し難しいですが、少なくともプロトタイプの実装はあります。そのため、私たちの開発は非常に具体的なものになります。

### 4.5.1 基礎技術

標準的なテンプレートインスタンス化マシンと G マシンは、簡約のたびに更新を実行します。(G マシンには末尾呼び出しがいくつか最適化されていますが、原理は同じです。) TIM はスパインのないマシンであるため、その更新手法はかなり異なったものになる必要があります。重要なアイデアは次のとおりです。

G マシンとは異なり、簡約のたびに更新は実行されません。代わりに、クロージャの評価が開始されるとき (つまり、クロージャに入ったとき)、次の手順が実行されます。

- 現在のスタックと、入力されているクロージャのアドレスは、マシン状態の新しいコンポーネントであるダンプにプッシュされます。
- クロージャの評価が完了するとトリガーされる「マウストラップ」が設定されます。
- クロージャの評価は、空のスタックから開始して通常どおりに実行されるようになりました。
- マウストラップがトリガーされると、クロージャは通常形式で更新され、古いスタックがダンプから復元されます。

「ネズミ捕り」は次のとおりです。クロージャの評価は新しいスタック上で実行されるため、Return 命令が空のスタックを見つけるか、スーパーコンビネータが適用される引数が少なすぎるため、評価は最終的に停止する必要があります。この時点で、式は (頭部) 正規形に達しているため、更新を実行する必要があります。

まず、値が整数型であるクロージャの更新に焦点を当てましょう。クロージャに入る直前に、新しい命令 `PushMarker x` が実行されるようにします。これにより、ダンプに情報をプッシュすることで更新メカニズムがセットアップされます。具体的には、`PushMarker x` はダンプにプッシュします<sup>13</sup>。

<sup>13</sup>TIM に関する論文では、この操作は「更新マーカーのプッシュ」と呼ばれることがよくあります。これは、「マーカー」の下にある引数を使用しようとすると更新がトリガーされるようにスタックが「マーク」されているためです。

- 現在のスタック。
- 現在のフレーム ポインター。更新するクロージャを含むフレームを指します。
- フレーム内で更新されるクロージャのインデックス  $x$ 。

現在のスタックをダンプに保存したので、PushMarker は空のスタックを使用して続行します。その状態遷移ルールは次のとおりです。

(4.15)	$\text{PushMarker } x : i \quad f \quad s \quad v \quad d \quad h \quad c$
$\Rightarrow$	$i \quad f \quad [] \quad v \quad (f, x, s) : d \quad h \quad c$

しばらくすると、クロージャの評価が完了します。その値は整数であるため、その値は値スタックの最上位にあり、スタックの最上位の継続に戻ることを期待して Return 命令が実行されます。ただし、この時点ではスタックは空になります。これが更新のトリガーです。ダンプがポップされ、更新が行われ、復元されたスタックを使用して Return 命令が再実行されます。このアクションは、次の遷移ルールによって説明されます。

(4.16)	$[\text{Return}] \quad f \quad [] \quad n : v \quad (f_u, x, s) : d \quad h \quad c$
$\Rightarrow$	$[\text{Return}] \quad f \quad s \quad n : v \quad d \quad h' \quad c$ ここで $h' = h[f_u : \langle \dots, d_{x-1}, (\text{intCode}, n), d_{x+1}, \dots \rangle]$
(4.16)	$[\text{Return}] \quad f \quad (i, f') : s \quad n : v \quad d \quad h \quad c$
$\Rightarrow$	$i \quad f' \quad s \quad v \quad d \quad h \quad c$

最初のルールは、フレーム  $f_u$  内の  $x$  番目のクロージャの更新をクロージャ  $(\text{intCode}, n)$  で記述します。これは、入力するとすぐに  $n$  を値スタックにプッシュして戻ります (セクション 4.3.1 を参照)。さらに更新を実行する必要がある場合に備えて、Return 命令が再試行されることに注意してください。これは、ルールの右側のコード シーケンスが依然として  $[\text{Return}]$  であるという事実によって示されます。

2 番目のルールは、ルール 4.11 を再度書き直したものです。実行する更新がない場合についても説明します。スタックの一番上の継続は、プログラムカウンターと現在のフレーム ポインターにロードされます。

- 4.5.2 PushMarker 命令のコンパイル
- 4.5.3 更新メカニズムの実装
- 4.5.4 間接更新の問題
- 4.5.5 `let(rec)` にバインドされた共有変数の更新
- 4.5.6 間接チェーンの排除
- 4.5.7 部分的な関数適用の更新



## 4.6 Mark 5: 構造化データ

このセクションでは、代数データ型を TIM に追加する方法を学習します。セクション 2.8.3 で説明されている高階関数を使用すると、このセクションの内容をまったく使用せずにデータ構造を実装することができます。しかし、そうするのはむしろ非効率的です。代わりに、より一般的なデータ構造を処理できるように、算術演算に使用したアプローチを開発します。

### 4.6.1 一般的なアプローチ

### 4.6.2 データ構造の遷移規則とコンパイル方式

### 4.6.3 試してみる

### 4.6.4 リストの表示

### 4.6.5 データ構造を直接使用する †

## 4.7 Mark 6: CAF とコードストア †

前に述べたように (セクション 4.2.3)、コード ストアを名前とコード シーケンスの関連付けリストとして表すという決定は、CAF が更新されないことを意味します。代わりに、呼び出されるたびにコードが実行されるため、作業が重複する可能性があります。この余分な作業は避けたいのですが、TIM の解決策は以前の実装ほど簡単ではありません。

テンプレート インスタンス化マシンと G マシンの場合、解決策は各スーパーコンビネーターを表すためにヒープ内にノードを割り当てることでした。CAF が呼び出されると、redex のルートはスーパーコンビネータ ノード自体であるため、ノードはリダクションの結果 (つまり、スーパーコンビネータ定義の右側のインスタンス) で更新されます。その後スーパーコンビネータを使用すると、元のスーパーコンビネータではなく、この更新されたノードが表示されます。問題は、TIM にはヒープ ノードがまったくないことです。ノードに対応するのはフレーム内のクロージャです。したがって、私たちがしなければならないことは、初期ヒープに、各スーパーコンビネータのクロージャを含む単一の巨大なフレーム、つまりグローバルフレームを割り当てることです。

コード ストアは、スーパーコンビネータ名をフレーム内のオフセットにマッピングする関連付けリスト  $g$  とともに、グローバル フレームのアドレス  $f_G$  によって表されます。Label アドレッシング モードは、この関連リストを使用してオフセットを見つけ、スーパーコンビネータのクロージャをグローバル フレームからフェッチします。Push Label の新しい遷移規則は、これらのアイデアを正式なものとします。

(4.21)

$$\begin{array}{c} \text{Push (Label } l) : i \quad f \quad s \quad h[f_G : \langle (i_1, f_1), \dots, (i_n, f_n) \rangle] \quad (f_G, g[l : k]) \\ \Rightarrow \quad i \quad f \quad (i_k, f_k) : s \quad h \quad (f_G, g) \end{array}$$

Enter Label のルールは、Push/Enter 関係から直接に従います。セクション 4.5.5 の let(rec) バインド変数のコンテキストで説明したように、グローバル フレーム内の各クロージャは自己更新クロージャです。let(rec) でバインドされた変数と同様に、スーパーコンビネータをスタックにプッシュするときは、(非更新の) 間接指定を使用する必要があります (セクション 4.5.5)。

### 4.7.1 CAF の実装

### 4.7.2 コードストアをより忠実にモデリングする

## 4.8 まとめ

最終的な TIM コンパイル スキームを図 4.7 と 4.8 にまとめます。明らかな疑問は、「TIM は G マシンより優れているのか、それとも劣っているのか?」ということです。答えるのは難しいです。私たちのプロトタイプは、設計の選択を検討するには非常に役立ちますが、本格的なパフォーマンスの比較には全く役に立ちません。たとえば、G マシン M<sub>kap</sub> と比較した Take 命令の相対コストをどのように確立できるでしょうか? 実際に比較できる唯一の尺度は、この 2 つのヒープ消費量です。

それでも、この章で行ったように、別の評価モデルを検討することは、TIM の側面と G マシンの側面を組み合わせた他の設計手段を示唆するため、非常に啓発的です。そうする試みの 1 つが Spineless Tagless G マシンです [Peyton Jones and Salkild 1989, Peyton Jones 1991]。これは TIM のスパインレスと更新メカニズムを採用していますが、そのスタックは (G マシンと同様に) ヒープオブジェクトへのポインタで構成されています。(TIM のような) コードとフレームのペアではありません。

```
module ParGM where
import Utils
import Language
--import GM
```

## 5 並列 G マシン

### 5.1 Introduction

この章では、並列 G マシン用の抽象マシンとコンパイラを開発します。これは、ESPRIT プロジェクト 415 の一部として開発された並列 G マシンの簡易バージョンに基づいています。興味のある読者は、簡単にアクセスできる [Kingdon et al 1991] を参照してください。[Peyton Jones 1989] では、並列グラフ簡約の一般的な紹介を見つけることができます。

#### 5.1.1 Parallel functional programming

#### 5.1.2 Parallel graph reduction

## **5.2 Mark 1: A minimal parallel G-machine**

### **5.2.1 Data type definitions**

### **5.2.2 The evaluator**

### **5.2.3 Compiling a program**

### **5.2.4 Printing the results**

## **5.3 Mark 2: The evaluate-and-die model**

### **5.3.1 The node data structure**

### **5.3.2 The instruction set**

## **5.4 Mark 3: A realistic parallel G-machine**

### **5.4.1 Scheduling policy**

### **5.4.2 Conservative and speculative parallelism**



## **5.5 Mark 4: A better way to handle blocking**

### **5.5.1 Data structures**

### **5.5.2 Locking and unlocking**

## 5.6 Conclusions

## 6 ラムダリフティング

この章では、本書で以前に見たマシンに受け入れられる一連のプログラムを拡張する方法を見ていきます。導入する拡張機能は、ローカル関数定義を許可することです。次に、代替アプローチに直面します。

- 環境に対処するためのメカニズムをマシンに追加します。
- ローカル関数定義がないようにプログラムを変換します。代わりに、すべての関数がスーパー コンビネータとして定義されます。

この本では、常に 2 番目のアプローチが取られると想定してきました。

この章は、完全な laziness の概念を紹介する適切なポイントでもあります。ここでも、関数型言語のこの望ましい最適化は、プログラム変換を使用して実現されます。

### 6.1 Introduction

## 6.2 Improving the expr data type

```
module Lambda where
import Utils
import Language
```

## **6.3 Mark 1: A simple lambda lifter**

### **6.3.1 Implementing a simple lambda lifter**

### **6.3.2 Free variables**

### **6.3.3 Generating supercombinators**

### **6.3.4 Making all the variables unique**

### **6.3.5 Collecting supercombinators**

## **6.4 Mark 2: Improving the simple lambda lifter**

### **6.4.1 Simple extensions**

### **6.4.2 Eliminating redundant supercombinators**

### **6.4.3 Eliminating redundant local definitions**

## 6.5 Mark 3: Johnsson-style lambda lifter

### 6.5.1 implementation

### 6.5.2 Abstracting free variables in functions

### 6.5.3 A tricky point †

## 6.6 Mark 4: A separate full laziness pass

### 6.6.1 A review of full laziness

### 6.6.2 Full-lazy lambda lifting in the presence of `let(rec)s`

### 6.6.3 Full laziness without lambda lifting

### 6.6.4 A full lazy lambda lifter

### 6.6.5 Separating the lambdas

### 6.6.6 Adding level numbers

### 6.6.7 Identifying MFEs

### 6.6.8 Renaming variables

### 6.6.9 Floating `let(rec)` expressions



## **6.7 Mark 5: Improvements to full laziness**

### **6.7.1 Adding case expressions**

### **6.7.2 Eliminating redundant supercombinators**

### **6.7.3 Avoiding redundant full laziness**

## **6.8 Mark 6: Dependency analysis †**

### **6.8.1 Strongly connected components**

### **6.8.2 Implementing a strongly connected component algorithm**

### **6.8.3 A dependency analysis**

## 6.9 Conclusions

## 7 A ユーティリティモジュール

この Appendix では、本全体で使用されるさまざまな便利な型と関数の定義を示します。

```
module Utils where

-- The following definitions are used to make some synonyms for routines
-- in the Gofer prelude to be more Miranda compatible
shownum n = show n
hd :: [a] -> a
hd = head -- in Gofer standard prelude
tl :: [a] -> [a]
tl = tail -- in Gofer standard prelude
zip2 :: [a] -> [b] -> [(a,b)]
zip2 = zip -- in Gofer standard prelude
-- can't do anything about # = length, since # not binary.
```

### 7.1 The heap type

抽象データ型の heap と address は、実装ごとにガベージコレクションされたノードのヒープを表すために使用されます。

#### 7.1.1 Specification

#### 7.1.2 Representation

## **7.2 The association list type**

## **7.3 Generating unique names**

### **7.3.1 Representation**

## **7.4 Sets**

### **7.4.1 Representation**

## 7.5 Other useful function definitions

## 8 B コア言語プログラムの例

この Appendix では、この本で開発された実装のいくつかをテストするのに役立ついくつかのコア言語プログラムを示します。それらは、プレリュード (セクション 1.4) で定義された関数が定義されていると想定しています。

### 8.1 基本プログラム

このセクションのプログラムでは、整数定数と関数の適用のみが必要です。

#### 8.1.1 超基本テスト

このプログラムは、かなり早く値 3 を返すはずです!

```
main = I 3
```

次のプログラムでは、3 を返す前にさらにいくつかの手順が必要です。

```
id = S K K ;  
main = id 3
```

これはかなりの数の id の関数適用を作成します (いくつ?)。

```
id = S K K ;  
main = twice twice twice id 3
```

#### 8.1.2 更新のテスト

このプログラムは、更新を行うシステムと行わないシステムの違いを示すはずです。更新が発生した場合、(I I I) の評価は 1 回だけ行う必要があります。更新しないと、2 回行われます。

```
main = twice (I I I) 3
```

#### 8.1.3 もっと興味深い例

この例では、リストの関数表現 (セクション 2.8.3 を参照) を使用して 4 の無限リストを作成し、次にその 2 番目の要素を取ります。head と tail の関数 (hd と tl) は、引数が空のリストの場合、abort を返します。abort スーパーコンビネータは、無限ループを生成するだけです。



```
cons a b cc cn = cc a b ;  
nil      cc cn = cn ;  
hd list = list K abort ;  
tl list = list K1 abort ;  
abort = abort ;  
infinite x = cons x (infinite x) ;  
main = hd (tl (infinite 4))
```

## 8.2 let と letrec

更新が実装されている場合、`id1` の評価が共有されるため、このプログラムは実装されていない場合よりも少ないステップで実行されます。

```
main = let id1 = I I I
      in id1 id1 3
```

ネストされた `let` 式もテストする必要があります。

```
oct g x = let h = twice g
          in let k = twice h
              in k (k x) ;
main = oct I 4
```

次のプログラムは、以前の `cons` や `nil` などの定義に基づく「関数リスト」を使用して、`letrec` をテストします。

```
infinite x = letrec xs = cons x xs
              in xs ;
main = hd (tl (tl (infinite 4)))
```

## 8.3 算術演算

### 8.3.1 条件分岐なし

まずは、条件式を必要としない簡単なテストから始めます。

```
main = 4*5+(2-5)
```

この次のプログラムは、正しく動作させるために関数呼び出しが必要です。  
`twice twice` を `twice twice twice` または `twice twice twice twice` に置き換えてみてください。どのような結果になるかを予想してください。

```
inc x = x+1;
main = twice twice inc 4
```

再び関数型リストを使って、`length` 関数を書きます。

```
length xs = xs length1 0 ;
length1 x xs = 1 + (length xs) ;
main = length (cons 3 (cons 3 (cons 3 nil)))
```

### 8.3.2 条件分岐あり

条件分岐ができれば、ようやく「面白い」プログラムが書けるようになります。例えば、階乗。

```
fac n = if (n==0) 1 (n * fac (n-1)) ;
main = fac 5
```

次のプログラムは、ユークリッドのアルゴリズムを用いて、2つの整数の最大公約数を計算するものです。

```
gcd a b = if (a==b)
            a
            if (a<b) (gcd b a) (gcd b (a-b)) ;
main = gcd 6 10
```

`nfib` 関数は、その結果（整数）が、実行中に何回関数が呼び出されたかをカウントしてくれるから面白いです。そのため、結果を実行時間で割ると、1秒あたりの関数呼び出し回数で性能を表すことができます。その結果、`nfib` はベンチマークとしてかなり広く使われています。しかし、特定の実装の「`nfib-number`」は、様々な特殊な最適化に決定的に依存しているため、非常に大きな塩梅で受け取る必要があります。

```
nfib n = if (n==0) 1 (1 + nfib (n-1) + nfib (n-2)) ;
main = nfib 4
```

## 8.4 データ構造

このプログラムは、降順の整数のリストを返します。評価者は、プログラムの結果としてリストを期待する必要があります。cons と nil は、それぞれ Pack{2,2} と Pack{1,0} としてプレリユードに実装されることが期待されています。

```
downfrom n = if (n == 0)
               nil
               (cons n (downfrom (n-1))) ;
main = downfrom 4
```

次のプログラムは、エラトステネスのふるいを実装して素数の無限リストを生成し、結果リストの最初のいくつかの要素を取ります。出力が生成されたときにインクリメンタルに出力されるように調整する場合は、take の呼び出しを削除して、無限リストを出力するだけで済みます。

```
main = take 3 (sieve (from 2)) ;

from n = cons n (from (n+1)) ;

sieve xs = case xs of
             <1> -> nil ;
             <2> p ps -> cons p (sieve (filter (nonMultiple p) ps)) ;

filter predicate xs
  = case xs of
        <1> -> nil ;
        <2> p ps -> let rest = filter predicate ps
                      in
                      if (predicate p) (cons p rest) rest ;

nonMultiple p n = ((n/p)*p) ~= n ;

take n xs = if (n==0)
               nil
               (case xs of
                   <1> -> nil ;
                   <2> p ps -> cons p (take (n-1) ps))
```